Infinite 3D medium, Isotropic Point Source, Linearly-Anisotropic Scattering

Exponential Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



c - single-scattering albedo
Σt - extinction coefficient

r - radial position coordinate in medium (distance from point source at origin)
u = cos θ - direction cosine
b - anisotropy parameter

Namespace

In[2459]:=
```
Begin["inf3Disopointlinanisoscatter`"]
```

Out[2459]=
```
inf3Disopointlinanisoscatter`
```

Util

In[2463]:=
$$\text{SA}[d\_, r\_] := d \frac{\text{Pi}^{d/2}}{\text{Gamma}\left[\frac{d}{2}+1\right]} r^{d-1}$$

Diffusion modes

In[2464]:=
$$\text{diffusionMode}[v\_, d\_, r\_] := (2\pi)^{-d/2} r^{1-\frac{d}{2}} v^{-1-\frac{d}{2}} \text{BesselK}\left[\frac{1}{2}(-2+d), \frac{r}{v}\right]$$

Fluence: exact solution

In[2465]:=
$$\text{Alinearaniso}[c0\_, g\_, v\_] := \frac{v\,(1-v^2)}{c0\,\left(v^2-1+c0+3\,g\,(1-c0)\,(3-c0-3\,(1-c0)\,(1-g\,c0)/v^2)\right)};$$

$$\text{glinearaniso}[c0\_, g\_, u\_] := 1\Big/\left(\left(\frac{\text{Pi}\,c0\,u}{2}\left(1+3\,g\,(1-c0)\,u^2\right)\right)^2 + \left(1+3\,g\,c0\,(1-c0)\,u^2-\left(1+3\,g\,(1-c0)\,u^2\right)\frac{c0}{2}\,u\,\text{Log}\left[\frac{1+u}{1-u}\right]\right)^2\right);$$

$$\text{v0linearaniso}[c\_, g\_] := \text{ReplaceAll}\left[\text{Abs}[v], \text{FindRoot}\left[1+\frac{3\,g\,c\,(1-c)}{v^2}-\left(1+\frac{3\,g\,(1-c)}{v^2}\right)\frac{c}{2\,v}\,\text{Log}\left[\frac{1+v}{1-v}\right], \{v, 1.1\}\right]\right];$$

In[2468]:=
$$\phi\text{exact}[r\_, \Sigma t\_, c\_, b\_] := \frac{\#\,\Sigma t}{2\,\text{Pi}\,r}\,\text{Alinearaniso}[c, b/3, \#]\,\text{Exp}[-\#\,r\,\Sigma t] + \frac{\Sigma t}{4\,\text{Pi}\,r}\,\text{NIntegrate}\left[\frac{1}{u^2}\,\text{glinearaniso}[c, b/3, u]\,\text{Exp}\left[-\Sigma t\,\frac{r}{u}\right], \{u, 0, 1\}\right]\,\&\left[\text{v0linearaniso}[c, b/3]\right]$$

Fluence: Rigorous Diffusion Approximation

In[2469]:=
$$\phi\text{rigourousDiffusion}[r\_, \Sigma t\_, c\_, b\_] := \frac{\#\,\Sigma t}{2\,\text{Pi}\,r}\,\text{Alinearaniso}[c, b/3, \#]\,\text{Exp}[-\#\,r\,\Sigma t]\,\&\left[\text{v0linearaniso}[c, b/3]\right]$$

Fluence: Classical Diffusion Approximation

In[2740]:=
$$\phi\text{Diffusion}[r\_, \Sigma t\_, c\_, b\_] := \frac{e^{-r\sqrt{(1-c)\,(3-b\,c)}\;\Sigma t}\,(3-b\,c)\,\Sigma t}{4\,\pi\,r}$$

Fluence: Grosjean Modified Diffusion Approximation

In[2741]:=
$$\phi\text{Grosjean}[r\_, \Sigma t\_, c\_, b\_] := \frac{E^{-r\,\Sigma t}}{4\,\text{Pi}\,r^2} + \frac{c}{1-c}\,\frac{1}{\Sigma t}\,\text{diffusionMode}\left[\frac{1}{\sqrt{3}\,\sqrt{\frac{(c-1)\,(-3+b\,c)}{6+b\,(-1+c)^2-3\,c}}\;\Sigma t}, 3, r\right]$$

In[2475]:=
```
Clear[a, b, c, Σt, r];
FullSimplify[inf3Disopointlinanisoscatter`ϕGrosjean[r, Σt, c, b],
 Assumptions → Σt > 0 && 0 < c < 1 && b > -1 && b < 1]
```

Out[2475]=
$$\frac{e^{-r\,\Sigma t} - \frac{3\,c\,(-3+b\,c)\,e^{-\sqrt{3}\sqrt{\frac{(-1+c)\,(-3+b\,c)}{6+b\,(-1+c)^2-3\,c}}\;r\,\Sigma t}\;r\,\Sigma t}{6+b\,(-1+c)^2-3\,c}}{4\,\pi\,r^2}$$

Nth-collided fluence - Gaussian approximation

In[2476]:=
$$\text{twomomentGaussian}[r\_, m0\_, m2\_] := \frac{3\,\sqrt{\frac{3}{2}}\;e^{-\frac{3\,m0\,r^2}{2\,m2}}\;m0^{5/2}}{2\,m2^{3/2}\,\pi^{3/2}}$$

In[2477]:=
$$\phi\text{Gaussian}[r\_, \Sigma t\_, c\_, b\_, n\_] := \text{twomomentGaussian}\left[r, \frac{c^n}{\Sigma t}, \frac{2\times 3^{-n}\left(b^{2+n} + 3^{2+n}\,(1+n) - 3^{1+n}\,b\,(2+n)\right)c^n}{(-3+b)^2\,\Sigma t^3}\right]$$

load MC data

In[2687]:=
```
ppoints[xs_, dr_, maxx_] :=
 Table[{dr (i) - 0.5 dr, xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[2643]:=
```
ppointsu[xs_, du_, Σt_] :=
 Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[2644]:=
```
fs = FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/MCdata/
    inf3D_isotropicpoint_linanisoscatter*"];
```

In[2645]:=
```
index[x_] := Module[{data, c, mfp, b},
    data = Import[x, "Table"];
    mfp = data[[1, 13]];
    c = data[[2, 3]];
    b = data[[1, 16]];
    {c, mfp, b, data}];
simulations = index /@ fs;
cs = Union[#[[1]] & /@ simulations]
```

Out[2647]=
```
{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}
```

In[2648]:=
```
mfps = Union[#[[2]] & /@ simulations]
```

Out[2648]=
```
{0.3, 1}
```

In[2649]:=
```
bs = Union[#[[3]] & /@ simulations]
```

Out[2649]=
```
{-0.9, 0.7}
```

In[2650]:=
```
numcollorders = inf3Disopointlinanisoscatter`simulations[[1]][[-1]][[2, 13]];
```

Compare Deterministic and MC

Mean Track Length

In[2651]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]},
 {ActionMenu["Set mfp", "mfp = " <> ToString[#] :> (mfp = #;) & /@ mfps],
  Dynamic[mfp]},
 {ActionMenu["Set b", "b = " <> ToString[#] :> (b = #;) & /@ bs], Dynamic[b]}}
```

Out[2651]=
```
{{ Set c , 0.9}, { Set mfp , 1}, { Set b , 0.7}}
```

In[2652]:=
```
data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp && #[[3]] == b &][[4]];
meanTL = data[[-1]]
```

$$\frac{mfp}{1 - c}$$

Out[2653]=
```
{Mean, track, length:, 1.42865}
```

Out[2654]=
```
1.42857
```

Fluence - Exact solution comparison to MC

In[•]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]},
 {ActionMenu["Set mfp", "mfp = " <> ToString[#] :> (mfp = #;) & /@ mfps],
  Dynamic[mfp]},
 {ActionMenu["Set b", "b = " <> ToString[#] :> (b = #;) & /@ bs], Dynamic[b]}}
```

Out[•]=
```
{{ Set c , 0.9}, { Set mfp , 1}, { Set b , 0.7}}
```

In[2724]:=
```
data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp && #[[3]] == b &][[4]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
pointsFluence = ppoints[data[[6]], dr, maxr];
exact1FluenceShallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 ϕexact[#[[1]], 1/mfp, c, b]}] & /@
   pointsFluence[[1 ;; 60]];
exact1Fluence = Quiet[{#[[1]], 4 Pi #[[1]]^2 ϕexact[#[[1]], 1/mfp, c, b]}] & /@
   pointsFluence[[1 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
    ListPlot[pointsFluence[[1 ;; 60]],
     PlotRange → All, PlotStyle → PointSize[.01]],
    ListPlot[exact1FluenceShallow, PlotRange → All, Joined → True],
    Frame → True,
    FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[pointsFluence, PlotRange → All, PlotStyle → PointSize[.01]],
    ListLogPlot[exact1Fluence, PlotRange → All, Joined → True],
    Frame → True,
    FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Exact solution\nInfinite 3D, isotropic point source,
    linearly-anisotropic scattering, fluence ϕ[r], c = " <>
   ToString[c] <> ", Σt = " <> ToString[1/mfp] <> ", b = " <> ToString[b]]
```

Exact solution

Infinite 3D, isotropic point source, linearly–anisotropic scattering, fluence ϕ[r], c = 0.9, Σ_t = 1, b = (

Out[2732]=



Fluence - Diffusion Approximations

In[∘]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] :→ (c = #;) & /@ cs], Dynamic[c]},
 {ActionMenu["Set mfp", "mfp = " <> ToString[#] :→ (mfp = #;) & /@ mfps],
  Dynamic[mfp]},
 {ActionMenu["Set b", "b = " <> ToString[#] :→ (b = #;) & /@ bs], Dynamic[b]}}
```

Out[∘]=
```
{{Set c, 0.9}, {Set mfp, 1}, {Set b, 0.7}}
```

In[2791]:=
```
data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp && #[[3]] == b &][[4]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
pointsFluence = ppoints[data[[6]], dr, maxr];
plotϕshallow = Quiet[Show[
    ListPlot[pointsFluence[[1 ;; 60]],
     PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[{
      4 Pi r^2 ϕGrosjean[r, 1 / mfp, c, b],
      4 Pi r^2 ϕDiffusion[r, 1 / mfp, c, b],
      4 Pi r^2 ϕrigourousDiffusion[r, 1 / mfp, c, b]
     }
     , {r, 0, maxr}, PlotRange → All, PlotStyle → {Red, Dashed, DotDashed},
     PlotLegends → Placed[{"Grosjean", "P1", "Rigorous Diffusion"}, {0.5, .2}]],
    Frame → True,
    FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[pointsFluence, PlotRange → All, PlotStyle → PointSize[.01]],
    LogPlot[{
      4 Pi r^2 ϕGrosjean[r, 1 / mfp, c, b],
      4 Pi r^2 ϕDiffusion[r, 1 / mfp, c, b],
      4 Pi r^2 ϕrigourousDiffusion[r, 1 / mfp, c, b]
     }
     , {r, 0, maxr}, PlotRange → All, PlotStyle → {Red, Dashed, DotDashed},
     PlotLegends → Placed[{"Grosjean", "P1", "Rigorous Diffusion"}, {0.3, .2}]],
    Frame → True,
    FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800], PlotLabel ->
  "Diffusion Approximations vs MC\nInfinite 3D, isotropic point source,
    linearly-anisotropic scattering, fluence ϕ[r], c = " <>
   ToString[c] <> ", Σt = " <> ToString[1 / mfp] <> ", b = " <> ToString[b]]
```

Diffusion Approximations vs MC
Infinite 3D, isotropic point source, linearly−anisotropic scattering, fluence ϕ[r], c = 0.9, Σ_t = 1, b = (

Out[2797]=



N-th collided Fluence - Approximations

In[3168]:=

```
{{ActionMenu["Set c", "c = " <> ToString[#] :→ (c = #;) & /@ cs], Dynamic[c]},
 {ActionMenu["Set mfp", "mfp = " <> ToString[#] :→ (mfp = #;) & /@ mfps],
  Dynamic[mfp]},
 {ActionMenu["Set collision order",
   "collisionOrder = " <> ToString[#] :→ (collisionOrder = #;) & /@
    Range[0, numcollorders - 1]], Dynamic[collisionOrder]},
 {ActionMenu["Set b", "b = " <> ToString[#] :→ (b = #;) & /@ bs], Dynamic[b]}}
```

Out[3168]=

```
{{Set c, 0.9}, {Set mfp, 1}, {Set collision order, 2}, {Set b, 0.7}}
```

```
In[3189]:= data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp && #[[3]] == b &][[4]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
fluencei = 3 numcollorders + 15 + collisionOrder;

pointsFluence = ppoints[data[[fluencei]], dr, maxr];
seriesclassical = c^collisionOrder
   SeriesCoefficient[ϕDiffusion[r, 1/mfp, C, b], {C, 0, collisionOrder}];
seriesG = c^collisionOrder SeriesCoefficient[
    ϕGrosjean[r, 1/mfp, C, b], {C, 0, collisionOrder}];
plotϕshallow = Quiet[Show[
     ListPlot[pointsFluence[[1 ;; 60]],
      PlotRange → All, PlotStyle → PointSize[.01]],
     Plot[{4 Pi r^2 seriesG, 4 Pi r^2 seriesclassical}, {r, 0, maxr},
      PlotRange → All, PlotStyle → {Red, Dashed},
      PlotLegends → Placed[{"Grosjean", "P1"}, {0.5, .2}]],
     Frame → True,
     FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
    ]];
logplotϕ = Quiet[Show[
     ListLogPlot[pointsFluence, PlotRange → All, PlotStyle → PointSize[.01]],
     LogPlot[{4 Pi r^2 seriesG, 4 Pi r^2 seriesclassical},
      {r, 0, maxr}, PlotRange → All, PlotStyle → {Red, Dashed},
      PlotLegends → Placed[{"Grosjean", "P1"}, {0.5, .2}]],
     Frame → True,
     FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
    ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800], PlotLabel ->
  "Diffusion Approximations\nInfinite 3D medium, isotropic point source,
     linearly-anisotropic scattering, n-th scattered fluence ϕ[r|" <>
   ToString[collisionOrder] <> "], c =" <> ToString[c] <> ", Σt = " <>
   ToString[1/mfp] <> ", b = " <> ToString[b]]
```

Diffusion Approximations
Infinite 3D medium, isotropic point source, linearly−anisotropic scattering, n−th scattered fluence ϕ[r|2], c =0.9

Out[3198]=





Compare moments of ϕ

In[•]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]},
 {ActionMenu["Set mfp", "mfp = " <> ToString[#] :> (mfp = #;) & /@ mfps],
  Dynamic[mfp]},
 {ActionMenu["Set b", "b = " <> ToString[#] :> (b = #;) & /@ bs], Dynamic[b]}}
```

Out[•]= {{ Set c , 0.9}, { Set mfp , 1}, { Set b , 0.7}}

In[2868]:=
```
data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp && #[[3]] == b &][[4]];
nummoments = data[[2, 15]];
ϕmoments = N[{data[[10]]}];
ks = Table[k, {k, 0, nummoments - 1}];
```

$$\text{analytic} = \left\{\frac{1}{1-c}\,\text{mfp},\ 0,\ \frac{-6}{(c-1)^2\,(c\,b-3)}\,\text{mfp}^3,\ 0,\ \text{mfp}^5\,\frac{24\,(4\,c-9)}{(c-1)^3\,(c\,b-3)^2}\right\};$$

```
j = Join[{ks}, {analytic}, ϕmoments];

TableForm[
 Join[{{"k", "analytic", "MC"}}, Transpose[j]]
]
```

Out[2874]//TableForm=

k	analytic	MC
0	10.	10.0102
1	0	40.0563
2	253.165	254.167
3	0	2173.52
4	23 073.2	23 352.3

Compare nth moments of C

2nd and 4th moments of the scalar collision rate density C(x) for the nth collision

In[3099]:=
$$\text{C2moment}[n\_, c\_, g\_, \text{mfp}\_] := c^{n-1}\left(n\,\left(2\,\text{mfp}^2\right) + \frac{g}{1-g}\,2\,\left(\text{mfp}^2\right)\left(n - \frac{1-g^n}{1-g}\right)\right)$$

In[3133]:=
$$\text{C4moment}[n\_, c\_, b\_, \text{mfp}\_] := \frac{c^{n-1}}{\text{mfp}}\left(\frac{1}{(-3+b)^4}\,4\times 3^{1-n}\,\text{mfp}^5\left(3^n\left(-6\,b\,(36+(33+5\,(-1+n))\,(-1+n)) + 9\,(18+5\,(-1+n))\,n + b^2\,(28+5\,(-1+n))\,(2+n)\right) + 2\,b^{1+n}\left(12\,(2+n) + b\,(-36-13\,(-1+n)+3\,b\,n)\right)\right)\right)$$

In[•]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]},
 {ActionMenu["Set mfp", "mfp = " <> ToString[#] :> (mfp = #;) & /@ mfps],
  Dynamic[mfp]},
 {ActionMenu["Set b", "b = " <> ToString[#] :> (b = #;) & /@ bs], Dynamic[b]}}
```

Out[•]= {{ Set c , 0.9}, { Set mfp , 1}, { Set b , 0.7}}

In[3162]:=
```
data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp && #[[3]] == b &][[4]];
nummoments = data[[2, 15]];
ϕmoments = data[[13 ;; 13 + numcollorders - 2]];
Show[
 ListLogPlot[#[[5]] & /@ ϕmoments, PlotRange → All],
 ListLogPlot[#[[3]] & /@ ϕmoments, PlotRange → All],
 ListLogPlot[#[[1]] & /@ ϕmoments, PlotRange → All],
 LogPlot[C2moment[n, c, b / 3, mfp], {n, 1, numcollorders}, PlotRange → All],
 LogPlot[C4moment[n, c, b, mfp], {n, 1, numcollorders}, PlotRange → All],
 LogPlot[c^(n-1), {n, 1, numcollorders}, PlotRange → All], PlotRange → All
]
```

Out[3165]=



Close namespace

```
End[]
```