Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Schlick

```
In[•]:= pSchlick[u_, k_] := 1/(4 Pi) ((1 - k^2)/(1 + k u)^2)
```

Normalization condition

```
In[•]:= Integrate[2 Pi pSchlick[u, k], {u, -1, 1}, Assumptions → -1 < k < 1]
Out[•]= 1
```

Mean-cosine

```
In[•]:= Integrate[2 Pi pSchlick[u, k] u, {u, -1, 1}, Assumptions → -1 < k < 1]
Out[•]= -(k - ArcTanh[k] + k^2 ArcTanh[k])/k^2
```

Legendre expansion coefficients

```
In[•]:= Integrate[2 Pi (2 k + 1) pSchlick[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
          {y, 0, Pi}, Assumptions → -1 < e < 1]
Out[•]= ConditionalExpression[1, e ≠ 0]

In[•]:= Integrate[2 Pi (2 k + 1) pSchlick[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
          {y, 0, Pi}, Assumptions → -1 < e < 1]
Out[•]= ConditionalExpression[-(3 (e + (-1 + e^2) ArcTanh[e]))/e^2, e ≠ 0]

In[•]:= Integrate[2 Pi (2 k + 1) pSchlick[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
          {y, 0, Pi}, Assumptions → -1 < e < 1]
Out[•]= ConditionalExpression[-(5 (-6 e + 4 e^3 - 6 (-1 + e^2) ArcTanh[e]))/(2 e^3), e ≠ 0]
```

```
In[•]:= Integrate[2 Pi (2 k + 1) pSchlick[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
         {y, 0, Pi}, Assumptions → -1 < e < 1]
```

$$\text{Out[•]= ConditionalExpression}\left[-\frac{7\left(30\,e-26\,e^3-6\left(5-6\,e^2+e^4\right)\text{ArcTanh}[e]\right)}{4\,e^4},\ e\neq 0\right]$$

sampling

```
In[•]:= cdf = Integrate[2 Pi pSchlick[u, e], {u, -1, x}, Assumptions → -1 < e < 1 && 0 < x < 1]
```

$$\text{Out[•]= }\frac{(1+e)\,(1+x)}{2+2\,e\,x}$$

```
In[•]:= Solve[cdf == k, x]
```

$$\text{Out[•]= }\left\{\left\{x\to\frac{1+e-2\,k}{-1-e+2\,e\,k}\right\}\right\}$$

```
In[•]:= With[{e = -.7},
          Show[
           Plot[2 Pi pSchlick[u, e], {u, -1, 1}],
           Histogram[Map[(1 + e - 2 #)/(-1 - e + 2 e #) &, Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"]
          ]
         ]
```

Out[•]=