Infinite 3D medium, Isotropic Point Source, Linearly-Anisotropic Scattering

Gamma-3 Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



c - single-scattering albedo
Σt - extinction coefficient

r - radial position coordinate in medium (distance from point source at origin)
u = cos θ - direction cosine
b - anisotropy parameter

Namespace

In[9469]:=
```
Begin["inf3DisopointlinanisoscatterGamma3`"]
```
Out[9469]=
```
inf3DisopointlinanisoscatterGamma3`
```

Analytical results

Collision rate density

collision rate density Cc due to correlated emission:

derivation

In[9470]:=
```
f00 = Fpc[0, 0, 1/2 Exp[-#] #^2 &];
f01 = Fpc[0, 1, 1/2 Exp[-#] #^2 &];
f11 = Fpc[1, 1, 1/2 Exp[-#] #^2 &];
o = 2;
Clear[A, b, c, r, h];
A[0] := 1; A[1] := b;
hsystem = Table[
   h[k] == 2/Pi c u F[k, 0] + c Sum[A[m] × h[m] × F[k, m], {m, 0, o - 1}], {k, 0, o - 1}];
hsystemsolve = Simplify[Solve[hsystem, Table[h[i], {i, 0, o - 1}]] /. A[1] → b /.
     F[0, 0] → f00 /. F[0, 1] → f01 /. F[1, 1] → f11 /. F[1, 0] → - f01]
```

Out[9477]=
$$\left\{\left\{h[0] \to \frac{2\,c\,u\,\left(-b\,c\,u+u^3+b\,c\,\text{ArcTan}[u]\right)}{\pi\left(b\,c\,u\,\left(-1+c-2\,u^2\right)+u^3\,\left(-c+\left(1+u^2\right)^2\right)+b\,c\,\left(-c+\left(1+u^2\right)^2\right)\,\text{ArcTan}[u]\right)},\right.\right.$$
$$\left.\left.h[1] \to \frac{2\,c\,u^5}{\pi\left(b\,c\,u\,\left(-1+c-2\,u^2\right)+u^3\,\left(-c+\left(1+u^2\right)^2\right)+b\,c\,\left(-c+\left(1+u^2\right)^2\right)\,\text{ArcTan}[u]\right)}\right\}\right\}$$

In[9478]:=
```
Clear[r];
(2 k + 1) 1/(4 Pi r c) (h[k]) j2[k, r u] /. k → 0 /. hsystemsolve // FullSimplify
```

Out[9478]=
$$\left\{\frac{u\,\left(-b\,c\,u+u^3+b\,c\,\text{ArcTan}[u]\right)\,\text{Sin}[r\,u]}{2\,\pi^2\,r\,\left(b\,c\,u\,\left(-1+c-2\,u^2\right)+u^3\,\left(-c+\left(1+u^2\right)^2\right)+b\,c\,\left(-c+\left(1+u^2\right)^2\right)\,\text{ArcTan}[u]\right)}\right\}$$

result

In[9479]:=
```
Ccexact[r_, t_, c_, b_] := NIntegrate[
   u (-b c u + u^3 + b c ArcTan[u]) Sin[r u] / (2 π^2 r (b c u (-1 + c - 2 u^2) + u^3 (-c + (1 + u^2)^2) + b c (-c + (1 + u^2)^2) ArcTan[u])),
   {u, 0, Infinity}, Method → "LevinRule"]
```

In[9480]:=
```
TraditionalForm[HoldForm[Cc[r] = Integrate[
    u (-b c u + u^3 + b c ArcTan[u]) Sin[r u] / (2 π^2 r (b c u (-1 + c - 2 u^2) + u^3 (-c + (1 + u^2)^2) + b c (-c + (1 + u^2)^2) ArcTan[u])),
    {u, 0, Infinity}]]]
```

Out[9480]//TraditionalForm=

$$C_c(r) = \int_0^\infty \frac{u\left(-b\,c\,u + u^3 + b\,c\tan^{-1}(u)\right)\sin(r\,u)}{2\,\pi^2\,r\left(b\,c\,u\left(-1+c-2\,u^2\right) + u^3\left(-c+\left(1+u^2\right)^2\right) + b\,c\left(-c+\left(1+u^2\right)^2\right)\tan^{-1}(u)\right)}\,du$$

load MC data

In[9481]:=
```
ppoints[xs_, dr_, maxx_] :=
  Table[{dr (i) - 0.5 dr, xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[9482]:=
```
ppointsu[xs_, du_, Σt_] :=
  Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[9483]:=
```
fs = FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/MCdata/
     inf3D_isotropicpoint_linanisoscatter_gamma3C*"];
```

In[9484]:=
```
index[x_] := Module[{data, c, mfp, b},
   data = Import[x, "Table"];
   mfp = data[[1, 13]];
   c = data[[2, 3]];
   b = data[[1, 16]];
   {c, mfp, b, data}];
simulations = index /@ fs;
cs = Union[#[[1]] & /@ simulations]
```

Out[9486]=
```
{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}
```

In[9487]:=
```
mfps = Union[#[[2]] & /@ simulations]
```

Out[9487]=
```
{0.3, 1}
```

In[9488]:=
```
bs = Union[#[[3]] & /@ simulations]
```

Out[9488]=
```
{-0.9, 0.7}
```

In[9489]:=
```
numcollorders = inf3Disopointlinanisoscatter`simulations[[1]][[-1]][[2, 13]];
```

Compare analytic and MC

Collision-rate density - Exact solution (1) comparison to MC

In[9490]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]},
 {ActionMenu["Set mfp", "mfp = " <> ToString[#] :> (mfp = #;) & /@ mfps],
  Dynamic[mfp]},
 {ActionMenu["Set b", "b = " <> ToString[#] :> (b = #;) & /@ bs], Dynamic[b]}}
```

Out[9490]=
```
{{Set c, 0.95}, {Set mfp, 1}, {Set b, -0.9}}
```

In[9500]:=
```
data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp && #[[3]] == b &][[4]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
MCCollisionRate = ppoints[data[[4]], dr, maxr];
exact1CRShallow = Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], 1/mfp, c, b]}] & /@
   MCCollisionRate[[1 ;; 60]];
exact1CR = Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], 1/mfp, c, b]}] & /@
   MCCollisionRate[[61 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
    ListPlot[MCCollisionRate[[1 ;; 60]],
     PlotRange → All, PlotStyle → PointSize[.01]],
    ListPlot[exact1CRShallow, PlotRange → All, Joined → True],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[MCCollisionRate, PlotRange → All, PlotStyle → PointSize[.01]],
    ListLogPlot[exact1CR, PlotRange → All, Joined → True],
    ListLogPlot[exact1CRShallow, PlotRange → All, Joined → True],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Exact solution (1)\nInfinite 3D, isotropic point source,
    linearly-anisotropic scattering, Gamma-3 random flight -
    correlated emission\nCollision-rate density Cc[r], c = " <>
   ToString[c] <> ", b = " <> ToString[b]]
```

Out[9508]=



Namespace

```
In[9509]:= End[]
Out[9509]= inf3DisopointlinanisoscatterGamma3`
```