Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Lambertian Sphere

geometrical optics far-field phase function of a white Lambertian sphere in 3D:
[Schoenberg 1929] - **doi**: 10.1007/978-3-642-90703-6_1
[Esposito and Lumme 1977, Blinn 1982, Porco et al. 2008]

In[∘]:=
```
pLambertSphere[u_] := (2 (Sqrt[1 - u^2] - u ArcCos[u]))/(3 π^2)
```

$$\text{pLambertSphere[u\_]} := \frac{2\left(\sqrt{1-u^2} - u\,\text{ArcCos[u]}\right)}{3\,\pi^2}$$

MC testing

Normalization condition

In[∘]:=
```
Integrate[2 Pi pLambertSphere[u], {u, -1, 1}]
```

Out[∘]= 1

forward scattering probability

In[∘]:=
```
Clear[u]; Integrate[2 Pi pLambertSphere[u], {u, 0, 1}]
```

Out[∘]= $\frac{1}{6}$

Mean cosine (g)

In[∘]:=
```
Integrate[2 Pi pLambertSphere[u] u, {u, -1, 1}]
```

Out[∘]= $-\frac{4}{9}$

Mean square cosine

In[∘]:=
```
Integrate[2 Pi pLambertSphere[u] u^2, {u, -1, 1}]
```

Out[∘]= $\frac{3}{8}$

This phase function is not particularly well approximated by Henyey Greenstein:

In[•]:=
```
Show[
 Plot[pHG[Cos[t], -4/9], {t, -Pi, Pi}, PlotRange → {0, 1}],
 Plot[pLambertSphere[Cos[t]], {t, -Pi, Pi}, PlotRange → All]
]
```

Out[•]=


Legendre expansion coefficients

In[•]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
 {y, 0, Pi}]
```

Out[•]= 1

In[•]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
 {y, 0, Pi}]
```

Out[•]= $-\frac{4}{3}$

In[•]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
 {y, 0, Pi}]
```

Out[•]= $\frac{5}{16}$

In[•]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
 {y, 0, Pi}]
```

Out[•]= 0

In[•]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 4,
 {y, 0, Pi}]
```

Out[•]= $\frac{1}{64}$

In[•]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 6,
 {y, 0, Pi}]
```

Out[•]= $\frac{13}{4096}$

In[∘]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 8,
  {y, 0, Pi}]
```

Out[∘]= $\frac{17}{16384}$

In[∘]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 10,
  {y, 0, Pi}]
```

Out[∘]= $\frac{343}{786432}$

Importance sampling:

The cosine of deflection can be sampled from:

In[∘]:=
```
Show[
 Histogram[Table[
   Sin[2 Pi RandomReal[]] √((1 - #1) (1 - #2)) - √(#1 #2) &[RandomReal[], RandomReal[]]
   , {i, Range[100 000]}], 50, "PDF"],
 Plot[2 Pi pLambertSphere[u], {u, -1, 1}]
]
```

Out[∘]=


Approximate CDF inverse:

In[∘]:=
```
lambertSphereApproxCDFi[x_] := 1 - 2 (1 - x^(1.01938` + 0.0401885` x))^0.397225`
```

In[ ]:=
```
Show[
 Histogram[Table[
   lambertSphereApproxCDFi[RandomReal[]]
   , {i, Range[100 000]}], 50, "PDF"],
 Plot[2 Pi pLambertSphere[u], {u, -1, 1}]
]
```

Out[ ]=