# Scattering Kernels in 3D

This is code to accompany the book:

## A Hitchhiker’s Guide to Multiple Scattering



www.eugenedeon.com/hitchhikers

---

## Gaussian

[Oblow et al. 1973 - effects of highly anisotropic scattering on monoenergetic neutron transport at deep penetrations] p.15:

In[1394]:=
```
pGaussian[u_, k_] := Exp[-(u - 1)^2/k] / ((π^(3/2) Erf[2/Sqrt[k]]) / Sqrt[1/k])
```

### Normalization condition

In[1395]:=
```
Integrate[2 Pi pGaussian[u, k], {u, -1, 1}, Assumptions → -1 < k < 1]
```

Out[1395]= $1$

### Mean-cosine

In[1400]:=
```
Integrate[2 Pi pGaussian[u, k] u, {u, -1, 1}, Assumptions → k > 0]
```

Out[1400]= $1 + \frac{\left(-1 + e^{-4/k}\right) \sqrt{k}}{\sqrt{\pi}\ \mathrm{Erf}\left[\frac{2}{\sqrt{k}}\right]}$

### Legendre expansion coefficients

In[1414]:=
```
Integrate[2 Pi (2 k + 1) pGaussian[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
  {y, 0, Pi}, Assumptions → e > 0]
```

Out[1414]= $1$

In[1417]:=
```
Integrate[2 Pi (2 k + 1) pGaussian[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
   {y, 0, Pi}, Assumptions → e > 0] /. e → k
```

Out[1417]= $3 + \frac{3 \left(-1 + e^{-4/k}\right) \sqrt{k}}{\sqrt{\pi}\ \mathrm{Erf}\left[\frac{2}{\sqrt{k}}\right]}$

In[1418]:=
```
Integrate[2 Pi (2 k + 1) pGaussian[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
  {y, 0, Pi}, Assumptions → e > 0] /. e → k
```

Out[1418]= $5 + \frac{15\,k}{4} - \frac{15\sqrt{k}}{\sqrt{\pi}\ \mathrm{Erf}\left[\frac{2}{\sqrt{k}}\right]}$

In[1419]:=
```
Integrate[2 Pi (2 k + 1) pGaussian[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
  {y, 0, Pi}, Assumptions → e > 0] /. e → k
```

Out[1419]= $\frac{7}{4}\left(4 + 15\,k + \frac{2\,e^{-4/k}\sqrt{k}\,\left(2 + 5\,k - e^{4/k}\,(12 + 5\,k)\right)}{\sqrt{\pi}\ \mathrm{Erf}\left[\frac{2}{\sqrt{k}}\right]}\right)$

## sampling

In[1404]:=
```
cdf = Integrate[2 Pi pGaussian[u, k], {u, -1, x}, Assumptions → 0 < x < 1 && k > 0]
```

Out[1404]= $1 - \frac{\mathrm{Erf}\left[\frac{1-x}{\sqrt{k}}\right]}{\mathrm{Erf}\left[\frac{2}{\sqrt{k}}\right]}$

In[1405]:=
```
Solve[cdf == xi, x]
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[1405]= $\left\{\left\{x \to 1 - \sqrt{k}\ \mathrm{InverseErf}\left[-(-1 + xi)\ \mathrm{Erf}\left[\frac{2}{\sqrt{k}}\right]\right]\right\}\right\}$

In[1407]:=
```
With[{k = .7},
 Show[
  Plot[2 Pi pGaussian[u, k], {u, -1, 1}],
  Histogram[Map[1 - √k InverseErf[- (-1 + #) Erf[2/√k]] &,
    Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"]
 ]
]
```

Out[1407]=