Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Draine

Draine, B.T. (2003) '*Scattering by interstellar dust grains. 1: Optical and ultraviolet*', ApJ., 598, 1017–25.

In[10790]:=
```
pDraine[u_, g_, α_] := 1/(4 Pi) ((1 - g^2)/(1 + g^2 - 2 g u)^(3/2) (1 + α u^2)/(1 + α (1 + 2 g^2)/3))
```

$$\text{pDraine}[u\_, g\_, \alpha\_] := \frac{1}{4\,\text{Pi}} \left(\frac{1 - g^2}{\left(1 + g^2 - 2\,g\,u\right)^{3/2}} \, \frac{1 + \alpha\,u^2}{1 + \alpha\,\left(1 + 2\,g^2\right) / 3} \right)$$

Normalization condition

In[ ]:=
```
Integrate[2 Pi pDraine[u, g, a], {u, -1, 1}, Assumptions → 0 < a < 1 && -1 < g < 1]
```

Out[ ]= 1

Mean-cosine

In[ ]:=
```
Integrate[2 Pi pDraine[u, g, a] u, {u, -1, 1}, Assumptions → 0 < a < 1 && -1 < g < 1]
```

Out[ ]=
$$\frac{3}{5}\left(g + \frac{2\,(1 + a)\,g}{3 + a + 2\,a\,g^2}\right)$$

In[ ]:=
$$\frac{3}{5}\left(g + \frac{2\,(1 + a)\,g}{3 + a + 2\,a\,g^2}\right) \text{ /. } a \to 0$$

Out[ ]= g

Legendre expansion coefficients

In[ ]:=
```
Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
  {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```

Out[ ]= 1

In[ ]:=
```
Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
  {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```

Out[ ]=
$$\frac{9\,g\,\left(5 + a\,\left(3 + 2\,g^2\right)\right)}{5\,\left(3 + a + 2\,a\,g^2\right)}$$

In[•]:=
```
Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
  {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```

Out[•]=
$$\frac{14\,a + 5\,(21 + 11\,a)\,g^2 + 36\,a\,g^4}{7\,(3 + a + 2\,a\,g^2)}$$

In[•]:=
```
Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
  {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```

Out[•]=
$$\frac{g\,(54\,a + 7\,(45 + 23\,a)\,g^2 + 100\,a\,g^4)}{15\,(3 + a + 2\,a\,g^2)}$$

sampling

In[•]:=
```
cdf = Integrate[2 Pi pDraine[u, g, a],
   {u, -1, x}, Assumptions → 0 < a < 1 && -1 < g < 1 && -1 < x < 1]
```

Out[•]=
$$\Big(3\,(-1+g)\,g^2\,\big(-1-g+\sqrt{1+g^2-2\,g\,x}\big) + a\,\big(2-2\,g^6-2\,g\,x-2\,\sqrt{1+g^2-2\,g\,x}+g^3\,\sqrt{1+g^2-2\,g\,x}+g^4\,(-2+x^2) + 2\,g^5\,\big(x+\sqrt{1+g^2-2\,g\,x}\big)-g^2\,\big(-2+x^2+\sqrt{1+g^2-2\,g\,x}\big)\big)\Big)\Big/ \Big(2\,g^3\,(3+a+2\,a\,g^2)\,\sqrt{1+g^2-2\,g\,x}\Big)$$

general case

In[•]:=

sampleDraineSimplified[xi_, g_, a_] := Module[{T1, T2, T3, T4, T4a, T5, T6},

$$T1 = (-1+g^2)\,\big(4\,g^2 + a\,(1+g^2)^2\big);$$
$$T2 = -1296\,(a - a\,g^2)\,(-a + a\,g^4)\,T1;$$
$$T3 = 3\,g^2\,(1 + g\,(-1 + 2\,xi)) + a\,\big(2 + g^2 + g^3\,(1 + 2\,g^2)\,(-1 + 2\,xi)\big);$$
$$T4a = 432\,(-a + a\,g^4)^3 + T2 + 432\,(a - a\,g^2)\,T3^2;$$
$$T4 = T4a + \sqrt{-4\,(-144\,a\,g^2 + 288\,a\,g^4 - 144\,a\,g^6)^3 + (T4a)^2};$$
$$T6 = \frac{1}{(a - a\,g^2)}\left(2\,(-a + a\,g^4) + \frac{48\times 2^{1/3}\,(-a\,g^2 + 2\,a\,g^4 - a\,g^6)}{T4^{1/3}} + \frac{T4^{1/3}}{3\times 2^{1/3}}\right);$$
$$T5 = 6\,(1 + g^2) + T6;$$
$$\frac{1 + g^2 - \left(\frac{1}{2}\sqrt{6\,(1+g^2) - T6 - \frac{8\,T3}{a\,(-1+g^2)\,\sqrt{T5}}} - \frac{\sqrt{T5}}{2}\right)^2}{2\,g}$$

]

In[10999]:=
```
With[{a = 6.7, g = .46}, Show[
  Histogram[Table[sampleDraineSimplified[RandomReal[], g, a], {i, 1, 100000}],
   100, "PDF", ScalingFunctions → "Log"],
  LogPlot[2 Pi pDraine[u, g, a], {u, -1, 1}, PlotRange → All]
 ]
]
```

Out[10999]=



special case g = 1/2, a = 12 (useful for approximating Mie scattering of water spheres in air)

In[ ]:=
```
pDraine[μ, 1 / 2, 12] // FullSimplify
```

Out[ ]=

$$\frac{3\,(1 + 12\,\mu^2)}{14\,\pi\,(5 - 4\,\mu)^{3/2}}$$

Simplification of an exact CDF inverse:

In[ ]:=
```
sampleDraineFog[xi_] := Module[{T1, T2, T3},
  T1 = Sqrt[(67 + 14 xi)^4 (5239 - 102376 xi + 492072 xi^2 + 105056 xi^3 + 5488 xi^4)];
  T2 = (883297 - 8820952 xi + 2597784 xi^2 + 735392 xi^3 + 38416 xi^4 + Sqrt[7] T1)^(1/3);
  T3 = Sqrt[((3954 × 2^(2/3) × 3^(1/3) - 56280 × 2^(2/3) × 3^(1/3) xi -
        5880 × 2^(2/3) × 3^(1/3) xi^2 + 912 T2 + 2^(1/3) × 3^(2/3) T2^2) / T2)];
  1/12 (-30 - T3 + Sqrt[(1824 + (6 × 2^(2/3) × 3^(1/3) (-659 + 9380 xi + 980 xi^2)) / T2 -
        2^(1/3) × 3^(2/3) T2 + (12 (67 + 14 xi)^2) / (T3))])
 ]
```

In[ ]:=
```
Show[Histogram[Table[sampleDraineFog[RandomReal[]], {i, 1, 100 000}],
   100, "PDF", ScalingFunctions → "Log"],
  LogPlot[2 Pi pDraine[u, .5, 12.], {u, -1, 1}, PlotRange → All]

]
```

Out[ ]=