Ei NDF

This is code to accompany the book:

A Hitchhiker’s Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

notation

u = **m** . **n** = Cos[θ_m]

α = roughness

Definitions and derivations

In[4444]:=

$$\text{Ei`D}[u\_,\alpha\_] := \frac{\text{Gamma}\left[0, \frac{-1+\frac{1}{u^2}}{\alpha^2}\right]}{\pi\, u^4\, \alpha^2}\ \text{HeavisideTheta}[u]$$

In[2623]:=

$$\text{Ei`}\Lambda[u\_,\alpha\_] := \frac{2\, e^{\frac{u^2}{(-1+u^2)\,\alpha^2}} \sqrt{1-u^2}\ \alpha \left(-\alpha^2 + u^2 \left(-1+\alpha^2\right)\right) + \sqrt{\pi}\ u \left(3\,\alpha^2 + u^2 \left(2 - 3\,\alpha^2\right)\right) \text{Erfc}\left[\frac{u}{\sqrt{1-u^2}\ \alpha}\right]}{6 \sqrt{\pi}\ u \left(-1+u^2\right) \alpha^2}$$

In[1331]:=

$$\text{Ei`}\sigma[u\_,\alpha\_] := \frac{1}{6\sqrt{\pi}\left(-1+u^2\right)\alpha^2} \Bigg(\alpha \left(3\sqrt{\pi}\ u \left(-1+u^2\right) \alpha + 2\, e^{\frac{u^2}{(-1+u^2)\,\alpha^2}} \sqrt{1-u^2}\ \left(-\alpha^2 + u^2\left(-1+\alpha^2\right)\right)\right) + 3\sqrt{\pi}\ u \left(-\alpha^2 + u^2 \left(-2+\alpha^2\right)\right) \text{Erf}\left[\frac{u}{\sqrt{1-u^2}\ \alpha}\right] + 2\sqrt{\pi}\ u^2\, \text{Abs}[u] \left(1 + 2\, \text{Erf}\left[\frac{u^2\sqrt{1-u^2}}{\alpha\, \text{Abs}[u] - u^2\, \alpha\, \text{Abs}[u]}\right]\right)\Bigg)$$

In[2624]:=

$$\text{Ei`G1}[u\_, a\_] := \frac{1}{1 + \text{Ei`}\Lambda[u, a]}$$

In[2689]:=

$$\text{FullSimplify}\left[\text{Ei`}\Lambda\left[u, \frac{u}{\sqrt{1-u^2}\ x}\right], \text{Assumptions} \rightarrow 0 < u < 1\ \&\&\ x > 0\right]$$

Out[2689]=

$$\frac{e^{-x^2}\left(1+x^2\right)}{3\sqrt{\pi}\ x} - \frac{1}{6}\left(3 + 2\,x^2\right) \text{Erfc}[x]$$

derivation

```
Beckmann`D[u_, α_] := e^(-(-1+1/u^2)/α^2)/(α^2 π u^4) HeavisideTheta[u]
```

$$\text{Beckmann`D}[u\_,\alpha\_] := \frac{e^{-\frac{-1+\frac{1}{u^2}}{\alpha^2}}}{\alpha^2\,\pi\,u^4}\ \text{HeavisideTheta}[u]$$

In[1304]:= Integrate[Beckmann`D[u, α Sqrt[m]], {m, 0, 1}, Assumptions → 0 < u < 1 && α > 0]

Out[1304]= $$\frac{\text{Gamma}\left[0, \frac{-1+\frac{1}{u^2}}{\alpha^2}\right]}{\pi\, u^4\, \alpha^2}$$

In[1326]:= Integrate[Beckmann`σ[u, α Sqrt[m]], {m, 0, 1}, Assumptions → -1 < u < 1 && α > 0]

Out[1326]= $$\frac{1}{6\sqrt{\pi}\left(-1+u^2\right)\alpha^2}\left(\alpha\left(3\sqrt{\pi}\,u\left(-1+u^2\right)\alpha+2\,e^{\frac{u^2}{\left(-1+u^2\right)\alpha^2}}\sqrt{1-u^2}\left(-\alpha^2+u^2\left(-1+\alpha^2\right)\right)\right)+ 3\sqrt{\pi}\,u\left(-\alpha^2+u^2\left(-2+\alpha^2\right)\right)\text{Erf}\left[\frac{u}{\sqrt{1-u^2}\,\alpha}\right]+ 2\sqrt{\pi}\,u^2\,\text{Abs}[u]\left(1+2\,\text{Erf}\left[\frac{u^2\sqrt{1-u^2}}{\alpha\,\text{Abs}[u]-u^2\,\alpha\,\text{Abs}[u]}\right]\right)\right)$$

In[1308]:= Integrate[Beckmann`Λ[u, α Sqrt[m]], {m, 0, 1}, Assumptions → 0 < u < 1 && α > 0]

Out[1308]= $$\frac{2\,e^{\frac{u^2}{\left(-1+u^2\right)\alpha^2}}\sqrt{1-u^2}\,\alpha\left(-\alpha^2+u^2\left(-1+\alpha^2\right)\right)+\sqrt{\pi}\,u\left(3\alpha^2+u^2\left(2-3\alpha^2\right)\right)\text{Erfc}\left[\frac{u}{\sqrt{1-u^2}\,\alpha}\right]}{6\sqrt{\pi}\,u\left(-1+u^2\right)\alpha^2}$$

shape invariant f(x)

In[1314]:= $$\text{FullSimplify}\left[\text{Ei`D}[u,\alpha]\,u^4\,\alpha^2\ /.\ u \to \frac{1}{\sqrt{1+x^2\alpha^2}},\ \text{Assumptions} \to 1-\frac{1}{\sqrt{1+x^2\alpha^2}} > 0\ \&\&\ x > 0\ \&\&\ \alpha > 0\right]$$

Out[1314]= $$\frac{\text{Gamma}\left[0, x^2\right]}{\pi}$$

height field normalization

In[1315]:= Integrate[2 Pi u Ei`D[u, α], {u, 0, 1}, Assumptions → 0 < α < 1]

Out[1315]= 1

distribution of slopes

In[1316]:=
```
FullSimplify[Ei`D[1/Sqrt[p^2 + q^2 + 1], α] (1/Sqrt[p^2 + q^2 + 1])^4,
  Assumptions → 0 < α < 1 && p > 0 && q > 0]
```

Out[1316]= $\frac{\text{Gamma}\left[0, \frac{p^2+q^2}{\alpha^2}\right]}{\pi\,\alpha^2}$

In[1318]:=
```
Ei`P22[p_, q_, α_] := Gamma[0, (p^2+q^2)/α^2]/(π α^2)
```

In[1319]:=
```
Integrate[Ei`P22[p, q, α], {p, -Infinity, Infinity},
  {q, -Infinity, Infinity}, Assumptions → 0 < α < 1]
```

Out[1319]= 1

compare σ to delta integral:

In[1322]:=
```
Delta`σ[u_, ui_] := Re[2 (Sqrt[1 - u^2 - ui^2] + u ui ArcCos[-(u ui)/(Sqrt[1 - u^2] Sqrt[1 - ui^2])])]
```

In[1332]:=
```
With[{α = .7},
  Plot[{
     Quiet[NIntegrate[Ei`D[ui, α] × Delta`σ[u, ui], {ui, 0, 1}]],
     Quiet[Ei`σ[u, α]]
    }, {u, -1, 1}]
]
```

Out[1332]=



importance sampling

In[4500]:=
```
With[{α = 1.7},
 Show[
  Histogram[Table[1/(√(1 - (α √RandomReal[])^2 (Log[RandomReal[]]))),
    {i, Range[10 000]}], 200, "PDF"],
  Plot[Ei`D[u, α] 2 Pi u, {u, 0, 1}, PlotRange → All]
 ]
]
```

General: Exp[−8.29137 × 10^8] is too small to represent as a normalized machine number; precision may be lost.

Out[4500]=