Infinite 3D medium, Isotropic Point Source, Rayleigh Scattering

Exponential Random Flight

This is code to accompany the book:

A Hitchhiker’s Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[3200]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



c - single-scattering albedo

Σt - extinction coefficient

r - radial position coordinate in medium (distance from point source at origin)

u = cos θ - direction cosine

Namespace

In[6395]:=
```
Begin["inf3DisopointRayleighscatter`"]
```
Out[6395]=
```
inf3DisopointRayleighscatter`
```

Util

In[∘]:=
```
SA[d_, r_] := d Pi^(d/2)/Gamma[d/2 + 1] r^(d-1)
```

$$\text{SA}[d\_, r\_] := d \frac{\text{Pi}^{d/2}}{\text{Gamma}\left[\frac{d}{2}+1\right]} r^{d-1}$$

Diffusion modes

In[∘]:=

$$\text{diffusionMode}[v\_, d\_, r\_] := (2\pi)^{-d/2} r^{1-\frac{d}{2}} v^{-1-\frac{d}{2}} \text{BesselK}\left[\frac{1}{2}(-2+d), \frac{r}{v}\right]$$

Analytical solutions

Fluence: exact solution

[Grosjean 1963 - A New Approximate One-Velocity Theory for Treating both Isotropic and Anisotropic Multiple Scattering Problems, p. 37]

In[6408]:=

$$\phi\text{exact}[r\_, \Sigma t\_, c\_] := \frac{\text{Exp}[-r\,\Sigma t]}{4\,\text{Pi}\,r^2} + \frac{c\,\Sigma t}{2\,\text{Pi}^2\,r} \text{NIntegrate}\Big[u \left(\frac{9u^2 - 3u\left(6 - 3c + 2u^2\right)\text{ArcTan}[u] + \left(9 + 6u^2 + 9u^4 - 3c\left(3 + u^2\right)\right)\text{ArcTan}[u]^2}{u\left(-9(-1+c)\,c\,u + 3c\,u^3 + 8u^5 + 3c\left(3(-1+c) + (-2+c)\,u^2 - 3u^4\right)\text{ArcTan}[u]\right)}\right) \text{Sin}[r\,\Sigma t\,u], \{u, 0, \text{Infinity}\}, \text{Method} \to \text{"LevinRule"}\Big]$$

In[4570]:=

$$\phi\text{exact}[r\_, \Sigma t\_, c\_, 1] := \frac{c\,\Sigma t}{2\,\text{Pi}^2\,r} \text{NIntegrate}\Big[u\left(\frac{9u^2 - 6u\left(3+u^2\right)\text{ArcTan}[u] + \left(9 + 6u^2 + 9u^4\right)\text{ArcTan}[u]^2}{8u^6}\right)\text{Sin}[r\,\Sigma t\,u], \{u, 0, \text{Infinity}\}, \text{Method} \to \text{"LevinRule"}\Big]$$

In[4571]:=

$$\phi\text{exact}[r\_, \Sigma t\_, c\_, 2] := \frac{c\,\Sigma t}{2\,\text{Pi}^2\,r} \text{NIntegrate}\Big[u\left(\frac{1}{64u^{11}} 3\left(-9u^3\left(3+u^2\right) + \left(81u^2 + 54u^4 + 57u^6\right)\text{ArcTan}[u] - u\left(81 + 81u^2 + 123u^4 + 35u^6\right)\text{ArcTan}[u]^2 + 3\left(3 + 2u^2 + 3u^4\right)^2\text{ArcTan}[u]^3\right)c\right) \text{Sin}[r\,\Sigma t\,u], \{u, 0, \text{Infinity}\}, \text{Method} \to \text{"LevinRule"}\Big]$$

In[4572]:=
```
ϕexact[r_, Σt_, c_, 3] := (c Σt)/(2 Pi^2 r) NIntegrate[
  u (1/(512 u^16) 9 (27 u^4 (3 + 2 u^2 + 3 u^4) - 12 u^3 (27 + 27 u^2 + 45 u^4 + 13 u^6) ArcTan[u] +
      6 u^2 (81 + 108 u^2 + 198 u^4 + 108 u^6 + 49 u^8) ArcTan[u]^2 - 4 u (81 + 135 u^2 + 270 u^4 +
         210 u^6 + 161 u^8 + 39 u^10) ArcTan[u]^3 + 3 (3 + 2 u^2 + 3 u^4)^3 ArcTan[u]^4) c^2)
    Sin[r Σt u], {u, 0, Infinity}, Method → "LevinRule"]
```

Rigorous asymptotic diffusion

In[6374]:=
```
Clear[c, b, g, x, v];
grosjeanΔ = buildΔ4[3] /. A[0] → 1 /. A[1] → 0 /. A[2] → 1/2;
dgrosjeanΔ = D[grosjeanΔ /. u → I v, c];
xpsolve = Solve[D[grosjeanΔ == 0 /. u → I x[c], c], x'[c]][[1, 1, -1]];
grosjeang = buildg4[3] /. A[0] → 1 /. A[1] → 0 /. A[2] → 1/2 /. u → I v;
```

In[6384]:=
```
grosjeang (xpsolve x[c] 2 /. x[c] → v)/dgrosjeanΔ /.
  Solve[grosjeanΔ == 0 /. u → I v, ArcTanh[v]] // FullSimplify
```

Out[6384]=
```
{-(32 v^6 (-1 + v^2))/(9 c ((-10 + c) (-1 + c)^2 + 2 (-1 + c) (-7 + 4 c) v^2 + (-6 + 5 c) v^4 + 2 v^6))}
```

In[6387]:=
```
grosjeanΔ /. u → I V
```

Out[6387]=
```
1 + (9 c)/(8 V^4) - (9 c^2)/(8 V^4) - (3 c)/(8 V^2) - (9 c ArcTanh[V])/(8 V^5) + (9 c^2 ArcTanh[V])/(8 V^5) +
  (3 c ArcTanh[V])/(4 V^3) - (3 c^2 ArcTanh[V])/(8 V^3) - (9 c ArcTanh[V])/(8 V)
```

In[6396]:=
```
rayleighv0inv[c_] :=
  ReplaceAll[Abs[V], FindRoot[1 + (9 c)/(8 V^4) - (9 c^2)/(8 V^4) - (3 c)/(8 V^2) - (9 c ArcTanh[V])/(8 V^5) +
     (9 c^2 ArcTanh[V])/(8 V^5) + (3 c ArcTanh[V])/(4 V^3) - (3 c^2 ArcTanh[V])/(8 V^3) - (9 c ArcTanh[V])/(8 V), {V, 0.8}]];
```

In[6492]:=
```
ϕrigourousDiffusion[r_, Σt_, c_] :=
  Σt/(4 Pi r) (-(32 #1^6 (-1 + #1^2))/(9 c ((-10 + c) (-1 + c)^2 + 2 (-1 + c) (-7 + 4 c) #1^2 + (-6 + 5 c) #1^4 + 2 #1^6)))
    Exp[-# r Σt] &[rayleighv0inv[c]]
```

load MC data

In[6398]:=
```
ppoints[xs_, dr_, maxx_] :=
  Table[{dr (i) - 0.5 dr, xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[6399]:=
```
ppointsu[xs_, du_, Σt_] :=
  Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[6400]:=
```
fs = FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/MCdata/
     inf3D_isotropicpoint_rayleighscatter*"];
```

In[6401]:=
```
index[x_] := Module[{data, α, Σt},
   data = Import[x, "Table"];
   Σt = data[[1, 13]];
   α = data[[2, 3]];
   {α, Σt, data}];
simulations = index /@ fs;
cs = Union[#[[1]] & /@ simulations]
```

Out[6403]= {0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}

In[6404]:=
```
mfps = Union[#[[2]] & /@ simulations]
```

Out[6404]= {0.3, 1}

In[6405]:=
```
numcollorders = simulations[[1]][[3]][[2, 13]];
maxr = simulations[[1]][[3]][[2, 5]];
dr = simulations[[1]][[3]][[2, 7]];
numr = Floor[maxr / dr];
```

Compare MC and deterministic

Mean Track Length

In[4945]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]},
  {ActionMenu["Set mfp", "mfp = " <> ToString[#] :> (mfp = #;) & /@ mfps],
   Dynamic[mfp]}}
```

Out[4945]= {{ Set c , 0.95}, { Set mfp , 0.3}}

In[4946]:=
```
data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp &][[-1]];
meanTL = data[[-1]]
```
$$\frac{mfp}{1 - c}$$

Out[4947]= {Mean, track, length:, 5.00103}

Out[4948]= 5.

Fluence - Exact solution comparison to MC

In[4949]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]},
  {ActionMenu["Set mfp", "mfp = " <> ToString[#] :> (mfp = #;) & /@ mfps],
   Dynamic[mfp]}}
```

Out[4949]= {{ Set c , 0.95}, { Set mfp , 0.3}}

In[6583]:=
```
data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp &][[-1]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
pointsFluence = ppoints[data[[6]], dr, maxr];
exact1FluenceShallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 ϕexact[#[[1]], 1/mfp, c]}] & /@
   pointsFluence[[1 ;; 60]];
exact1Fluence = Quiet[{#[[1]], 4 Pi #[[1]]^2 ϕexact[#[[1]], 1/mfp, c]}] & /@
   pointsFluence[[1 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
    ListPlot[pointsFluence[[1 ;; 60]],
     PlotRange → All, PlotStyle → PointSize[.01]],
    ListPlot[exact1FluenceShallow, PlotRange → All, Joined → True],
    Plot[4 Pi r^2 ϕrigourousDiffusion[r, 1/mfp, c],
     {r, 0, maxr}, PlotStyle → Dashed, PlotRange → All],
    Frame → True,
    FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[pointsFluence, PlotRange → All, PlotStyle → PointSize[.01]],
    ListLogPlot[exact1Fluence, PlotRange → All, Joined → True],
    LogPlot[4 Pi r^2 ϕrigourousDiffusion[r, 1/mfp, c],
     {r, 0, maxr}, PlotStyle → Dashed],
    Frame → True,
    FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Exact solution (continuous) and Rigorous Asymptotic Diffusion
     (dashed)\nInfinite 3D, isotropic point source, Rayleigh scattering,
     fluence ϕ[r], c = " <> ToString[c] <> ", Σt = " <> ToString[1/mfp]]
```

Exact solution (continuous) and Rigorous Asymptotic Diffusion (dashed)
Infinite 3D, isotropic point source, Rayleigh scattering, fluence ϕ[r], c = 0.9, Σ_t = 3.33333

Out[6591]=





N-th collided Fluence - Exact solution comparison to MC

```
In[4501]:= {{ActionMenu["Set c", "c = " <> ToString[#] :→ (c = #;) & /@ cs], Dynamic[c]},
  {ActionMenu["Set mfp", "mfp = " <> ToString[#] :→ (mfp = #;) & /@ mfps],
   Dynamic[mfp]}}

Out[4501]= {{Set c, 0.95}, {Set mfp, 0.3}}

In[4573]:= data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp &][[3]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
fluencei1 = 3 numcollorders + 15 + 1;
fluencei2 = 3 numcollorders + 15 + 2;
fluencei3 = 3 numcollorders + 15 + 3;

pointsFluence1 = ppoints[data[[fluencei1]], dr, maxr];
pointsFluence2 = ppoints[data[[fluencei2]], dr, maxr];
pointsFluence3 = ppoints[data[[fluencei3]], dr, maxr];
exact1Fluence1Shallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 ϕexact[#[[1]], 1/mfp, c, 1]}] & /@
   pointsFluence1[[1 ;; 60]];
exact1Fluence1 = Quiet[{#[[1]], 4 Pi #[[1]]^2 ϕexact[#[[1]], 1/mfp, c, 1]}] & /@
   pointsFluence[[61 ;; -1 ;; 10]];
exact1Fluence2Shallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 ϕexact[#[[1]], 1/mfp, c, 2]}] & /@
   pointsFluence1[[1 ;; 60]];
exact1Fluence2 = Quiet[{#[[1]], 4 Pi #[[1]]^2 ϕexact[#[[1]], 1/mfp, c, 2]}] & /@
   pointsFluence[[61 ;; -1 ;; 10]];
exact1Fluence3Shallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 ϕexact[#[[1]], 1/mfp, c, 3]}] & /@
   pointsFluence1[[1 ;; 60]];
exact1Fluence3 = Quiet[{#[[1]], 4 Pi #[[1]]^2 ϕexact[#[[1]], 1/mfp, c, 3]}] & /@
   pointsFluence[[61 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
    ListPlot[{pointsFluence1[[1 ;; 60]], pointsFluence2[[1 ;; 60]],
      pointsFluence3[[1 ;; 60]]}, PlotRange → All, PlotStyle → PointSize[.01]],
    ListPlot[{exact1Fluence1Shallow, exact1Fluence2Shallow,
      exact1Fluence3Shallow}, PlotRange → All, Joined → True],
    Frame → True,
    FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[{pointsFluence1, pointsFluence2, pointsFluence3},
     PlotRange → All, PlotStyle → PointSize[.01]],
    ListLogPlot[{exact1Fluence1Shallow, exact1Fluence2Shallow,
      exact1Fluence3Shallow}, PlotRange → All, Joined → True],
```

```
    ListLogPlot[{exact1Fluence1, exact1Fluence2, exact1Fluence3},
     PlotRange → All, Joined → True],
    Frame → True,
    FrameLabel -> {{4 Pi r^2 ϕ[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Exact solution (Fourier)\nInfinite 3D medium, isotropic point
     source, Rayleigh scattering, n-th scattered fluence ϕ[r|" <>
   ToString[collisionOrder] <> "], c =" <> ToString[c] <>
   ", Σt = " <> ToString[1 / mfp]]
```

Out[4590]=

Exact solution (Fourier)
Infinite 3D medium, isotropic point source, Rayleigh scattering, n–th scattered fluence ϕ[r|1], c =0.95, Σ_t





Compare moments of ϕ

In[4361]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] ⧴ (c = #;) & /@ cs], Dynamic[c]},
 {ActionMenu["Set mfp", "mfp = " <> ToString[#] ⧴ (mfp = #;) & /@ mfps],
  Dynamic[mfp]}}
```

Out[4361]= {{ Set c , 0.95}, { Set mfp , 0.3}}

mfp 1

In[4362]:=
```
mfp = 1;
sims1 = Select[simulations, #[[2]] == mfp &];
```

In[4364]:=
```
Show[
 ListLogPlot[{
   {#[[-1, 2, 3]], #[[-1, 10, 1]]} & /@ sims1,
   {#[[-1, 2, 3]], #[[-1, 10, 3]]} & /@ sims1,
   {#[[-1, 2, 3]], #[[-1, 10, 5]]} & /@ sims1,
   {#[[-1, 2, 3]], #[[-1, 10, 7]]} & /@ sims1,
   {#[[-1, 2, 3]], #[[-1, 10, 9]]} & /@ sims1
  }],
```

$$\text{LogPlot}\left[\left\{\frac{\text{mfp}}{1-c}, \frac{2\,\text{mfp}^3}{(-1+c)^2}, -\frac{120\,(-2+c)\,\text{mfp}^5}{(-10+c)\,(-1+c)^3}, \frac{720\,(100+c\,(-116+37\,c))\,\text{mfp}^7}{(-10+c)^2\,(-1+c)^4}, \frac{5760\,(49\,000+c\,(-93\,580+c\,(64\,230+c\,(-15\,937+256\,c))))\,\text{mfp}^9}{7\,(-10+c)^3\,(-1+c)^5}\right\},\right.$$

```
  {c, 0, .999}, PlotRange → All]
]
```

Out[4364]=



mfp 0.3

In[4365]:=
```
mfp = 0.3;
sims1 = Select[simulations, #[[2]] == mfp &];
```

In[4367]:=
```
Show[
 ListLogPlot[{
   {#[[-1, 2, 3]], #[[-1, 10, 1]]} & /@ sims1,
   {#[[-1, 2, 3]], #[[-1, 10, 3]]} & /@ sims1,
   {#[[-1, 2, 3]], #[[-1, 10, 5]]} & /@ sims1,
   {#[[-1, 2, 3]], #[[-1, 10, 7]]} & /@ sims1,
   {#[[-1, 2, 3]], #[[-1, 10, 9]]} & /@ sims1
  }],
 LogPlot[{mfp/(1 - c), 2 mfp^3/(-1 + c)^2, -120 (-2 + c) mfp^5/((-10 + c) (-1 + c)^3), 720 (100 + c (-116 + 37 c)) mfp^7/((-10 + c)^2 (-1 + c)^4),
   5760 (49 000 + c (-93 580 + c (64 230 + c (-15 937 + 256 c)))) mfp^9/(7 (-10 + c)^3 (-1 + c)^5)},
  {c, 0, .999}, PlotRange → All]
]
```

Out[4367]=