Infinite Flatland medium, Isotropic Line Source, Isotropic Scattering

Exponential Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[2222]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation

α - single-scattering albedo
Σt - extinction coefficient
x - scalar position coordinate in medium (distance from line source at origin)
u = cos θ - direction cosine

Analytic solutions

Generalized Caseology quantities for Flatland [d'Eon 2017]

In[2187]:= flatland\`CaseN0[c_, v_] := $\frac{c^2 v^2}{2 \pi \left(-1+v^2\right)^{3/2}}$

In[2188]:= flatland\`Casev0[c_] := $\frac{1}{\sqrt{1-c^2}}$

In[2189]:=
```
FullSimplify[flatland`CaseN0[c, flatland`Casev0[c]], Assumptions → 0 < c < 1]
```

Out[2189]= $\frac{\sqrt{1-c^2}}{2 c \pi}$

In[2190]:= flatland\`Caseψ0[u_, v0_, c_, z_] := $\frac{c}{2 \text{Pi}} \frac{v0}{v0 - \text{Sign}[z]\, u}$

In[2191]:= flatland`CaseN[c_, v_] := $v\left(\text{flatland`Case}\lambda[v, c]^2 + \left(\frac{c\,v}{2}\right)^2\right)$

In[2193]:= flatland`Caseλ[v_, c_] := $\frac{\sqrt{1-v^2}}{2}$

Fluence - Rigorous Diffusion Approximation

In[2319]:= infFlatlandisolineisoscatter`ϕrigourousDiffusion[x_, Σt_, α_] :=
$\frac{1}{2\,\text{Pi}}\ \frac{E^{-\text{Abs}[x]\ \Sigma t/\#}}{\text{flatland`CaseN0}[\alpha, \#]}$ &[flatland`Casev0[α]]

In[2320]:= FullSimplify[infFlatlandisolineisoscatter`ϕrigourousDiffusion[x, Σt, α],
Assumptions → 0 < α < 1 && Σt > 0]

Out[2320]= $\frac{e^{-\sqrt{1-\alpha^2}\ \Sigma t\ \text{Abs}[x]}\ \alpha}{\sqrt{1-\alpha^2}}$

Fluence - Exact solution 1 - Caseology Generalization

In[2321]:= infFlatlandisolineisoscatter`ϕexact1[x_, Σt_, α_] :=
infFlatlandisolineisoscatter`ϕrigourousDiffusion[x, Σt, α] +
$\frac{1}{2\,\text{Pi}}$ NIntegrate$\left[\frac{e^{-\Sigma t\ \text{Abs}[x]\ /v}}{\text{flatland`CaseN}[\alpha, v]}\ \frac{\sqrt{1-v^2}}{2}, \{v, 0, 1\}\right]$

Fluence - Exact solution 2 - Fourier Transform

In[2302]:= infFlatlandisolineisoscatter`ϕexact2[x_, Σt_, c_] :=
NIntegrate$\left[\left(z\left(c\,\pi + 2\sqrt{1-c^2+z^2} + 2\,c\,\text{ArcSin}\left[\frac{c}{\sqrt{1+z^2}}\right]\right)\text{BesselJ}[0, z\,\text{Abs}[x]]\right)\Big/\right.$
$\left.\left(2\,\pi\left(1-c^2+z^2\right)^{3/2}\right), \{z, 0, \text{Infinity}\}\right]$

In[2307]:= infFlatlandisolineisoscatter`ϕexact3[x_, Σt_, c_] :=
$\frac{\text{BesselK}[0, \text{Abs}[x]]}{\pi}$ + NIntegrate[
$\left(c\,z\left(\pi + \pi\,z^2 + 2\,c\sqrt{1-c^2+z^2} + 2\left(1+z^2\right)\text{ArcSin}\left[\frac{c}{\sqrt{1+z^2}}\right]\right)\text{BesselJ}[0, z\,\text{Abs}[x]]\right)\Big/$
$\left(2\,\pi\left(1+z^2\right)\left(1-c^2+z^2\right)^{3/2}\right)$, {z, 0, Infinity}]

In[2201]:= infFlatlandisolineisoscatter`ϕexact4[x_, Σt_, c_] :=
$\frac{\text{BesselK}[0, \text{Abs}[x]]}{\pi} + \frac{c\,e^{-\sqrt{1-c^2}\ \text{Abs}[x]}}{2\sqrt{1-c^2}}$ + NIntegrate$\Big[\frac{1}{\pi}$
$c\,z\left(\frac{c}{\left(1+z^2\right)\left(1-c^2+z^2\right)} + \frac{\text{ArcSin}\left[\frac{c}{\sqrt{1+z^2}}\right]}{\left(1-c^2+z^2\right)^{3/2}}\right)\text{BesselJ}[0, \text{Abs}[x]\,z], \{z, 0, \text{Infinity}\}\Big]$

In[2202]:= infFlatlandisolineisoscatter\`ϕexact5[x_, Σt_, c_] :=

$$\frac{\text{BesselK}[0,\text{Abs}[x]]}{\pi}+\frac{c\,e^{-\sqrt{1-c^2}\,\text{Abs}[x]}}{2\sqrt{1-c^2}}+\text{NIntegrate}\left[\frac{c\,z\left(\frac{\text{ArcSin}\left[\frac{c}{\sqrt{1+z^2}}\right]}{(1-c^2+z^2)^{3/2}}\right)\text{BesselJ}[0,\text{Abs}[x]\,z]}{\pi},\{z,0,\text{Infinity}\}\right]+$$

$$\text{Chop}\left[\frac{1}{4\pi^2\,\text{Abs}[x]}\left(\text{Abs}[x]\,\text{MeijerG}\left[\left\{\left\{1,1,\frac{3}{2}\right\},\{\}\right\},\left\{\left\{\frac{3}{2}\right\},\{\}\right\},\frac{2\,i}{\text{Abs}[x]},\frac{1}{2}\right]+\right.\right.$$

$$\pi\left(-2\,i+i\,\pi\,\text{Abs}[x]\,\text{BesselI}[0,\text{Abs}[x]]+\right.$$

$$4\,\text{Abs}[x]\,\text{BesselK}\left[0,\sqrt{1-c^2}\,\text{Abs}[x]\right]-\text{Abs}[x]\,\text{BesselI}[0,\text{Abs}[x]]\,\text{Log}[4]+$$

$$2\,\text{Abs}[x]\,\text{BesselI}[0,\text{Abs}[x]]\,\text{Log}[\text{Abs}[x]]-\pi\,\text{Abs}[x]\,\text{StruveH}[0,i\,\text{Abs}[x]]+$$

$$\left.\left.\left.2\,\text{Abs}[x]\,\text{Hypergeometric0F1Regularized}^{(1,0)}\left[1,\frac{\text{Abs}[x]^2}{4}\right]\right)\right)\right]$$

In[2203]:= infFlatlandisolineisoscatter\`ϕexact5[x_, Σt_, c_] :=

$$\frac{\text{BesselK}[0,\text{Abs}[x]]}{\pi}+\frac{c\,e^{-\sqrt{1-c^2}\,\text{Abs}[x]}}{2\sqrt{1-c^2}}+\text{NIntegrate}\left[\frac{c\,z\left(\frac{\text{ArcSin}\left[\frac{c}{\sqrt{1+z^2}}\right]}{(1-c^2+z^2)^{3/2}}\right)\text{BesselJ}[0,\text{Abs}[x]\,z]}{\pi},\{z,0,\text{Infinity}\}\right]+$$

$$\frac{\text{BesselK}\left[0,\sqrt{1-c^2}\,\text{Abs}[x]\right]}{\pi}-\frac{\text{BesselI}[0,\text{Abs}[x]]\,\text{Log}[4]}{4\pi}+$$

$$\frac{\text{BesselI}[0,\text{Abs}[x]]\,\text{Log}[\text{Abs}[x]]}{2\pi}+\frac{\text{Hypergeometric0F1Regularized}^{(1,0)}\left[1,\frac{\text{Abs}[x]^2}{4}\right]}{2\pi}+$$

$$\text{Re}\left[\frac{\text{MeijerG}\left[\left\{\left\{1,1,\frac{3}{2}\right\},\{\}\right\},\left\{\left\{\frac{3}{2}\right\},\{\}\right\},\frac{2\,i}{\text{Abs}[x]},\frac{1}{2}\right]}{4\pi^2}\right]$$

Nth-scattered Fluence

In[2408]:= infFlatlandisolineisoscatter\`ϕ[r_, Σt_, c_, n_] :=

$$\frac{2^{-n/2}\,c^n\,\Sigma t^{\frac{n}{2}}\,\text{Abs}[r]^{n/2}\,\text{BesselK}\left[\frac{n}{2},\Sigma t\,\text{Abs}[r]\right]}{\sqrt{\pi}\,\text{Gamma}\left[\frac{1+n}{2}\right]}$$

In[2409]:= Limit[infFlatlandisolineisoscatter\`ϕ[x, 1, c, n],
x → 0, Assumptions → 0 < c < 1 && n > 0 && n ∈ Integers]

Out[2409]=
$$\frac{c^n\,\text{Gamma}\left[\frac{n}{2}\right]}{2\sqrt{\pi}\,\text{Gamma}\left[\frac{1+n}{2}\right]}$$

Moments

In[2343]:=
```
infFlatlandisolineisoscatter`ϕm[c_, Σt_, m_, n_] :=
 If[EvenQ[m], (2^m c^n Σt^(-m-1) Gamma[(1+m)/2] Gamma[1/2 (1 + m + n)])/(Sqrt[π] Gamma[(1+n)/2]), 0]
```

In[2344]:=
```
infFlatlandisolineisoscatter`ϕm[c_, Σt_, m_?IntegerQ] :=
 If[EvenQ[m], -1/π (1 - c^2)^(-1-m/2) Σt^(-m-1)
   (-c π m! - 2^m Gamma[(1 + m)/2]^2 Hypergeometric2F1[-1/2, -m/2, 1/2, c^2]), 0]
```

In[2345]:=
```
TableForm[
 FullSimplify[Table[infFlatlandisolineisoscatter`ϕm[c, 1, m], {m, 0, 6, 2}]]]
```

Out[2345]//TableForm=

$$\frac{1}{1-c}$$

$$\frac{1}{(-1+c)^2}$$

$$\frac{3\,(-3+c)}{(-1+c)^3}$$

$$\frac{45\,(5+(-4+c)\,c)}{(-1+c)^4}$$

Radiance (Angular Flux)

Generalized Caseology - Asymptotic Solution:

In[2347]:=
```
infFlatlandisolineisoscatter`LrigourousDiffusion[z_, u_, Σt_, α_] :=
 1/(2 Pi) flatland`Caseψ0[u, #, α, z] E^(-Abs[z] Σt/#)/(flatland`CaseN0[α, #]) &[flatland`Casev0[α]]
```

Generalized Caseology - Exact Solution:

```
infFlatlandisolineisoscatter`Lexact[z_, u_, Σt_, α_] :=
 infFlatlandisolineisoscatter`LrigourousDiffusion[z, u, Σt, α] +
  1/(2 Pi) (flatland`Caseλ[u, α] ⅇ^(-Abs[z] Σt/u)/(flatland`CaseN[α, u])
      HeavisideTheta[1 - u] HeavisideTheta[u] Sqrt[1 - u^2]/2
    + 1/Pi NIntegrate[ⅇ^(-Abs[z] Σt/v)/(flatland`CaseN[α, v]) α/2 v/(v - u) Sqrt[1 - v^2]/2
      , {v, 0, u, 1}, Method -> "PrincipalValue", PrecisionGoal -> 5])
```

load MC data

Compare Deterministic and MC

Fluence - Exact solution I (Caseology) comparison to MC

In[2322]:=
```
Manipulate[
 If[Length[infFlatlandisolineisoscatter`simulations] > 0,
  Module[{data, maxz, dz, pointsϕ,
    plotpointsϕ, logplotϕ, plotϕ, exact1points, numpoints, skip},
   data = SelectFirst[infFlatlandisolineisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt &][[3]];
   maxz = data[[2, 5]];
   dz = data[[2, 7]];

   pointsϕ = data[[4]];

   (* divide by Σt to convert collision density into fluence *)
   plotpointsϕ = infFlatlandisolineisoscatter`ppoints[pointsϕ, dz, maxz, Σt];

   exact1points =
    Quiet[{#[[1]], infFlatlandisolineisoscatter`ϕexact1[#[[1]], Σt, α]}] & /@
     plotpointsϕ;

   numpoints = Length[plotpointsϕ];
   skip = Floor[numpoints 6/7 1/2];

   plotϕ = Quiet[Show[
      (*ListPlot[exact1points[[skip;;-skip]],PlotRange→{0,6},Joined→True],*)
      Plot[infFlatlandisolineisoscatter`ϕexact1[z, Σt, α],
       {z, -maxz/7, maxz/7}, PlotRange → All],
      Plot[infFlatlandisolineisoscatter`ϕrigourousDiffusion[z, Σt, α],
       {z, -maxz/7, maxz/7}, PlotRange → All, PlotStyle → {Red, Dashed}],
      ListPlot[plotpointsϕ[[skip ;; -skip]], PlotRange → All,
       PlotStyle → {Black, PointSize[.01]}],
      Frame → True,
      FrameLabel -> {{ ϕ[z],}, {z,}}
     ]];
   logplotϕ = Quiet[Show[
      ListLogPlot[exact1points, PlotRange → All, Joined → True],
      LogPlot[infFlatlandisolineisoscatter`ϕrigourousDiffusion[z, Σt, α],
       {z, -maxz, maxz}, PlotRange → All, PlotStyle → {Red, Dashed}],
      ListLogPlot[plotpointsϕ[[1 ;; -1 ;; 3]], PlotRange → All,
       PlotStyle → {Black, PointSize[.01]}],
      Frame → True,
```

```
      FrameLabel -> {{ϕ[z],}, {z,}}
     ]];
   Show[GraphicsGrid[{{plotϕ, logplotϕ}}, ImageSize → 800],
    PlotLabel -> "Exact solution 1 (thin), Rigorous Diffusion
        component (dashed)\nInfinite Flatland, isotropic line
        source, isotropic scattering, fluence ϕ[z], α = "<>
      ToString[α] <> ", Σt = " <> ToString[Σt]]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
, {{α, 0.95}, infFlatlandisolineisoscatter`alphas},
{{Σt, 1}, infFlatlandisolineisoscatter`muts}]
```

Out[2322]=



N-th collided Fluence - Exact solution (1) comparison to MC

```
Manipulate[
 If[Length[infFlatlandisolineisoscatter`simulations] > 0,
  Module[{data, maxz, dz, pointsϕ,
    plotpointsϕ, logplotϕ, plotϕ, exact1points, numorders},
   data = SelectFirst[infFlatlandisolineisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt &][[3]];
   maxz = data[[2, 5]];
   dz = data[[2, 7]];
```

```
   numorders = data[[2, 13]];

   pointsϕ = data[[9 + numorders + n + 1]];

   (* divide by Σt to convert collision density into fluence *)
   plotpointsϕ = infFlatlandisolineisoscatter`ppoints[pointsϕ, dz, maxz, Σt];

   exact1points =
    Quiet[{#[[1]], infFlatlandisolineisoscatter`ϕ[#[[1]], Σt, α, n]}] & /@
     plotpointsϕ;

   plotϕ = Quiet[Show[
       ListPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
       Plot[ infFlatlandisolineisoscatter`ϕ[Abs[z], Σt, α, n],
        {z, -maxz, maxz}, PlotRange → All],
       Frame → True,
       FrameLabel -> {{ ϕ["z|" <> ToString[n]],}, {z,}}
      ]];
   logplotϕ = Quiet[Show[
       ListLogPlot[plotpointsϕ, PlotRange → All, PlotStyle → PointSize[.01]],
       ListLogPlot[exact1points, PlotRange → All, Joined → True],
       Frame → True,
       FrameLabel -> {{ ϕ["z|" <> ToString[n]],}, {z,}}
      ]];
   Show[GraphicsGrid[{{plotϕ, logplotϕ}}, ImageSize → 800], PlotLabel ->
     "Exact solution (1)\nInfinite Flatland, isotropic line source,
        isotropic scattering, n-th scattered fluence ϕ[z|n], α = " <>
       ToString[α] <> ", Σt = " <> ToString[Σt]]
  ]
  ,
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.99}, infFlatlandisolineisoscatter`alphas},
 {{Σt, 1}, infFlatlandisolineisoscatter`muts},
 {{n, 1}, Range[0, If[NumberQ[infFlatlandisolineisoscatter`numcollorders],
    infFlatlandisolineisoscatter`numcollorders, 1]]}
]
```


α 0.99
Σt 1 3
n 16
Exact solution (1)
Infinite 3D, isotropic plane source, isotropic scattering, n–th scattered fluence ϕ[z|n], α = 0.9
ϕ(z|16)
0.08
0.06
0.04
0.02
0.00
−5
0
5
z
0.100
0.010
0.001
10^−4
10^−5
−6
−4
−2

Fluence Moments

Moments of the total fluence

In[2338]:=
```
Manipulate[
 If[Length[infFlatlandisolineisoscatter`simulations] > 0,
  Module[{data, nummoments, ϕmoments, ks, analytic, j},
   data = SelectFirst[infFlatlandisolineisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt &][[3]];
   nummoments = data[[2, 15]];
   ϕmoments = N[{data[[6]] / Σt}];
   ks = {Table[k, {k, 0, nummoments - 1}]};
   analytic = {Table[
      Re[infFlatlandisolineisoscatter`ϕm[α, Σt, m]], {m, 0, nummoments - 1}]};
   j = Join[ks, analytic, ϕmoments];
   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.95}, infFlatlandisolineisoscatter`alphas},
 {{Σt, 3}, infFlatlandisolineisoscatter`muts}]
```

Out[2338]=

α 0.999

Σt 1 3

k	analytic	MC
0	1000.	999.287
1	0	-127.272
2	$1. \times 10^{6}$	1.00735×10^{6}
3	0	-1.27689×10^{6}
4	6.003×10^{9}	6.11971×10^{9}

Moments of the n-th collided fluence

In[2331]:=
```
Manipulate[
 If[Length[infFlatlandisolineisoscatter`simulations] > 0,
  Module[{data, nummoments, ϕmoments, ks, analytic, j},
   data = SelectFirst[infFlatlandisolineisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt &][[3]];
   nummoments = data[[2, 15]];
   ϕmoments = N[{data[[9 + n]]} / Σt];
   ks = Table[k, {k, 0, nummoments - 1}];
   analytic = Table[Re[infFlatlandisolineisoscatter`ϕm[α, Σt, k, n]], {k, ks}];
   j = Join[{ks}, {analytic}, ϕmoments];

   TableForm[
    Join[{{"k", "analytic", "MC"}}, Transpose[j]]
   ]
  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.95}, infFlatlandisolineisoscatter`alphas},
 {{Σt, 3}, infFlatlandisolineisoscatter`muts},
 {{n, 11}, Range[If[NumberQ[infFlatlandisolineisoscatter`numcollorders],
    infFlatlandisolineisoscatter`numcollorders, 1]]}
]
```

Out[2331]=

α 0.7

Σt 1 3

n 8

k	analytic	MC
0	0.019216	0.0192302
1	0	0.0000183559
2	0.019216	0.0192049
3	0	0.0000175097
4	0.0704587	0.069785

Angular Distributions

Exact solution (Caseology)

In[2387]:=
```
Manipulate[
 If[Length[infFlatlandisolineisoscatter`simulations] > 0,
  Module[{data, numorders, pointsu, plotpointsu, du, r, dz, maxz, zsim},
   data = SelectFirst[infFlatlandisolineisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt &][[3]];
   numorders = data[[2, 13]];
   du = data[[2, 9]];
   dz = data[[2, 7]];
   maxz = data[[2, 5]];

   pointsu = data[[9 + 2 numorders + n]];

   zsim = dz * n - 0.5 dz - maxz;

   plotpointsu = infFlatlandisolineisoscatter`ppointsu[pointsu, du, Σt];
   pp = Show[
     ListPlot[plotpointsu, PlotRange → All, PlotStyle → Black,
      Frame → True,
      FrameLabel -> {{ L[zsim, u]/Sqrt[1 - u^2],}, {u,}}],
     Plot[1/Pi (Sqrt[1 - u^2])^-1 Pi infFlatlandisolineisoscatter`LrigourousDiffusion[
        zsim, u, Σt, α], {u, -1, 1}, PlotStyle → {Red, Dashed}
     ],
     Quiet[Plot[1/Pi (Sqrt[1 - u^2])^-1 Pi
        infFlatlandisolineisoscatter`Lexact[zsim, u, Σt, α], {u, -1, 1}
      ]],
     PlotLabel -> "Infinite Flatland Medium, isotropic line
         source, isotropic scattering\nAngular distribution:
         Exact solution 1\n L[z,u]/Sqrt[1 - u^2] , z = " <>
       ToString[zsim] <> ", α = " <> ToString[α] <> ", Σ_t = " <> ToString[Σt]
    ]

  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.9}, infFlatlandisolineisoscatter`alphas},
 {{Σt, 1}, infFlatlandisolineisoscatter`muts},
 {{n, 127}, Range[If[NumberQ[infFlatlandisolineisoscatter`numz],
    infFlatlandisolineisoscatter`numz, 1]]}]
```

Out[2387]=



In[2369]:= infFlatlandisolineisoscatter`LexactFourier2[z_, u_, Σt_, c_] :=

$$\frac{1}{2\,\mathrm{Pi}}\,\frac{1}{\mathrm{Pi}}\,\mathrm{NIntegrate}\left[\frac{(\mathrm{Cos}[k\,z\,\Sigma t]+k\,u\,\mathrm{Sin}[k\,z\,\Sigma t])}{(1+k^2\,u^2)}\,\frac{1}{\left(1-c\,\sqrt{\frac{1}{1+k^2}}\right)},\ \{k,\ 0,\ \mathrm{Infinity}\}\right]$$

In[2370]:=
```
Manipulate[
 If[Length[infFlatlandisolineisoscatter`simulations] > 0,
  Module[{data, numorders, pointsu, plotpointsu, du, r, dz, maxz, zsim},
   data = SelectFirst[infFlatlandisolineisoscatter`simulations,
      #[[1]] == α && #[[2]] == Σt &][[3]];
   numorders = data[[2, 13]];
   du = data[[2, 9]];
   dz = data[[2, 7]];
   maxz = data[[2, 5]];

   pointsu = data[[9 + 2 numorders + n]];

   zsim = dz * n - 0.5 dz - maxz;

   plotpointsu = infFlatlandisolineisoscatter`ppointsu[pointsu, du, Σt];
   pp = Show[
     ListPlot[plotpointsu, PlotRange → All, PlotStyle → Black,
      Frame → True,
      FrameLabel -> {{ Pi L[zsim, u],}, {u,}}],
     Plot[1/Pi (Sqrt[1 - u^2])^-1 Pi infFlatlandisolineisoscatter`LrigourousDiffusion[
        zsim, u, Σt, α], {u, -1, 1}, PlotStyle → {Red, Dashed}
     ],
     Quiet[Plot[1/Pi (Sqrt[1 - u^2])^-1 Pi
        infFlatlandisolineisoscatter`LexactFourier2[zsim, u, Σt, α], {u, -1, 1}
      ]],
     PlotLabel -> "Infinite 3D Medium, isotropic plane
        source, isotropic scattering\nAngular distribution:
        rigorous diffusion approximation\nπ L[z,u], z = " <>
      ToString[zsim] <> ", α = " <> ToString[α] <> ", Σt = " <> ToString[Σt]
    ]

  ],
  Text[
   "Uh oh!  Couldn't find MC data.  Try to evaluate this entire notebook and
     ensure the data path is setup correctly."]
 ]
 , {{α, 0.9}, infFlatlandisolineisoscatter`alphas},
 {{Σt, 1}, infFlatlandisolineisoscatter`muts},
 {{n, 127}, Range[If[NumberQ[infFlatlandisolineisoscatter`numz],
    infFlatlandisolineisoscatter`numz, 1]]}]
```

Out[2370]=


α 0.5
Σt 1 3
n 128
Infinite 3D Medium, isotropic plane source, isotropic scattering
Angular distribution: rigorous diffusion approximation
π L[z,u], z = 0.5, α = 0.5, Σt = 1
π L(0.5, u)
0.0
0.2
0.4
0.6
0.8
−1.0
−0.5
0.0
0.5
1.0
u