Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Anli-Gungor phase function

In[289]:=
```
Clear[pAG]; pAG[u_, t_] := 1/(4 Pi) 1/Sqrt[1 + t^2 - 2 t u]
```

$$\text{pAG}[u\_, t\_] := \frac{1}{4\,\text{Pi}}\frac{1}{\sqrt{1+t^2-2\,t\,u}}$$

In[291]:=
```
pHGplot = Show[
  Plot[pAG[Cos[t], .8], {t, -Pi, Pi}, PlotRange → {0, .4}],
  Plot[pAG[Cos[t], .6], {t, -Pi, Pi}, PlotRange → All],
  Plot[pAG[Cos[t], .5], {t, -Pi, Pi}, PlotRange → All],
  Plot[pAG[Cos[t], .4], {t, -Pi, Pi}, PlotRange → All],
  Plot[pAG[Cos[t], .3], {t, -Pi, Pi}, PlotRange → All],
  Frame → True,
  ImageSize → 400,
  FrameLabel →
   {{p[Cos[θ]],}, {θ, "Anli-Gungor Scattering, t = 0.3, 0.4, 0.5, 0.6, 0.8"}}]
```

Out[291]=



In[298]:=
```
Show[
 ParametricPlot[{Sin[t], Cos[t]} (1),
  {t, -Pi, Pi}, PlotRange → All, PlotStyle → Dashed],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pAG[Cos[t], 0.95]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pAG[Cos[t], 0.9]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pAG[Cos[t], 0.8]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[{Sin[t], Cos[t]} (1 + pAG[Cos[t], 0.7]),
  {t, -Pi, Pi}, PlotRange → All],
 ParametricPlot[
  {Sin[t], Cos[t]} (1 + pAG[Cos[t], 0.3]), {t, -Pi, Pi}, PlotRange → All]
]
```

Out[298]=



Normalization condition

In[300]:=
```
Integrate[2 Pi pAG[u, t], {u, -1, 1}, Assumptions → -1 < t < 1]
```

Out[300]= 1

Mean Cosine

In[301]:=
```
Integrate[2 Pi pAG[u, t] u, {u, -1, 1}, Assumptions → -1 < t < 1]
```

Out[301]= $\frac{t}{3}$

Back-scattering fraction

In[302]:=
```
FullSimplify[Integrate[2 Pi pAG[u, t], {u, -1, 0}, Assumptions → t > -1 && t < 1],
 Assumptions → -1 < g < 1]
```

Out[302]= $\frac{1}{1+t+\sqrt{1+t^2}}$

Legendre expansion coefficients

In[303]:=
```
Integrate[2 Pi (2 k + 1) pAG[u, t] LegendreP[k, u] /. k → 0,
 {u, -1, 1}, Assumptions → t > -1 && t < 1]
```

Out[303]= 1

In[304]:=
```
Integrate[2 Pi (2 k + 1) pAG[u, t] LegendreP[k, u] /. k → 1,
 {u, -1, 1}, Assumptions → t > -1 && t < 1]
```

Out[304]= t

In[305]:=
```
Integrate[2 Pi (2 k + 1) pAG[u, t] LegendreP[k, u] /. k → 2,
 {u, -1, 1}, Assumptions → t > -1 && t < 1]
```

Out[305]= t^2

In[306]:=
```
Integrate[2 Pi (2 k + 1) pAG[u, t] LegendreP[k, u] /. k → 3,
 {u, -1, 1}, Assumptions → t > -1 && t < 1]
```

Out[306]= t^3

Legendre Approximations

In[315]:=
```
With[{t = 0.6},
 Plot[{pAG[u, t], 1/(4 Pi) Sum[t^k LegendreP[k, u], {k, 0, 3}]},
  {u, -1, 1}, PlotRange → All]
]
```

Out[315]=



sampling

In[316]:=
```
cdf = Integrate[2 Pi pAG[u, t], {u, -1, x}, Assumptions → t > -1 && t < 1 && x < 1]
```

Out[316]=
$$\frac{1 + t - \sqrt{1 + t^2 - 2\,t\,x}}{2\,t}$$

In[317]:=
```
Solve[cdf == e, x]
```

Solve : There may be values of the parameters for which some or all solutions are not valid.

Out[317]= $\{\{x \to -1 + 2\,e + 2\,e\,t - 2\,e^2\,t\}\}$

In[345]:=
```
t = 0.95;
Show[
 Plot[2 Pi pAG[u, t], {u, -1, 1}],
 Histogram[
  Map[-1 + 2 # + 2 # t - 2 #^2 t &, Table[RandomReal[], {i, 1, 1 000 000}]], 150, "PDF"]
]
Clear[t];
```

Out[346]=