Infinite 3D medium, Isotropic Point Source, Rayleigh Scattering

Cauchy Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[•]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



c - single-scattering albedo
r - radial position coordinate in medium (distance from point source at origin)
u = cos θ - direction cosine

Namespace

In[3075]:=
```
Begin["inf3DisopointRayleighscatterCauchy`"]
```
Out[3075]=
```
inf3DisopointRayleighscatterCauchy`
```

Analytical results

Collision rate density

collision rate density Cc due to correlated emission:

derivation

In[3119]:=
```
Clear[cpc, c];
cpc[s_] := c 2/(Pi (1 + s^2))
```

In[3121]:=
```
f00 = Fpc[0, 0, cpc, u];
f10 = Fpc[1, 0, cpc, u];
f11 = Fpc[1, 1, cpc, u];
f20 = Fpc[2, 0, cpc, u];
f22 = Fpc[2, 2, cpc, u];
```

In[3126]:=
```
o = 3;
Clear[A, b, c, r, h, F];
A[n_] := 0;
A[0] := 1;
A[1] := 0;
A[2] := 1/2;
hsystem =
  Table[h[k] == 2/Pi u F[k, 0] + Sum[A[m] × h[m] × F[k, m], {m, 0, o - 1}], {k, 0, o - 1}];
hsystemsolve = Simplify[
  Solve[hsystem, Table[h[i], {i, 0, o - 1}]] /. F[0, 0] → f00 /. F[0, 1] → -f10 /.
      F[1, 1] → f11 /. F[1, 0] → f10 /.
    F[2, 0] → f20 /. F[0, 2] → f20 /. F[2, 2] → f22]
```

Out[3133]=

$$\left\{\left\{h[0] \to \left(2\,c\,u\left(8\,e^{u}\left(-1+e^{u}\right)u^{5}-9\,c\left(20\,e^{2u}+4\left(5+4\,u+u^{2}\right)-e^{u}\left(40+16\,u+8\,u^{2}+4\,u^{3}+u^{4}\right)\right)\right)\right)\Big/\left(\pi\left(8\,e^{2u}u^{6}+9\,c^{2}\left(20\,e^{2u}+4\left(5+4\,u+u^{2}\right)-e^{u}\left(40+16\,u+8\,u^{2}+4\,u^{3}+u^{4}\right)\right)-3\,c\,e^{u}\,u\left(-4\left(18+18\,u+8\,u^{2}+2\,u^{3}+u^{4}\right)+e^{u}\left(72-4\,u^{2}+3\,u^{4}\right)\right)\right)\right),\right.$$

$$h[1] \to -\left(\left(c\left(-4\,u^{2}\left(-6+u^{2}+e^{-u}\left(6+2\,u\left(3+u\right)\right)\right)F[1,2]-\frac{8\,u^{5}\,\text{MeijerG}\left[\left\{\left\{\frac{1}{2}\right\},\{\}\right\},\left\{\left\{\frac{1}{2},\frac{1}{2}\right\},\{-1\}\right\},\frac{u^{2}}{4}\right]}{\sqrt{\pi}}+\frac{1}{\sqrt{\pi}}\right.\right.\right.$$

$$\left.\left.c\,e^{-u}\left(e^{u}\left(216-12\,u^{2}+u^{4}\right)-4\left(54+54\,u+24\,u^{2}+6\,u^{3}+u^{4}\right)\right)\text{MeijerG}\left[\left\{\left\{\frac{1}{2}\right\},\{\}\right\},\left\{\left\{\frac{1}{2},\frac{1}{2}\right\},\{-1\}\right\},\frac{u^{2}}{4}\right]\right)\right)\Big/$$

$$\left(\pi\,u^{4}\left(8+\frac{9\,c^{2}\,e^{-2u}\left(20\,e^{2u}+4\left(5+4\,u+u^{2}\right)-e^{u}\left(40+16\,u+8\,u^{2}+4\,u^{3}+u^{4}\right)\right)}{u^{6}}-\frac{3\,c\,e^{-u}\left(-4\left(18+18\,u+8\,u^{2}+2\,u^{3}+u^{4}\right)+e^{u}\left(72-4\,u^{2}+3\,u^{4}\right)\right)}{u^{5}}\right)\right),$$

$$h[2] \to \left(8\,c\,e^{u}\,u^{4}\left(e^{u}\left(-6+u^{2}\right)+2\left(3+3\,u+u^{2}\right)\right)\right)\Big/\left(\pi\left(8\,e^{2u}u^{6}+9\,c^{2}\left(20\,e^{2u}+4\left(5+4\,u+u^{2}\right)-e^{u}\left(40+16\,u+8\,u^{2}+4\,u^{3}+u^{4}\right)\right)-3\,c\,e^{u}\,u\left(-4\left(18+18\,u+8\,u^{2}+2\,u^{3}+u^{4}\right)+e^{u}\left(72-4\,u^{2}+3\,u^{4}\right)\right)\right)\right)\Big\}\Big\}$$

In[3135]:=
```
Clear[r, c];
First[(2 k + 1) 1/(4 Pi c) (h[k]) u SphericalBesselJ[k, r u] /. k → 0 /. hsystemsolve //
  FullSimplify]
```

Out[3135]=

$$\left(u\left(8\,e^{u}\left(-1+e^{u}\right)u^{5}+9\,c\left(-20\,e^{2u}-4\,(5+u\,(4+u))+e^{u}\left(40+u\left(16+u\left(8+u\,(4+u)\right)\right)\right)\right)\right)\right.$$
$$\left.\text{Sin}[r\,u]\right)\Big/\left(2\,\pi^{2}\,r\left(8\,e^{2u}u^{6}+3\,c\,e^{u}\,u\left(72+e^{u}\left(-72+4\,u^{2}-3\,u^{4}\right)+4\,u\left(18+u\left(8+u\left(2+u\right)\right)\right)\right)+9\,c^{2}\left(20\,e^{2u}+4\,(5+u\,(4+u))-e^{u}\left(40+u\left(16+u\left(8+u\,(4+u)\right)\right)\right)\right)\right)\right)$$

result

In[3136]:=
```
Ccexact[r_, c_] := NIntegrate[
  (u (8 ℯ^u (-1 + ℯ^u) u^5 + 9 c (-20 ℯ^(2 u) - 4 (5 + u (4 + u)) + ℯ^u (40 + u (16 + u (8 + u (4 + u)))))))
     Sin[r u]) /
   (2 π^2 r (8 ℯ^(2 u) u^6 + 3 c ℯ^u u (72 + ℯ^u (-72 + 4 u^2 - 3 u^4) + 4 u (18 + u (8 + u (2 + u)))) +
       9 c^2 (20 ℯ^(2 u) + 4 (5 + u (4 + u)) - ℯ^u (40 + u (16 + u (8 + u (4 + u))))))),
  {u, 0, Infinity}, Method → "ExtrapolatingOscillatory"]
```

load MC data

In[3137]:=
```
ppoints[xs_, dr_, maxx_] :=
 Table[{dr (i) - 0.5 dr, xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[3138]:=
```
ppointsu[xs_, du_, Σt_] :=
 Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[3139]:=
```
fs = FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/MCdata/
     inf3D_isotropicpoint_rayleighscatter_cauchy*"];
```

In[3140]:=
```
index[x_] := Module[{data, c},
   data = Import[x, "Table"];
   c = data[[2, 3]];
   {c, data}];
simulations = index /@ fs;
cs = Union[#[[1]] & /@ simulations]
```

Out[3142]= {0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}

In[3143]:=
```
numcollorders = simulations[[1]][[-1]][[2, 13]];
```

Compare analytic and MC

Collision-rate density - Exact solution - comparison to MC

In[3144]:=
```
{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]}
```

Out[3144]= { Set c , 0.95}

In[3145]:=
```
data = SelectFirst[simulations, #[[1]] == c &][[2]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
MCCollisionRate = ppoints[data[[4]], dr, maxr];
exact1CRShallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]}] & /@ MCCollisionRate[[1 ;; 60]];
exact1CR = Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]}] & /@
   MCCollisionRate[[61 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
    ListPlot[MCCollisionRate[[1 ;; 60]],
     PlotRange → All, PlotStyle → PointSize[.01]],
    ListPlot[exact1CRShallow, PlotRange → All, Joined → True, PlotStyle → Black],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[MCCollisionRate, PlotRange → All, PlotStyle → PointSize[.01]],
    ListLogPlot[exact1CR, PlotRange → All, Joined → True, PlotStyle → Black],
    ListLogPlot[exact1CRShallow,
     PlotRange → All, Joined → True, PlotStyle → Black],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Infinite 3D, isotropic point source,
     Rayleigh scattering, Cauchy random flight - correlated
     emission\nCollision-rate density Cc[r], c = " <> ToString[c]]
```

Out[3153]=



Namespace

In[3155]:=
```
End[]
```

Out[3155]=
```
inf3DisopointRayleighscatterCauchy`
```