Random variate generation/Importance sampling using integral transforms

Eugene d'Eon
NVIDIA
Wellington, New Zealand
April 5, 2020   (updated May 10, 2020)

Abstract

We discuss various analytic and exact methods for deriving importance-sampling procedures for 1D distributions f(r) on the half line r ∈ [0,∞). Specifically:

- Finding convolutions using the forward Laplace transform
- Sampling as a superposition of exponentials using the inverse Laplace transform
- The Gaussian transform
- Generalized Box-Muller projections
- Sampling using a track-length estimator

1. Introduction

Simulating the linear transport of light or particles often involves drawing random variates from some given distribution on the half line, such as free-path length sampling in homogeneous classical media where we draw samples from the collision-rate density $\sigma_t e^{-\sigma_t r}$ using the CDF inversion method, leading to

```
r = -Log[1 - RandomReal[]] / σt.
```

There are times where we wish to sample other distributions f(r) on the half line where the CDF inversion method is not possible analytically. For example, in non-classical transport where free-path lengths between collision are not exponentially-distributed, we need sampling procedures for a large class of distributions that give the chord-lengths in a given class of random microstructure.

In this paper we discuss methods for sampling from a 1D distribution f(r) that is normalized on the half line $\int_0^\infty f(r)\, dr = 1$. In most cases, more than 1 random number will be required to sample f(r).

2. Deconvolution

If f(r) can be expressed as a convolution of more than 1 distribution, each of which has known sampling procedures, then f(r) can be sampled as a whole by sampling each of the distributions in its convolution and summing the values r_i. The Laplace transform can be used to factor f(r) into its deconvolved components using the convolution property of Laplace transforms: the Laplace transform of a convolution is the product of the two Laplace transforms, $\mathcal{L}_r[f_1(r) * f_2(r)](s) = \mathcal{L}_r[f_1(r)](s)\ \mathcal{L}_r[f_2(r)](s)$.

Example 2.1:

Consider the Erlang-2 / Gamma-2 distribution, f(r) = $e^{-r}\, r$.

In[•]:=
```
Clear[f];
f[r_] := Exp[-r] r;
 Integrate[f[r], {r, 0, Infinity}]
```

Out[•]= 1

We can’t analytically invert the CDF of f(r):

In[•]:=
```
Integrate[f[r], {r, 0, k}, Assumptions → k > 0]
```

Out[•]= $1 - e^{-k}\,(1 + k)$

Using the Laplace transform of f(r):

In[•]:=
```
LaplaceTransform[f[r], r, s]
```

Out[•]= $\frac{1}{(1 + s)^2}$

we note that f is the convolution of a function with itself, whose laplace transform is $\frac{1}{1+s}$. Using the inverse Laplace transform, we find

In[•]:=
```
InverseLaplaceTransform[1/(1 + s), s, r]
```

Out[•]= e^{-r}

We find the exponential, which we know how to sample. Therefore, Erlang-2 can be sampled using the sum of 2 calls to sample() procedure for the exponential:

In[•]:=
```
sampleExp[] := -Log[RandomReal[]];
```

In[•]:=
```
Show[
 Histogram[
  Table[sampleExp[] + sampleExp[], {i, Range[100 000]}], 200, "PDF"],
 Plot[f[r], {r, 0, 7}, PlotRange → All]
]
```

Out[•]=



In general the Gamma distribution f(r) = $e^{-r}\, r^{a-1}$ / Gamma[a] can be sampled as the sum of a exponential random variates. In the case of non – integer a, Marsaglia's method can be used.

3. Discrete Decomposition

Sometimes f(r) can be written as the sum of several non-negative distributions f(r) = $\sum_{i=1}^{n} w_i\, f_i(r)$, each of which is normalized $\int_0^\infty f_i(r)\, dr = 1$ and an analytic sampling procedure is known for each $f_i(r)$. In this case, one random number can be used to first select one of the f_i distributions using a CDF built from the weights w_i, and then a second independent random number can be used to sample f_i. The resulting distance r is then distributed by $f(r)$.

4. Continuous superpositions

A continuous generalization of the previous discrete decomposition is when f(r) can be expressed as a continuous superposition of a family of distributions g(r,s), where an exact analytic sampling procedure for g(r,s) is known: f(r) = $\int_0^\infty g(r, s)\, w(s)\, ds$, where s is a parameter for distribution g (such as the mean of g) and w(s) ≥ 0, for s ≥ 0. If a sampling procedure is also known for w(s), then we can first sample s from w, and then sample g(r,s) with the sampled s parameter value. See also [Devroye 2006 - Section 1.2].

Example 4.1 - BesselK$_0$

Consider the distribution $\frac{2}{\pi} K_0(r)$:

In[•]:=
```
Clear[f];
f[r_] := 2/Pi BesselK[0, r];
Integrate[f[r], {r, 0, Infinity}]
```

Out[•]= 1

We can write f(r) as the superposition of exponentials:

In[1097]:=
```
w[s_] := 2/(π s Sqrt[-1 + s^2]) HeavisideTheta[s - 1]
```

In[•]:=
```
Integrate[w[s] s Exp[-s r], {s, 0, Infinity}, Assumptions → r > 0]
```

Out[•]= $\frac{2\,\mathrm{BesselK}[0,r]}{\pi}$

We check to see we can sample inverse mfp s from w(s). The CDF of the weight function w(s) is

In[1101]:=
```
Integrate[w[s], {s, 1, S}, Assumptions → S > 1]
```

Out[1101]= $\frac{2\,\mathrm{ArcSec}[S]}{\pi}$

Here we integrate starting from s = 1, since the Heaviside function in w[s] makes the distribution 0 below s = 1. We can invert the result,

In[•]:=
```
Solve[ξ == 2 ArcSec[S]/π, S]
```

Out[•]= $\{\{S \to \mathrm{ConditionalExpression}[\mathrm{Sec}[\frac{\pi\,\xi}{2}], (\xi \neq 1 \,\&\&\, 0 < \pi\,\mathrm{Re}[\xi] < 2\,\pi) \,||\, (\pi\,\mathrm{Re}[\xi] == 0 \,\&\&\, \pi\,\mathrm{Im}[\xi] \geq 0) \,||\, (\pi\,\mathrm{Re}[\xi] == 2\,\pi \,\&\&\, \pi\,\mathrm{Im}[\xi] \leq 0)]\}\}$

So we can sample inverse mfp s using $\mathrm{Sec}[\frac{\pi\,\xi_1}{2}]$ where ξ_1 is a uniform random real in [0,1], and then sample the exponential using: $-\log(\xi_2)/s$:

In[•]:=
```
Show[
 Histogram[Table[(-Log[RandomReal[]]/Sec[π RandomReal[]/2]), {i, Range[50 000]}], 450, "PDF"],
 Plot[f[r], {r, 0, 10}, PlotRange → {0, 3}]
]
```

Out[•]=



5. Finding continuous superpositions of a given f(r)

When the inverse Laplace transform $\mathcal{L}_r^{-1}[f(r)](s)$ is known (superposition of exponentials)

When the inverse Laplace transform $\mathcal{L}_r^{-1}[f(r)](s)$ of f(r) is known, then a weight function for a superposition of exponentials s $e^{-s\,r}$ is $w(s) = \frac{\mathcal{L}_r^{-1}[f(r)](s)}{s}$. If w(s) can be sampled, then we have a new 2-random-number sampling procedure for f(r).

Example 5.1:

Consider the distribution f(r) given by:

In[∘]:=
$$f[r\_] := \sqrt{\frac{2}{\pi}} - e^{\frac{r^2}{2}}\, r\, \mathrm{Erfc}\left[\frac{r}{\sqrt{2}}\right]$$

We check the inverse Laplace:

In[∘]:=
$$\frac{1}{s}\,\mathrm{FullSimplify}[\mathrm{InverseLaplaceTransform}[f[r], r, s], \mathrm{Assumptions} \to s > 0]$$

Out[∘]=
$$e^{-\frac{s^2}{2}} \sqrt{\frac{2}{\pi}}$$

This is a weighting function w(s), which is simply a Gaussian/Normal distribution, which is easily sampled.

In[∘]:=
```
Show[
 Histogram[
  Table[-Log[RandomReal[]] / Abs[RandomVariate[NormalDistribution[]]], {i, Range[30 000]}],
  {0, 10, 0.03}, "PDF"],
 Plot[f[r], {r, 0, 10}, PlotRange → All]
]
```

Out[∘]=


When the inverse Laplace transform $\mathcal{L}_r^{-1}\left[f\left(\sqrt{r}\right)\right](s)$ is known (superpositions of Gaussians)

Sometimes the inverse Laplace transform of f(r) is not known but the inverse Laplace transform of $f\left(\sqrt{r}\right)$ is. In this case, it may be possible to sample f(r) using the Gaussian (or Normal) distribution. To do this, use the Gaussian transform, described next.

6. The Gaussian Transform

The 1D Gaussian/Normal distribution on the half line with variance s is:

In[•]:=
$$\texttt{Gaussian[r\_, s\_] :=} \frac{2}{\sqrt{2\,\texttt{Pi}\,s}}\,\texttt{Exp}\left[-r^2/(2\,s)\right]$$

and is normalized on the half line:

```
In[•]:= Integrate[Gaussian[r, s], {r, 0, Infinity}, Assumptions → s > 0]
Out[•]= 1
```

This one-sided Gaussian/Normal distribution can be sampled using inverseErf[] or the Box-Muller algorithm.

The Gaussian transform [Alecu et al. 2006] can be used to solve for a weight function w[s] such that f(r) can be expressed as a superposition of Gaussians: f(r) = $\int_0^\infty \text{Gaussian}(r, s)\, w(s)\, ds$.

Example 6.1 - w[s] is Erlang-2/Gamma-2:

```
In[•]:= w[s_] := Exp[-s] s
In[•]:= Integrate[w[s], {s, 0, Infinity}]
Out[•]= 1
```

Find f(r) given w(s) as the weighting function of Gaussians:

```
In[•]:= f = Integrate[w[s] × Gaussian[r, s], {s, 0, Infinity}, Assumptions → r > 0]
```

Out[•]=
$$\frac{e^{-\sqrt{2}\,r}\left(1+\sqrt{2}\,r\right)}{\sqrt{2}}$$

Try the Gaussian transform to find w(s) given f(r):

In[•]:=
$$\frac{1}{2\,s}\sqrt{\frac{\texttt{Pi}}{2\,s}}\;\texttt{InverseLaplaceTransform}\left[f\;/.\;r\to\sqrt{r},\,r,\,t\right]\;/.\;t\to\frac{1}{2\,s}$$

Out[•]= $e^{-s}\,s$

Example 6.2 - BesselK$_0$

In[•]:=
$$\texttt{w[s\_] :=} \frac{\texttt{Exp}[-s]\,s^{-1/2}}{\sqrt{\texttt{Pi}}}$$

In[∘]:= `Integrate[w[s], {s, 0, Infinity}]`

Out[∘]= 1

Find f(r) given w(s) as the weighting function of Gaussians:

In[∘]:= `f = Integrate[w[s] × Gaussian[r, s], {s, 0, Infinity}, Assumptions → r > 0]`

Out[∘]= $\frac{2\sqrt{2}\ \text{BesselK}\left[0, \sqrt{2}\ r\right]}{\pi}$

Try the Gaussian transform to find w(s) given f(r):

In[∘]:= $\frac{1}{2\,s}\sqrt{\frac{\text{Pi}}{2\,s}}\ \text{InverseLaplaceTransform}\left[f\ /.\ r \to \sqrt{r}\text{, r, t}\right]\ /.\ t \to \frac{1}{2\,s}$

Out[∘]= $\frac{e^{-s}\sqrt{\frac{1}{s}}}{\sqrt{\pi}}$

Example 6.3

In[∘]:= $\text{w[s\_] :=}\ \frac{1}{1+s}$

Find f(r) given w(s) as the weighting function of Gaussians:

In[∘]:= `f = Integrate[w[s] × Gaussian[r, s], {s, 0, Infinity}, Assumptions → r > 0]`

Out[∘]= $e^{\frac{r^2}{2}}\sqrt{2\,\pi}\ \text{Erfc}\left[\frac{r}{\sqrt{2}}\right]$

Try the Gaussian transform to find w(s) given f(r):

In[∘]:= $\frac{1}{2\,s}\sqrt{\frac{\text{Pi}}{2\,s}}\ \text{InverseLaplaceTransform}\left[f\ /.\ r \to \sqrt{r}\text{, r, t}\right]\ /.\ t \to \frac{1}{2\,s}\ \text{// Simplify}$

Out[∘]= $\frac{1}{1+s}$

Example 6.4

In[∘]:= $\text{w[s\_] :=}\ \frac{1}{s\,(1+s)}$

Find f(r) given w(s) as the weighting function of Gaussians:

In[∘]:= `f = Integrate[w[s] × Gaussian[r, s], {s, 0, Infinity}, Assumptions → r > 0]`

Out[∘]= $\frac{2}{r} - e^{\frac{r^2}{2}}\sqrt{2\,\pi}\ \text{Erfc}\left[\frac{r}{\sqrt{2}}\right]$

Try the Gaussian transform to find w(s) given f(r):

In[•]:= $\frac{1}{2\,s}\sqrt{\frac{Pi}{2\,s}}$ InverseLaplaceTransform[f /. r → $\sqrt{r}$, r, t] /. t → $\frac{1}{2\,s}$ // Simplify

Out[•]= $\frac{1}{s+s^2}$

Example 6.5: Cauchy superposition of Gaussians

In[•]:= w[s_] := $\frac{1}{1+s^2}\,\frac{2}{Pi}$

In[•]:= Integrate[w[s], {s, 0, Infinity}]

Out[•]= 1

Find f(r) given w(s) as the weighting function of Gaussians:

In[•]:= f = Integrate[w[s] × Gaussian[r, s], {s, 0, Infinity}, Assumptions → r > 0]

Out[•]= $\frac{2\left(\pi\,\mathrm{Cos}\left[\frac{r^2}{2}\right]\left(1-2\,\mathrm{FresnelC}\left[\frac{r}{\sqrt{\pi}}\right]\right)+\pi\left(1-2\,\mathrm{FresnelS}\left[\frac{r}{\sqrt{\pi}}\right]\right)\mathrm{Sin}\left[\frac{r^2}{2}\right]\right)}{\pi^{3/2}}$

In[•]:= Plot[$\frac{2\left(\pi\,\mathrm{Cos}\left[\frac{r^2}{2}\right]\left(1-2\,\mathrm{FresnelC}\left[\frac{r}{\sqrt{\pi}}\right]\right)+\pi\left(1-2\,\mathrm{FresnelS}\left[\frac{r}{\sqrt{\pi}}\right]\right)\mathrm{Sin}\left[\frac{r^2}{2}\right]\right)}{\pi^{3/2}}$, {r, 0, 5}]

Out[•]=


Try the Gaussian transform to find w(s) given f(r):

In[•]:= $\frac{1}{2\,s}\sqrt{\frac{Pi}{2\,s}}$ InverseLaplaceTransform[f /. r → $\sqrt{r}$, r, t] /. t → $\frac{1}{2\,s}$ // Simplify

Out[•]= $\frac{1}{\sqrt{2}\,\pi}\left(\frac{1}{s}\right)^{3/2}$ InverseLaplaceTransform[$\pi\,\mathrm{Cos}\left[\frac{r}{2}\right]\left(1-2\,\mathrm{FresnelC}\left[\frac{\sqrt{r}}{\sqrt{\pi}}\right]\right)+\pi\left(1-2\,\mathrm{FresnelS}\left[\frac{\sqrt{r}}{\sqrt{\pi}}\right]\right)\mathrm{Sin}\left[\frac{r}{2}\right]$, r, $\frac{1}{2\,s}$]

Mathematica is not always able to find it.

7. Generalized Box-Muller

We now show how to solve for isotropic random flights in d dimension that project onto a single axis to give some desired symmetric distribution. This is a generalization of polar methods [Devroye 2006 - Section 1.2].

We need the surface area of the sphere of radius r in dD:

In[∘]:=
```
SA[d_, r_] := d Pi^(d/2)/Gamma[d/2 + 1] r^(d-1)
```

$$SA[d\_, r\_] := d \frac{Pi^{d/2}}{Gamma\left[\frac{d}{2}+1\right]} r^{d-1}$$

and the inverse Fourier/Hankel transform in dD [Dutka 1985]

In[∘]:=
$$inv[d\_, f\_] := r^{1-d/2} (2\,Pi)^{-d/2}\, Integrate\left[z^{d/2}\, BesselJ[d/2-1, z\, r]\, f, \{z, 0, Infinity\}, Assumptions \to d \geq 1 \,\&\&\, r > 0 \,\&\&\, assumps\right]$$

Given a desired distribution f(r) on the half line we wish to sample, we first form the Fourier transform

$$F = \sqrt{2\,Pi}\; FourierTransform\left[\frac{1}{2} f[Abs[r]], r, z\right]$$

and then search for some dimension d such that the inverse Hankel transform

$$p = inv[d, F] \times SA[d, r]$$

gives a distribution p(r) that we can sample analytically. If so, then sampling a random isotropic direction in dD, and then a random step from p(r), and projecting onto a single axis, and taking the abs(x), gives a sample procedure for f(r).

Example 7.1 - Box Muller sampling of the Gaussian

Suppose we wish to sample a random variate from a one-sided Gaussian/Normal distribution

In[∘]:=
```
f[r_] := Gaussian[r, 1]
```

In[∘]:=
```
f[r]
```

Out[∘]= $e^{-\frac{r^2}{2}} \sqrt{\frac{2}{\pi}}$

In[∘]:=
```
Integrate[f[r], {r, 0, Infinity}]
```

Out[∘]= 1

Following the 2 steps above, and knowing that the Box-Muller transform works in Flatland, d = 2, we find

In[∘]:= $F = \sqrt{2\,Pi}\; FourierTransform\left[\frac{1}{2} f[Abs[r]], r, z\right]$

Out[∘]= $e^{-\frac{z^2}{2}}$

In[∘]:=
```
p = inv[2, F] × SA[2, r]
```

Out[∘]= $e^{-\frac{r^2}{2}} r$

We find that a single-step random flight in 2D with an isotropic starting direction and free-path length drawn from $e^{-\frac{r^2}{2}} r$ will produce a Gaussian random number when projected onto one axis.

Sampling p by CDF inverse leads to:

In[∘]:=
```
Integrate[ⅇ^(-r^2/2) r, {r, 0, R}, Assumptions → R > 0]
```

Out[∘]= $1 - e^{-\frac{R^2}{2}}$

In[∘]:=
```
Solve[1 - ⅇ^(-R^2/2) == xi, R]
```

Out[∘]= $\left\{\left\{R \to \text{ConditionalExpression}\left[-\sqrt{2}\sqrt{2\,i\,\pi\,c_1 + \text{Log}\left[\frac{1}{1-\text{xi}}\right]},\ c_1 \in \mathbb{Z}\right]\right\},\right.$
$\left.\left\{R \to \text{ConditionalExpression}\left[\sqrt{2}\sqrt{2\,i\,\pi\,c_1 + \text{Log}\left[\frac{1}{1-\text{xi}}\right]},\ c_1 \in \mathbb{Z}\right]\right\}\right\}$

And so we sample a radius using $\sqrt{2}\sqrt{\text{Log}\left[\frac{1}{1-\text{xi}}\right]}$

In[∘]:=
```
Show[
 Histogram[
  Table[Abs[√2 √(Log[1/(1 - RandomReal[])]) Cos[2 Pi RandomReal[]]],
   {i, Range[100 000]}], 200, "PDF"],
 Plot[f[r], {r, 0, 7}, PlotRange → All]
]
```

Out[∘]=



Example 7.2 - Sampling GGX distribution of visible normals

In sampling the distribution of visible normals for the GGX normal distribution the following distribution requires sampling [Heitz and d'Eon 2014] in order to sample slope q conditioned on having sampled slope p:

$$f[q\_] := \frac{2}{\pi\left(1 + q^2\right)^2}.$$

This was sampled using an approximate inverse in [Heitz and d'Eon 2014] and later an exact scheme was provided [Heitz 2018]. However, the generalized Box-Muller approach in 2D also works. We find (using a variant for f(q) on the full line):

In[∘]:= F = $\sqrt{2\,\mathrm{Pi}}$ FourierTransform[f[Abs[r]], r, z]

Out[∘]= $e^{-\mathrm{Abs}[z]}\,(1+\mathrm{Abs}[z])$

In[∘]:= p = inv[2, F] × SA[2, r]

Out[∘]= $\frac{3\,r}{(1+r^2)^{5/2}}$

This distribution is easily sampled

In[∘]:= Integrate$\left[\frac{3\,r}{(1+r^2)^{5/2}}, \{r, 0, R\}, \text{Assumptions} \to R > 0\right]$

Out[∘]= $1-\frac{1}{(1+R^2)^{3/2}}$

In[∘]:= Solve$\left[xi == 1-\frac{1}{(1+R^2)^{3/2}}, R\right]$

Out[∘]= $\left\{\left\{R \to -\sqrt{-1+\frac{1}{(1-xi)^{2/3}}}\right\}, \left\{R \to \sqrt{-1+\frac{1}{(1-xi)^{2/3}}}\right\}\right\}$

In[∘]:=

```
Show[
 Plot[f[q] , {q, -3, 3}],
 Histogram[
  Table[
   Sin[2 Pi RandomReal[]] Sqrt[-1 + 1/(1 - RandomReal[])^(2/3)]
   , {i, Range[10 000]}]
  , 500, "PDF"]
]
```

Out[∘]=



8. Combining various techniques

We consider here an example that combines several of the above techniques to sample the distribution of visible normals for GGX by using the exact sampling method for Beckmann. This is far less efficient than known sampling procedures for the GGX vNDF sampling [Heitz 2018]. However, the procedure highlights several of the above approaches. Based on the observation that GGX is simply a superposition of Beckmanns:

In[1825]:= $$\text{DBeckmann}[u\_, \alpha\_] := \frac{1}{\text{Pi}\,\alpha^2\,(u)^4}\,E^{-\left(\frac{\left(\frac{1}{(u)^2}-1\right)}{\alpha^2}\right)}$$

In[•]:= $$\text{DGGX}[u\_, \alpha\_] := \frac{\alpha^2\,\frac{1}{(u)^4}}{\pi\left(\alpha^2+\left(\frac{1}{(u)^2}-1\right)\right)^2}$$

In[•]:= $$\text{Integrate}\left[\frac{2\,e^{-\frac{\text{GGXm}^2}{m^2}}\,\text{GGXm}^2}{m^3}\,\text{DBeckmann}[u, m], \{m, 0, \text{Infinity}\},\ \text{Assumptions} \to \text{GGXm} > 0\,\&\&\,0 < u < 1\right] - \text{DGGX}[u, \text{GGXm}]\ //\ \text{FullSimplify}$$

Out[•]= 0

we find that the GGX vNDF can be sampled using an exact vNDF procedure for Beckmann [Heitz and d'Eon 2014] by first sampling Beckmann roughness m' from:

In[•]:= $$\text{mprime}[m\_, u\_] := -\frac{e^{\frac{m}{-1+u^2}}\,(-1+u)}{\sqrt{\pi}\,\sqrt{m-m\,u^2}}+\frac{e^{-m}\,u\left(1+\text{Erf}\left[u\sqrt{-\frac{m}{-1+u^2}}\right]\right)}{1+u}$$

where u = cos
θ_i and m is the roughness of the desired GGX NDF. To sample the $m'(m)$ distribution we consider :

In[•]:= $$\text{LaplaceTransform}[\text{mprime}[m, u], m, s]$$

Out[•]= $$\frac{u}{(1+s)\,(1+u)}+\frac{u}{(1+s)\sqrt{1+(1+s)\left(-1+\frac{1}{u^2}\right)}\,(1+u)}-\frac{-1+u}{\sqrt{1+s-s\,u^2}}$$

We separate this into 3 components $\overline{m}(s) = \text{m1}(s) + \text{m2}(s) + \text{m3}(s)$:

In[•]:= $$\text{m1}[s\_] := \frac{u}{1+u}\,\frac{1}{1+s}$$

We recognize this as an exponential distribution from $\frac{1}{1+s}$. The second term can simplify,

In[•]:= $$\text{FullSimplify}\left[\frac{u}{\sqrt{1+(1+s)\left(-1+\frac{1}{u^2}\right)}\,(1+u)}, \text{Assumptions} \to s > 0\,\&\&\,0 < u < 1\right]$$

Out[•]= $$\frac{u^2}{(1+u)\,\sqrt{1+s-s\,u^2}}$$

In[•]:= $$\text{m2}[s\_] := \frac{u^2}{1+u}\,\frac{1}{1+s}\,\frac{1}{\sqrt{1+s-s\,u^2}}$$

We recognize this as the convolution of an exponential $\frac{1}{1+s}$ with a new distribution $\frac{1}{\sqrt{1+s-s\,u^2}}$, which also appears in the third term:

In[•]:= $$\text{m3}[s\_] := (1-u)\,\frac{1}{\sqrt{1+s-s\,u^2}}$$

We now have three terms. Their selection weights are found from s = 0,

In[•]:=
```
{m1[s], m2[s], m3[s]} /. s → 0
```

Out[•]= $\left\{\frac{u}{1+u}, \frac{u^2}{1+u}, 1-u\right\}$

Which we verify sum to 1:

In[•]:=
```
% // Total // FullSimplify
```

Out[•]= 1

We just need to find a sampling procedure for the inverse Laplace transform of $\frac{1}{\sqrt{1+s-s\,u^2}}$,

In[•]:= $\text{InverseLaplaceTransform}\left[\frac{1}{\sqrt{1+s-s\,u^2}}, s, r\right]$

Out[•]= $\frac{e^{-\frac{r}{1-u^2}}}{\sqrt{\pi}\,\sqrt{r}\,\sqrt{1-u^2}}$

Try CDF inversion,

In[•]:= $\text{FullSimplify}\left[\text{Integrate}\left[\frac{e^{-\frac{r}{1-u^2}}}{\sqrt{\pi}\,\sqrt{r}\,\sqrt{1-u^2}}, \{r, 0, R\}, \text{Assumptions} \to R > 0\right],\right.$
$\left.\text{Assumptions} \to 0 < u < 1\;\&\&\; R > 0\right]$

Out[•]= $\text{Erf}\left[\sqrt{-\frac{R}{-1+u^2}}\right]$

In[•]:= $\text{Solve}\left[xi == \text{Erf}\left[\sqrt{-\frac{R}{-1+u^2}}\right], R\right]$

... Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[•]= $\left\{\left\{R \to -\left(-1+u^2\right)\,\text{InverseErf}[xi]^2\right\}\right\}$

We can sample m2 and m3 using this, assuming we have an efficient InverseErf(),

```
sampleprime[u_] := Module[{xi1},
  xi1 = RandomReal[]; (* term selection variable *)
  If[xi1 < u/(1 + u),
   -Log[RandomReal[]] (* sample m1(s) *)
   ,
   If[xi1 < u/(1 + u) + (1 - u),
    -(-1 + u^2) InverseErf[RandomReal[]]^2 (* sample m3 *)
    ,
    -Log[RandomReal[]] - (-1 + u^2) InverseErf[RandomReal[]]^2
    (* sample m2 - by convolution *)
   ]
  ]
 ]
```

Verify our derivation:

```
In[•]:= With[{u = 0.3},
  Show[
   Histogram[Table[sampleprime[u], {i, Range[100 000]}
    ], 1300, "PDF"],
   Plot[mprime[m, u], {m, 0, 5}, PlotRange → {0, 5}]

  ]
 ]
```

Out[•]=


9. Track-length estimators

In the case that a track-length estimator [Heitz and Belcour 2018] is used to sample f(r) by scoring at all positions in [0,r] when distance r is sampled, we note that the r can be sampled as a superposition of exponentials by sampling inverse mfp s from $\mathcal{L}_r^{-1}[f(r)](s)$ as opposed to $\frac{\mathcal{L}_r^{-1}[f(r)](s)}{s}$ for standard sampling, which follows from the property of the Laplace transform of the integral of a function f(r).

Example 9.1:

Suppose we want to sample f(r) using a track-length estimator where f(r) is given as:

In[∘]:=
```
f[r_] := 1 + r + ⅇ^r r (2 + r) ExpIntegralEi[-r]
```

In[∘]:=
```
Integrate[f[r], {r, 0, Infinity}]
```

Out[∘]= 1

Knowing that f(r) is given by a Laplace transform (contrived example)

In[∘]:=
```
LaplaceTransform[2 s/(1 + s)^3, s, r] // FullSimplify
```

Out[∘]= $1 + r + e^r\, r\, (2 + r)\, \mathrm{ExpIntegralEi}[-r]$

We can sample s from $\frac{2\,s}{(1+s)^3}$ and then r from $s\, e^{-r s}$ and the track – length estimator will sample f(r).

CDF sampling of $\frac{2\,s}{(1+s)^3}$ yields

In[∘]:=
```
Solve[xi == Integrate[2 s/(1 + s)^3, {s, 0, S}, Assumptions → S > 0], S]
```

Out[∘]= $\left\{\left\{S \to \frac{-\sqrt{xi} - xi}{-1 + xi}\right\}, \left\{S \to \frac{\sqrt{xi} - xi}{-1 + xi}\right\}\right\}$

Verify that a track-length estimator samples f(r):

In[∘]:=
```
tls = Sum[HeavisideTheta[# - t] &[-Log[RandomReal[]]/((-Sqrt[#] - #)/(-1 + #) &[RandomReal[]])], {i, Range[700]}]/700;
```

In[∘]:=
```
Show[
 Plot[f[t], {t, 0, 6}, PlotRange → {0, 1}],
 Plot[tls, {t, 0, 16}, Filling → Axis, PlotRange → {0, 1}]
]
```

Out[∘]=



10. References

- Alecu et al. 2006 - The Gaussian Transform of Distributions: Definition, Computation and Application. *IEEE TRANSACTIONS ON SIGNAL PROCESSING*, VOL. **54**, NO. 8, AUGUST 2006. doi: 10.1109/TSP.2006.877657
- Devroye, L. 2006 - Nonuniform random variate generation, in *Handbooks in operations research and management science* ***13****, Chapter 4, 83-121.* doi: 10.1016/S0927-0507(06)13004-2
- Dutka, J. 1985. On the problem of random flights. *Arch. Hist. Exact Sci.* **32**(3–4): 351–75. doi: 10.1007/BF00348451
- Heitz, E. and d’Eon, E. 2014 - Importance sampling microfacet-based BSDFs using the distribution of visible normals. In *Proceedings of the 25th Eurographics Symposium on Rendering*, Eurographics Association, Aire-la-Ville, Switzerland, EGSR ’14, 103–112. URL: https://hal.inria.fr/hal-00996995v2.
- Heitz and Belcour 2018 - A note on track-length sampling with non-exponential distributions. https://hal.inria.fr/hal-01788593v1
- Heitz, E. 2018 - Sampling the GGX Distribution of Visible Normals, *JCGT.* http://jcgt.org/published/0007/04/01/