4D Half Space, Albedo Problem, Isotropic Scattering

Exponential Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[2388]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation

α - single-scattering albedo
Σt - extinction coefficient
x - depth in medium (positive inside the scattering half space)
u = cos θ - direction cosine

Load MC Data

Delta Incidence

In[2389]:=
```
FourDhalfspaceAlbedoProblemIsotropic`deltafs = FileNames[
    "code/4D_medium/halfSpace/albedoProblem/data/albedoproblem_delta*.txt"];
```

In[2390]:=
```
FourDhalfspaceAlbedoProblemIsotropic`indexdelta[x_] := Module[{data, α, Σt, ui},
    data = Import[x, "Table"];
    Σt = data[[1, -1]];
    α = data[[2, 3]];
    ui = data[[1, -4]];
    {α, Σt, ui, data}];
```

In[2391]:=
```
FourDhalfspaceAlbedoProblemIsotropic`simulationsdelta =
  FourDhalfspaceAlbedoProblemIsotropic`indexdelta /@
   FourDhalfspaceAlbedoProblemIsotropic`deltafs;
FourDhalfspaceAlbedoProblemIsotropic`alphas =
 Union[#[[1]] & /@ FourDhalfspaceAlbedoProblemIsotropic`simulationsdelta]
```

Out[2392]=
```
{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}
```

In[2393]:=
```
FourDhalfspaceAlbedoProblemIsotropic`muts =
 Union[#[[2]] & /@ FourDhalfspaceAlbedoProblemIsotropic`simulationsdelta]
```

Out[2393]=
```
{1}
```

In[2394]:=
```
FourDhalfspaceAlbedoProblemIsotropic`uis =
 Union[#[[3]] & /@ FourDhalfspaceAlbedoProblemIsotropic`simulationsdelta]
```

Out[2394]=
```
{0.1, 0.25, 0.5, 1}
```

In[2395]:=
```
FourDhalfspaceAlbedoProblemIsotropic`numcollorders =
  FourDhalfspaceAlbedoProblemIsotropic`simulationsdelta[[1]][[4]][[2, 13]];
FourDhalfspaceAlbedoProblemIsotropic`maxz =
 FourDhalfspaceAlbedoProblemIsotropic`simulationsdelta[[1]][[4]][[2, 7]];
FourDhalfspaceAlbedoProblemIsotropic`dz =
 FourDhalfspaceAlbedoProblemIsotropic`simulationsdelta[[1]][[4]][[2, 9]];
FourDhalfspaceAlbedoProblemIsotropic`numz =
  Floor[FourDhalfspaceAlbedoProblemIsotropic`maxz /
    FourDhalfspaceAlbedoProblemIsotropic`dz];
```

Util

In[2398]:=
```
FourDhalfspaceAlbedoProblemIsotropic`plotpoints[data_, du_] :=
  Table[{du i - 0.5 du, data[[i]]}, {i, 1, Length[data]}];
```

In[2399]:=
```
FourDhalfspaceAlbedoProblemIsotropic`plotpoints2[data_, du_] :=
  Table[{du i - 0.5 du, data[[i]] Sqrt[1 - (du i - 0.5 du)^2]/2}, {i, 1, Length[data]}];
```

In[2400]:=
```
FourDhalfspaceAlbedoProblemIsotropic`plotpoints[data_, d_, min_] :=
  Table[{min + d i - 0.5 d, data[[i]]}, {i, 1, Length[data]}];
```

H-function

In[2401]:=
```
FourDhalfspaceAlbedoProblemIsotropic`H[c_, u_] :=
  Exp[-1/Pi NIntegrate[u/(1 + u^2 t^2) Log[1 - c 2 (-1 + Sqrt[1 + t^2])/t^2], {t, 0, Infinity}]]
```

Albedo

Exact Solution

In[2402]:=
```
FourDhalfspaceAlbedoProblemIsotropic`albedoexact[α_, ui_] :=
  1 - √(1 - α) FourDhalfspaceAlbedoProblemIsotropic`H[α, ui]
```

Albedo Benchmarks

Albedo - Delta Illumination - General incidence

In[2403]:=
```
p = Manipulate[
  Module[{Rs},
   Rs = {#[[1]], #[[4]][[3, 3]]} & /@ Select[
      FourDhalfspaceAlbedoProblemIsotropic`simulationsdelta, #[[3]] == ui &];
   Quiet[Show[
     Show[
      ListLogLinearPlot[Rs, PlotStyle → {PointSize[0.01], Black}],
      LogLinearPlot[Quiet[FourDhalfspaceAlbedoProblemIsotropic`albedoexact[
         c, ui]], {c, 0.001, .9999}, PlotRange → All]
     ], Frame → True, FrameLabel → {{R[α, ui],},
       {"Single scattering albedo: α", "Total Reflectance/Albedo R(α,uᵢ):
          isotropically-scattering half space\nincidence (uᵢ=" <>
        ToString[ui] <> "), indexed-matched boundary"}}
    ]]
  ]
  , {ui, FourDhalfspaceAlbedoProblemIsotropic`uis}]
```

Out[2403]=