Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

---

vMF (spherical Gaussian) Scattering

[Pomraning and Prinja 1995] - "Transverse Diffusion of a Collimated Particle Beam" https://doi.org/10.1007/BF02178551

In[∘]:=
```
pVMF[u_, k_] := k/(4 Pi Sinh[k]) Exp[k u]

Show[
 Plot[pVMF[Cos[t], 5.8], {t, -Pi, Pi}, PlotRange → All],
 Plot[pVMF[Cos[t], 15], {t, -Pi, Pi}, PlotRange → All],
 Plot[pVMF[Cos[t], 30], {t, -Pi, Pi}, PlotRange → All],

 Frame → True,
 FrameLabel → {{p[Cos[θ]],}, {θ, "vMF, k = {5.8,15,30}"}}]
```



Normalization condition

```
Integrate[2 Pi pVMF[u, k], {u, -1, 1}, Assumptions → k > 0]
```

1

Mean cosine (g)

```
Integrate[2 Pi u pVMF[u, k], {u, -1, 1}, Assumptions → k > 0]
```

$$-\frac{1}{k} + \mathrm{Coth}[k]$$

Legendre expansion coefficients

In[•]:=
```
Integrate[2 Pi (2 o + 1) pVMF[u, k] LegendreP[o, u] /. o → 0,
  {u, -1, 1}, Assumptions → k > 0]
```

Out[•]= 1

In[•]:=
```
Integrate[2 Pi (2 o + 1) pVMF[u, k] LegendreP[o, u] /. o → 1,
  {u, -1, 1}, Assumptions → k > 0]
```

Out[•]= $$-\frac{3}{k} + 3\,\mathrm{Coth}[k]$$

In[•]:=
```
Integrate[2 Pi (2 o + 1) pVMF[u, k] LegendreP[o, u] /. o → 2,
  {u, -1, 1}, Assumptions → k > 0]
```

Out[•]= $$\frac{5\left(3 + k^2 - 3\,k\,\mathrm{Coth}[k]\right)}{k^2}$$

In[•]:=
```
Integrate[2 Pi (2 o + 1) pVMF[u, k] LegendreP[o, u] /. o → 3,
  {u, -1, 1}, Assumptions → k > 0]
```

Out[•]= $$\frac{7\left(-3\left(5 + 2\,k^2\right) + k\left(15 + k^2\right)\mathrm{Coth}[k]\right)}{k^3}$$

```
Integrate[2 Pi (2 o + 1) pVMF[u, k] LegendreP[o, u] /. o → 4,
  {u, -1, 1}, Assumptions → k > 0]
```

$$\frac{9\left(105 + 45\,k^2 + k^4 - 5\,k\left(21 + 2\,k^2\right)\mathrm{Coth}[k]\right)}{k^4}$$

sampling

```
k = 3;
Show[Histogram[
  Map[Log[E^-k (1 - #) + E^k #]/k &, Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"],
 Plot[2 Pi pVMF[u, k], {u, -1, 1}, PlotRange → All]

]
Clear[k];
```



When cosine u has been sampled with random variable ξ, what is the PDF at the sampled direction in terms of ξ?

In[•]:=
```
FullSimplify[pVMF[Log[E^-k (1 - #) + E^k #]/k &[ξ], k], Assumptions → k > 0 && 0 < ξ < 1]
```

Out[•]=

$$\frac{k\left(-1+2\xi+\text{Coth}[k]\right)}{4\pi}$$