Infinite 3D medium, Isotropic Point Source, Rayleigh Scattering

Gamma-4 Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[•]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



c - single-scattering albedo
Σt - extinction coefficient
r - radial position coordinate in medium (distance from point source at origin)
u = cos θ - direction cosine
b - anisotropy parameter

Namespace

In[653]:=
```
Begin["inf3DisopointRayleighscatterGamma4`"]
```
Out[653]=
```
inf3DisopointRayleighscatterGamma4`
```

Analytical results

Collision rate density

collision rate density Cc due to correlated emission:

derivation

In[654]:=
```
pc[s_] := 1/6 Exp[-s] s^3
```

In[655]:=
```
f00 = Fpc[0, 0, pc];
f01 = Fpc[0, 1, pc];
f11 = Fpc[1, 1, pc];
f20 = Fpc[2, 0, pc];
f22 = Fpc[2, 2, pc];
```

In[660]:=
```
o = 3;
Clear[A, b, c, r, h];
A[n_] := 0;
A[0] := 1;
A[1] := 0;
A[2] := 1/2;
hsystem = Table[
   h[k] == 2/Pi c u F[k, 0] + c Sum[A[m] × h[m] × F[k, m], {m, 0, o - 1}], {k, 0, o - 1}];
hsystemsolve = Simplify[Solve[hsystem, Table[h[i], {i, 0, o - 1}]] /. F[0, 0] → f00 /.
        F[0, 1] → f01 /. F[1, 1] → f11 /. F[1, 0] → -f01 /.
     F[2, 0] → f20 /. F[0, 2] → f20 /. F[2, 2] → f22]
```
Out[667]=
$$\{\{h[0] \to \left(2\,c\,u\left(2\,u^5\left(-3-2\,u^2+u^4\right)-9\,c\,u\left(-3-u^2+u^4\right)+9\,c\left(-3-2\,u^2+u^4\right)\text{ArcTan}[u]\right)\right)\Big/\left(3\,\pi\left(u\left(-2\left(u+u^3\right)^4+3\,c^2\left(-3-u^2+u^4\right)+c\left(9+33\,u^2+46\,u^4+25\,u^6+3\,u^8\right)\right)-3\,c\left(1+u^2\right)\left(c\left(-3+u^2\right)+3\left(1+u^2\right)^3\right)\text{ArcTan}[u]\right)\right),$$
$$h[1] \to \left(8\,c\,u^2\left(6\,u^5\left(1+u^2\right)^3+27\,c\left(1+u^2\right)^3\text{ArcTan}[u]+c\,u\left(-27-72\,u^2-60\,u^4-11\,u^6+3\,u^5\,F[1,2]+9\,u^7\,F[1,2]+9\,u^9\,F[1,2]+3\,u^{11}\,F[1,2]\right)\right)\right)\Big/\left(9\,\pi\left(1+u^2\right)^2\left(2\,u^5\left(1+u^2\right)^4+3\,c^2\,u\left(3+u^2-u^4\right)-c\,u\left(9+33\,u^2+46\,u^4+25\,u^6+3\,u^8\right)+3\,c\left(1+u^2\right)\left(c\left(-3+u^2\right)+3\left(1+u^2\right)^3\right)\text{ArcTan}[u]\right)\right),$$
$$h[2] \to -\left(\left(16\,c\,u^8\left(1+u^2\right)\right)\Big/\left(3\,\pi\left(u\left(-2\left(u+u^3\right)^4+3\,c^2\left(-3-u^2+u^4\right)+c\left(9+33\,u^2+46\,u^4+25\,u^6+3\,u^8\right)\right)-3\,c\left(1+u^2\right)\left(c\left(-3+u^2\right)+3\left(1+u^2\right)^3\right)\text{ArcTan}[u]\right)\right)\right)\}\}$$

In[668]:=
```
Clear[r];
(2 k + 1) 1/(4 Pi r c) (h[k]) j2[k, r u] /. k → 0 /. hsystemsolve // FullSimplify
```

Out[668]=
$$\left\{\left(u\left(-2u^5\left(-3+u^2\right)\left(1+u^2\right)+9cu\left(-3-u^2+u^4\right)-9c\left(-3+u^2\right)\left(1+u^2\right)\text{ArcTan}[u]\right)\right.$$
$$\left.\text{Sin}[ru]\right)\Big/$$
$$\left(6\pi^2 r\left(2u^5\left(1+u^2\right)^4+3c^2u\left(3+u^2-u^4\right)-cu\left(1+u^2\right)\left(9+u^2\left(6+u^2\right)\left(4+3u^2\right)\right)+\right.\right.$$
$$\left.\left.3c\left(1+u^2\right)\left(c\left(-3+u^2\right)+3\left(1+u^2\right)^3\right)\text{ArcTan}[u]\right)\right)\}$$

result

In[669]:=
```
Ccexact[r_, c_] := NIntegrate[
  (u (-2 u^5 (-3 + u^2) (1 + u^2) + 9 c u (-3 - u^2 + u^4) - 9 c (-3 + u^2) (1 + u^2) ArcTan[u])
     Sin[r u]) / (6 π^2 r (2 u^5 (1 + u^2)^4 + 3 c^2 u (3 + u^2 - u^4) - c u (1 + u^2)
       (9 + u^2 (6 + u^2) (4 + 3 u^2)) + 3 c (1 + u^2) (c (-3 + u^2) + 3 (1 + u^2)^3) ArcTan[u])),
  {u, 0, Infinity}, Method → "LevinRule"]
```

load MC data

In[670]:=
```
ppoints[xs_, dr_, maxx_] :=
 Table[{dr (i) - 0.5 dr, xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[671]:=
```
ppointsu[xs_, du_, Σt_] :=
 Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[672]:=
```
fs = FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/MCdata/
     inf3D_isotropicpoint_rayleighscatter_gamma4C*"];
```

In[673]:=
```
index[x_] := Module[{data, c},
   data = Import[x, "Table"];
   c = data[[2, 3]];
   {c, data}];
simulations = index /@ fs;
cs = Union[#[[1]] & /@ simulations]
```

Out[675]=
$$\{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999\}$$

In[676]:=
```
numcollorders = simulations[[1]][[-1]][[2, 13]];
```

Compare analytic and MC

Collision-rate density - Exact solution (1) comparison to MC

In[633]:=
```
{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]}
```

Out[633]=
{Set c, 0.7}

In[677]:=
```
data = SelectFirst[simulations, #[[1]] == c &][[2]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
MCCollisionRate = ppoints[data[[4]], dr, maxr];
exact1CRShallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]}] & /@ MCCollisionRate[[1 ;; 60]];
exact1CR = Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]}] & /@
   MCCollisionRate[[61 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
    ListPlot[MCCollisionRate[[1 ;; 60]],
     PlotRange → All, PlotStyle → PointSize[.01]],
    ListPlot[exact1CRShallow, PlotRange → All, Joined → True],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[MCCollisionRate, PlotRange → All, PlotStyle → PointSize[.01]],
    ListLogPlot[exact1CR, PlotRange → All, Joined → True],
    ListLogPlot[exact1CRShallow, PlotRange → All, Joined → True],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Infinite 3D, isotropic point source,
     Rayleigh scattering, Gamma-4 random flight - correlated
     emission\nCollision-rate density Cc[r], c = " <> ToString[c]]
```

Out[685]=



Namespace

In[686]:=
```
End[]
```

Out[686]=
```
inf3DisopointRayleighscatterGamma4`
```