Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker’s Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Draine

Draine, B.T. (2003) ‘*Scattering by interstellar dust grains. 1: Optical and ultraviolet*’, ApJ., 598, 1017–25.

In[∘]:=
$$\text{pDraine}[u\_, g\_, \alpha\_] := \frac{1}{4\,\text{Pi}}\left(\frac{1-g^2}{\left(1+g^2-2\,g\,u\right)^{3/2}}\;\frac{1+\alpha\,u^2}{1+\alpha\left(1+2\,g^2\right)/3}\right)$$

Normalization condition

```
In[∘]:= Integrate[2 Pi pDraine[u, g, a], {u, -1, 1}, Assumptions → 0 < a < 1 && -1 < g < 1]
Out[∘]= 1
```

Mean-cosine

```
In[∘]:= Integrate[2 Pi pDraine[u, g, a] u, {u, -1, 1}, Assumptions → 0 < a < 1 && -1 < g < 1]
```

Out[∘]=
$$\frac{3}{5}\left(g+\frac{2\,(1+a)\,g}{3+a+2\,a\,g^2}\right)$$

In[∘]:=
$$\frac{3}{5}\left(g+\frac{2\,(1+a)\,g}{3+a+2\,a\,g^2}\right)\;/.\;a\to 0$$

Out[∘]= g

Legendre expansion coefficients

```
In[∘]:= Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
          {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
Out[∘]= 1
```

```
In[∘]:= Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
          {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```

Out[∘]=
$$\frac{9\,g\left(5+a\left(3+2\,g^2\right)\right)}{5\left(3+a+2\,a\,g^2\right)}$$

In[•]:=
```
Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
  {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```

Out[•]=

$$\frac{14\,a + 5\,(21 + 11\,a)\,g^2 + 36\,a\,g^4}{7\,(3 + a + 2\,a\,g^2)}$$

In[•]:=
```
Integrate[2 Pi (2 k + 1) pDraine[Cos[y], g, a] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
  {y, 0, Pi}, Assumptions → 0 < a < 1 && -1 < g < 1]
```

Out[•]=

$$\frac{g\,(54\,a + 7\,(45 + 23\,a)\,g^2 + 100\,a\,g^4)}{15\,(3 + a + 2\,a\,g^2)}$$

sampling

In[•]:=
```
cdf = Integrate[2 Pi pDraine[u, g, a],
   {u, -1, x}, Assumptions → 0 < a < 1 && -1 < g < 1 && -1 < x < 1]
```

Out[•]=

$$\Big(3\,(-1 + g)\,g^2\,\big(-1 - g + \sqrt{1 + g^2 - 2\,g\,x}\big) + a\,\big(2 - 2\,g^6 - 2\,g\,x - 2\,\sqrt{1 + g^2 - 2\,g\,x} + g^3\,\sqrt{1 + g^2 - 2\,g\,x} + g^4\,(-2 + x^2) + 2\,g^5\,\big(x + \sqrt{1 + g^2 - 2\,g\,x}\big) - g^2\,\big(-2 + x^2 + \sqrt{1 + g^2 - 2\,g\,x}\big)\big)\Big) \Big/ \Big(2\,g^3\,(3 + a + 2\,a\,g^2)\,\sqrt{1 + g^2 - 2\,g\,x}\Big)$$