Infinite 3D medium, Isotropic Point Source, Isotropic Scattering

Cauchy Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

```
In[•]:= SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



c - single-scattering albedo
r - radial position coordinate in medium (distance from point source at origin)
u = cos θ - direction cosine

Namespace

In[3002]:=
```
Begin["inf3DisopointIsoscatterCauchy`"]
```
Out[3002]=
```
inf3DisopointIsoscatterCauchy`
```

Analytical results

In[3059]:=
$$\texttt{pc[s\_] :=} \frac{2}{\texttt{Pi}\,(1+s^2)}$$

Collision rate density

collision rate density Cc due to correlated emission:

derivation

In[3003]:=
```
Clear[cpc, c];
```
$$\texttt{cpc[s\_] := c}\,\frac{2}{\texttt{Pi}\,(1+s^2)}$$

In[3010]:=
```
f00 = Fpc[0, 0, cpc, u];
```

In[3011]:=
```
o = 1;
Clear[A, b, c, r, h, F];
A[n_] := 0;
A[0] := 1;
A[1] := 0;
A[2] := 0;
hsystem =
```
$$\texttt{Table}\left[\texttt{h[k]} == \frac{2}{\texttt{Pi}}\,\texttt{u F[k, 0] + Sum[A[m]} \times \texttt{h[m]} \times \texttt{F[k, m], \{m, 0, o - 1\}], \{k, 0, o - 1\}}\right];$$
```
hsystemsolve = Simplify[
  Solve[hsystem, Table[h[i], {i, 0, o - 1}]] /. F[0, 0] → f00 /. F[0, 1] → -f10 /.
       F[1, 1] → f11 /. F[1, 0] → f10 /.
     F[2, 0] → f20 /. F[0, 2] → f20 /. F[2, 2] → f22]
```
Out[3018]=
$$\left\{\left\{h[0] \to -\frac{2\,c\,u\,(-1+\text{Cosh}[u]-\text{Sinh}[u])}{\pi\,(-c+u+c\,\text{Cosh}[u]-c\,\text{Sinh}[u])}\right\}\right\}$$

In[3019]:=
```
Clear[r, c];
```
$$(2\,k+1)\,\frac{1}{\texttt{4 Pi c}}\,(\texttt{h[k]})\;\texttt{u SphericalBesselJ[k, r u] /. k → 0 /. hsystemsolve //}$$
```
 FullSimplify
```
Out[3019]=
$$\left\{\frac{(-1+e^u)\,u^2\,\text{Sinc}[r\,u]}{2\,\pi^2\,(c-c\,e^u+e^u\,u)}\right\}$$

result

In[3020]:=
```
Ccexact[r_, c_] := NIntegrate[((-1 + ℯ^u) u^2 Sinc[r u])/(2 π^2 (c - c ℯ^u + ℯ^u u)),
  {u, 0, Infinity}, Method → "ExtrapolatingOscillatory"]
```

$$\mathrm{Ccexact}[r\_, c\_] := \mathrm{NIntegrate}\left[\frac{\left(-1 + e^{u}\right) u^2\, \mathrm{Sinc}[r\, u]}{2\, \pi^2 \left(c - c\, e^{u} + e^{u}\, u\right)}, \{u, 0, \mathrm{Infinity}\}, \mathrm{Method} \to \text{"ExtrapolatingOscillatory"}\right]$$

load MC data

In[3023]:=
```
ppoints[xs_, dr_, maxx_] :=
 Table[{dr (i) - 0.5 dr, xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[3024]:=
```
ppointsu[xs_, du_, Σt_] :=
 Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[3025]:=
```
fs = FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/MCdata/
    inf3D_isotropicpoint_isotropicscatter_cauchy*"];
```

In[3026]:=
```
index[x_] := Module[{data, c},
   data = Import[x, "Table"];
   c = data[[2, 3]];
   {c, data}];
simulations = index /@ fs;
cs = Union[#[[1]] & /@ simulations]
```

Out[3028]=
```
{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}
```

In[3029]:=
```
numcollorders = simulations[[1]][[-1]][[2, 13]];
```

Compare analytic and MC

Collision-rate density - Exact solution - comparison to MC

In[3030]:=
```
{ActionMenu["Set c", "c = " <> ToString[#] :→ (c = #;) & /@ cs], Dynamic[c]}
```

Out[3030]=
```
{ Set c , 0.5}
```

In[3040]:=
```
data = SelectFirst[simulations, #[[1]] == c &][[2]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
MCCollisionRate = ppoints[data[[4]], dr, maxr];
exact1CRShallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]} & /@ MCCollisionRate[[1 ;; 60]];
exact1CR = Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]}] & /@
   MCCollisionRate[[61 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
    ListPlot[MCCollisionRate[[1 ;; 60]],
     PlotRange → All, PlotStyle → PointSize[.01]],
    ListPlot[exact1CRShallow, PlotRange → All, Joined → True, PlotStyle → Black],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[MCCollisionRate, PlotRange → All, PlotStyle → PointSize[.01]],
    ListLogPlot[exact1CR, PlotRange → All, Joined → True, PlotStyle → Black],
    ListLogPlot[exact1CRShallow,
     PlotRange → All, Joined → True, PlotStyle → Black],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Infinite 3D, isotropic point source,
    Isotropic scattering, Cauchy random flight - correlated
    emission\nCollision-rate density Cc[r], c = " <> ToString[c]]
```

Out[3048]=



Namespace

In[3074]:=
```
End[]
```

Out[3074]=
```
inf3DisopointIsoscatterCauchy`
```