Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker’s Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Lambertian Sphere

geometrical optics far-field phase function of a white Lambertian sphere in 3D:
[Lambert 1760] - Photometria - Part VI - p470 - §1059
[Schoenberg 1929] - **doi**: 10.1007/978-3-642-90703-6_1
[Esposito and Lumme 1977, Blinn 1982, Porco et al. 2008]

In[∘]:=
```
pLambertSphere[u_] := (2 (Sqrt[1 - u^2] - u ArcCos[u]))/(3 π^2)
```

From Lambert’s work in latin:



Lambert’s table:

In[∘]:=
```
Table[{t, Pi pLambertSphere[Cos[t/180 Pi]]}, {t, 0, 180, 10}] // N // TableForm
```

```
0.`       0.`
10.`      0.0003749265567811918`
20.`      0.002972067797415395`
30.`      0.009878250529659252`
40.`      0.022915701605859425`
50.`      0.04352493713274053`
60.`      0.07266518736281957`
70.`      0.11073707843638177`
80.`      0.15753138394817434`
90.`      0.2122065907891938`
100.`     0.273296835726127 9`
110.`     0.33875050732016093`
120.`     0.4059985206961529`
130.`     0.47205001025710003`
140.`     0.5336119970185115`
150.`     0.5872285197192851`
160.`     0.629433814988021`
170.`     0.6569134285649197`
180.`     0.6666666666666666`
```

Elongatio ☽ a ☉ v	illuminatio plani a phaſe ☽ . c	Elongatio ☽ a ☉ v	illuminatio plani a phaſe ☽ , c
0°	0,0000	90°	0,2122
10	0,0004	100	0,2733
20	0,0030	110	0,3387
30	0,0099	120	0,4060
40	0,0229	130	0,4720
50	0,0435	140	0,5336
60	0,0727	150	0,5872
70	0,1107	160	0,6294
80	0,1576	170	0,6569
90	0,2122	180	0,6666

Thanks for Lionel Simonot for pointing me to Lambert's work for this phase function.

MC testing

Normalization condition

In[∘]:= `Integrate[2 Pi pLambertSphere[u], {u, -1, 1}]`

Out[∘]= 1

forward scattering probability

In[∘]:= `Clear[u]; Integrate[2 Pi pLambertSphere[u], {u, 0, 1}]`

Out[∘]= $\frac{1}{6}$

Mean cosine (g)

In[∘]:=
```
Integrate[2 Pi pLambertSphere[u] u, {u, -1, 1}]
```

Out[∘]= $-\frac{4}{9}$

Mean square cosine

In[∘]:=
```
Integrate[2 Pi pLambertSphere[u] u^2, {u, -1, 1}]
```

Out[∘]= $\frac{3}{8}$

This phase function is not particularly well approximated by Henyey Greenstein:

In[∘]:=
```
Show[
 Plot[pHG[Cos[t], -4/9], {t, -Pi, Pi}, PlotRange → {0, 1}],
 Plot[pLambertSphere[Cos[t]], {t, -Pi, Pi}, PlotRange → All]
]
```

Out[∘]=



Legendre expansion coefficients

In[∘]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
 {y, 0, Pi}]
```

Out[∘]= 1

In[∘]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
 {y, 0, Pi}]
```

Out[∘]= $-\frac{4}{3}$

In[∘]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
 {y, 0, Pi}]
```

Out[∘]= $\frac{5}{16}$

In[∘]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
  {y, 0, Pi}]
```
Out[∘]= 0

In[∘]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 4,
  {y, 0, Pi}]
```
Out[∘]= $\frac{1}{64}$

In[∘]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 6,
  {y, 0, Pi}]
```
Out[∘]= $\frac{13}{4096}$

In[∘]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 8,
  {y, 0, Pi}]
```
Out[∘]= $\frac{17}{16\,384}$

In[∘]:=
```
Integrate[2 Pi (2 k + 1) pLambertSphere[Cos[y]] LegendreP[k, Cos[y]] Sin[y] /. k → 10,
  {y, 0, Pi}]
```
Out[∘]= $\frac{343}{786\,432}$

Importance sampling:

The cosine of deflection can be sampled from:

In[∘]:=
```
Show[
 Histogram[Table[
   Sin[2 Pi RandomReal[]] √((1 - #1) (1 - #2)) - √(#1 #2) &[RandomReal[], RandomReal[]]
   , {i, Range[100 000]}], 50, "PDF"],
 Plot[2 Pi pLambertSphere[u], {u, -1, 1}]
]
```
Out[∘]=


Approximate CDF inverse:

In[•]:=

```
lambertSphereApproxCDFi[x_] := 1 - 2 (1 - x^(1.01938` + 0.0401885` x))^0.397225`
```

In[•]:=

```
Show[
 Histogram[Table[
   lambertSphereApproxCDFi[RandomReal[]]
   , {i, Range[100 000]}], 50, "PDF"],
 Plot[2 Pi pLambertSphere[u], {u, -1, 1}]
]
```

Out[•]=