Infinite 3D medium, Isotropic Point Source, Rayleigh Scattering

Gamma-3 Random Flight

This is code to accompany the book:

A Hitchhiker’s Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[ ]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



c - single-scattering albedo

Σt - extinction coefficient

r - radial position coordinate in medium (distance from point source at origin)

u = cos θ - direction cosine

b - anisotropy parameter

Namespace

In[595]:=
```
Begin["inf3DisopointRayleighscatterGamma3`"]
```

Out[595]=
```
inf3DisopointRayleighscatterGamma3`
```

Analytical results

Collision rate density

collision rate density Cc due to correlated emission:

derivation

In[596]:= $pc[s\_] := \frac{1}{2} Exp[-s]\, s^2$

In[611]:=
```
f00 = Fpc[0, 0, pc];
f01 = Fpc[0, 1, pc];
f11 = Fpc[1, 1, pc];
f20 = Fpc[2, 0, pc];
f22 = Fpc[2, 2, pc];
```

In[616]:=
```
o = 3;
Clear[A, b, c, r, h];
A[n_] := 0;
A[0] := 1;
A[1] := 0;
A[2] := 1/2;
hsystem = Table[
   h[k] == 2/Pi c u F[k, 0] + c Sum[A[m] × h[m] × F[k, m], {m, 0, o - 1}], {k, 0, o - 1}];
hsystemsolve = Simplify[Solve[hsystem, Table[h[i], {i, 0, o - 1}]] /. F[0, 0] → f00 /.
        F[0, 1] → f01 /. F[1, 1] → f11 /. F[1, 0] → -f01 /.
     F[2, 0] → f20 /. F[0, 2] → f20 /. F[2, 2] → f22]
```

Out[623]=
```
{{h[0] → -((2 c u (63 c u^2 + 8 u^6 - 36 c u (2 + u^2) ArcTan[u] + 9 c (1 + u^2)^2 ArcTan[u]^2)) /
        (π (-u^2 (-63 c^2 + 8 u^4 (1 + u^2)^2 + 6 c (9 + 16 u^2 + 5 u^4)) + 6 c u
              (-6 c (2 + u^2) + (1 + u^2)^2 (9 + u^2)) ArcTan[u] + 9 c^2 (1 + u^2)^2 ArcTan[u]^2))),
  h[1] → (4 c u^3 (4 u^5 (1 + u^2)^2 + 3 c (1 + u^2)^2 ArcTan[u]
           (-9 - u^2 + u F[1, 2] + 2 u^3 F[1, 2] + u^5 F[1, 2]) + c u
           (27 + 48 u^2 + 19 u^4 - 3 u F[1, 2] - 11 u^3 F[1, 2] - 13 u^5 F[1, 2] - 5 u^7 F[1, 2]))) /
      (π (1 + u^2)^2 (u^2 (-63 c^2 + 8 u^4 (1 + u^2)^2 + 6 c (9 + 16 u^2 + 5 u^4)) +
           6 c u (6 c (2 + u^2) - (1 + u^2)^2 (9 + u^2)) ArcTan[u] - 9 c^2 (1 + u^2)^2 ArcTan[u]^2)),
  h[2] → -((8 c u^4 (-u (3 + 5 u^2) + 3 (1 + u^2)^2 ArcTan[u])) /
        (π (-u^2 (-63 c^2 + 8 u^4 (1 + u^2)^2 + 6 c (9 + 16 u^2 + 5 u^4)) + 6 c u
              (-6 c (2 + u^2) + (1 + u^2)^2 (9 + u^2)) ArcTan[u] + 9 c^2 (1 + u^2)^2 ArcTan[u]^2)))}}
```

In[624]:=
```
Clear[r];
(2 k + 1) 1/(4 Pi r c) (h[k]) j2[k, r u] /. k → 0 /. hsystemsolve // FullSimplify
```

Out[624]=
```
{(u (63 c u^2 + 8 u^6 + 9 c ArcTan[u] (-4 u (2 + u^2) + (1 + u^2)^2 ArcTan[u])) Sin[r u]) /
  (2 π^2 r (u^2 (-63 c^2 + 8 u^4 (1 + u^2)^2 + 6 c (9 + 16 u^2 + 5 u^4)) +
       3 c ArcTan[u] (12 c u (2 + u^2) - 2 u (1 + u^2)^2 (9 + u^2) - 3 c (1 + u^2)^2 ArcTan[u])))}
```

result

In[625]:=
```
Ccexact[r_, c_] := NIntegrate[
  (u (63 c u^2 + 8 u^6 + 9 c ArcTan[u] (-4 u (2 + u^2) + (1 + u^2)^2 ArcTan[u])) Sin[r u]) /
   (2 π^2 r (u^2 (-63 c^2 + 8 u^4 (1 + u^2)^2 + 6 c (9 + 16 u^2 + 5 u^4)) +
        3 c ArcTan[u] (12 c u (2 + u^2) - 2 u (1 + u^2)^2 (9 + u^2) - 3 c (1 + u^2)^2 ArcTan[u]))),
  {u, 0, Infinity}, Method → "LevinRule"]
```

load MC data

```
In[626]:= ppoints[xs_, dr_, maxx_] :=
            Table[{dr (i) - 0.5 dr, xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]

In[627]:= ppointsu[xs_, du_, Σt_] :=
            Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]

In[628]:= fs = FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/MCdata/
                 inf3D_isotropicpoint_rayleighscatter_gamma3C*"];

In[629]:= index[x_] := Module[{data, c},
              data = Import[x, "Table"];
              c = data[[2, 3]];
              {c, data}];
          simulations = index /@ fs;
          cs = Union[#[[1]] & /@ simulations]

Out[631]= {0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}

In[632]:= numcollorders = simulations[[1]][[-1]][[2, 13]];
```

Compare analytic and MC

Collision-rate density - Exact solution (1) comparison to MC

```
In[633]:= {ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]}

Out[633]= {Set c, 0.7}
```

In[643]:=
```
data = SelectFirst[simulations, #[[1]] == c &][[2]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
MCCollisionRate = ppoints[data[[4]], dr, maxr];
exact1CRShallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]} & /@ MCCollisionRate[[1 ;; 60]];
exact1CR = Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]}] & /@
   MCCollisionRate[[61 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
     ListPlot[MCCollisionRate[[1 ;; 60]],
      PlotRange → All, PlotStyle → PointSize[.01]],
     ListPlot[exact1CRShallow, PlotRange → All, Joined → True],
     Frame → True,
     FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
     ListLogPlot[MCCollisionRate, PlotRange → All, PlotStyle → PointSize[.01]],
     ListLogPlot[exact1CR, PlotRange → All, Joined → True],
     ListLogPlot[exact1CRShallow, PlotRange → All, Joined → True],
     Frame → True,
     FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Infinite 3D, isotropic point source,
    Rayleigh scattering, Gamma-3 random flight - correlated
    emission\nCollision-rate density Cc[r], c = " <> ToString[c]]
```

Out[651]=



Namespace

In[652]:=
```
End[]
```

Out[652]=
```
inf3DisopointRayleighscatterGamma3`
```