Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker’s Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Gegenbauer Scattering

In[ ]:= pGegenbauer[u_, g_, a_] :=

$$\frac{\left(1 + g^2 - 2\,g\,u\right)^{-(a+1)}}{\frac{\left((1-g)^{-2\,a} - (1+g)^{-2\,a}\right)\pi}{a\,g}}$$

```
Show[
 Plot[pGegenbauer[Cos[t], 0.5, 1], {t, -Pi, Pi}, PlotRange → All],
 Plot[pGegenbauer[Cos[t], 0.5, 3], {t, -Pi, Pi}, PlotRange → All],
 Plot[pGegenbauer[Cos[t], 0.5, 5], {t, -Pi, Pi}, PlotRange → All],

 Frame → True,
 FrameLabel →
  {{p[Cos[θ]],}, {θ, "Gegenbauer Scattering, G = 0.5, a = 1, 3, 5"}}]
```



Normalization condition

```
Integrate[2 Pi pGegenbauer[u, g, a], {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]
1
```

Mean cosine (g)

```
Integrate[2 Pi u pGegenbauer[u, g, a], {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]
```

$$\frac{(1+g)^{2a}\left(1-2ag+g^2\right)-(1-g)^{2a}\left(1+2ag+g^2\right)}{2(-1+a)g\left((1-g)^{2a}-(1+g)^{2a}\right)}$$

Legendre expansion coefficients

```
Integrate[2 Pi (2 k + 1) pGegenbauer[u, g, a] LegendreP[k, u] /. k → 0,
 {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]
```

1

```
FullSimplify[Integrate[2 Pi (2 k + 1) pGegenbauer[u, g, a] LegendreP[k, u] /. k → 3,
  {u, -1, 1}, Assumptions → -1 ≤ g ≤ 1 && a > 0]]
```

$$-\left(7\left(24a^2g^2\left(1+g^2\right)\left((1-g)^{2a}-(1+g)^{2a}\right)+3\left(5+3g^2+3g^4+5g^6\right)\left((1-g)^{2a}-(1+g)^{2a}\right)+8a^3g^3\left((1-g)^{2a}+(1+g)^{2a}\right)+2ag\left(15+14g^2+15g^4\right)\left((1-g)^{2a}+(1+g)^{2a}\right)\right)\right)\Big/\left(8(-3+a)(-2+a)(-1+a)g^3\left((1-g)^{2a}-(1+g)^{2a}\right)\right)$$

sampling

```
g = -0.8;
a = -1.2;
Show[Histogram[Map[(1 + g^2 - (# (1 - g)^(-2 a) - (-1 + #) (1 + g)^(-2 a))^(-1/a))/(2 g) &,
    Table[RandomReal[], {i, 1, 100 000}]], 100, "PDF"],
  Plot[2 Pi pGegenbauer[u, g, a], {u, -1, 1}, PlotRange → All]

]
Clear[g, a];
```



When cosine u has been sampled with random variable ξ, what is the PDF at the sampled direction in terms of ξ?

In[∘]:=
```
FullSimplify[pGegenbauer[(1 + g^2 - (# (1 - g)^(-2 a) - (-1 + #) (1 + g)^(-2 a))^(-1/a))/(2 g) &[ξ], g, a],
  Assumptions → a > 0 && -1 < g < 1 && 0 < ξ < 1]
```

Out[∘]=

$$\frac{a\, g \left(\left(-(1+g)^{-2a}(-1+\xi)+(1-g)^{-2a}\,\xi\right)^{-1/a}\right)^{-1-a}}{\left((1-g)^{-2a}-(1+g)^{-2a}\right)\pi}$$