K_0 NDF

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

notation

u = **m** . **n** = Cos[θ_m]

α = roughness

Definitions and derivations

In[4508]:=
$$\text{K0`D}[u\_, \alpha\_] := \frac{2\,\text{BesselK}\left[0, \frac{2\sqrt{1-u^2}}{u\,\alpha}\right]}{\pi\,u^4\,\alpha^2}\,\text{HeavisideTheta}[u]$$

In[1414]:=
$$\text{K0`}\sigma[u\_, \alpha\_] := u\left(1 + \frac{e^{-\frac{2u}{\sqrt{1-u^2}\,\alpha}}\sqrt{1-u^2}\,\alpha}{4\,u}\right)\text{HeavisideTheta}[u] + \text{HeavisideTheta}[-u]\,\frac{1}{4}\,e^{\frac{2u}{\sqrt{1-u^2}\,\alpha}}\sqrt{1-u^2}\,\alpha$$

In[2650]:=
$$\text{K0`}\Lambda[u\_, \alpha\_] := \frac{e^{-\frac{2u}{\sqrt{1-u^2}\,\alpha}}\sqrt{1-u^2}\,\alpha}{4\,u}$$

In[2651]:=
$$\text{FullSimplify}\left[\text{K0`}\Lambda\left[u, \frac{u}{\sqrt{1-u^2}\,x}\right], \text{Assumptions} \to 0 < u < 1 \,\&\&\, x > 0\right]$$

Out[2651]=
$$\frac{e^{-2x}}{4\,x}$$

derivation

In[2751]:=
$$\text{Beckmann`D}[u\_, \alpha\_] := \frac{e^{-\frac{-1+\frac{1}{u^2}}{\alpha^2}}}{\alpha^2\,\pi\,u^4}\,\text{HeavisideTheta}[u]$$

In[2752]:=
$$\text{Integrate}\left[\text{Beckmann`D}\left[u, \alpha\sqrt{m}\right]\,\text{Exp}[-m],\ \{m, 0, \text{Infinity}\}, \text{Assumptions} \to 0 < u < 1 \,\&\&\, \alpha > 0\right]$$

Out[2752]=
$$\frac{2\,\text{BesselK}\left[0, \frac{2\sqrt{1-u^2}}{u\,\alpha}\right]}{\pi\,u^4\,\alpha^2}$$

In[2125]:= `Plot[{K0`D[u, m], Beckmann`D[u, m]} /. m → .5 // Evaluate, {u, 0, 1}]`

General: Exp[−9.58482 × 10^9] is too small to represent as a normalized machine number; precision may be lost.

Out[2125]=



The NDF is singular at u = 1 - no matter how rough you make it, it still has arbitrarily large BRDF values near the specular reflection direction and so always remains shiny.

In[2124]:= `Limit[K0`D[u, a], u → 1, Assumptions → a > 0, Direction → "FromBelow"]`

Out[2124]= ∞

shape invariant f(x)

In[1373]:= $\text{FullSimplify}\left[\text{K0`D}[u, \alpha]\, u^4 \alpha^2 \text{ /. } u \to \frac{1}{\sqrt{1 + x^2 \alpha^2}}, \text{Assumptions} \to 1 - \frac{1}{\sqrt{1 + x^2 \alpha^2}} > 0 \,\&\&\, x > 0 \,\&\&\, \alpha > 0\right]$

Out[1373]= $\frac{2\,\text{BesselK}[0, 2\,x]}{\pi}$

height field normalization

In[1376]:= `NIntegrate[2 Pi u K0`D[u, .6], {u, 0, 1}]`

Out[1376]= 1.

distribution of slopes

In[1379]:= $\text{FullSimplify}\left[\text{K0`D}\left[\frac{1}{\sqrt{p^2 + q^2 + 1}}, \alpha\right] \left(\frac{1}{\sqrt{p^2 + q^2 + 1}}\right)^4, \text{Assumptions} \to 0 < \alpha < 1 \,\&\&\, p > 0 \,\&\&\, q > 0\right]$

Out[1379]= $\frac{2\,\text{BesselK}\left[0, \frac{2\sqrt{p^2+q^2}}{\alpha}\right]}{\pi\,\alpha^2}$

In[1380]:= $\text{K0`P22}[p\_, q\_, \alpha\_] := \frac{2\,\text{BesselK}\left[0, \frac{2\sqrt{p^2+q^2}}{\alpha}\right]}{\pi\,\alpha^2}$

In[1381]:=
```
Integrate[K0`P22[p, q, α], {p, -Infinity, Infinity},
 {q, -Infinity, Infinity}, Assumptions → 0 < α < 1]
```

Out[1381]= 1

compare σ to delta integral:

In[1383]:=
$$\text{Delta`}\sigma[\text{u\_}, \text{ui\_}] := \text{Re}\left[2\left(\sqrt{1-u^2-ui^2} + u\,ui\,\text{ArcCos}\left[-\frac{u\,ui}{\sqrt{1-u^2}\,\sqrt{1-ui^2}}\right]\right)\right]$$

In[1415]:=
```
With[{α = .8},
 Plot[{
   Quiet[NIntegrate[K0`D[ui, α] × Delta`σ[u, ui], {ui, 0, 1}]],
   Quiet[K0`σ[u, α]]
  }, {u, -1, 1}]
]
```

Out[1415]=



importance sampling

In[4511]:=
```
With[{α = 1.7},
 Show[
  Histogram[Table[1/(√(1 - (α √(-Log[RandomReal[]]))^2 (Log[RandomReal[]]))),
    {i, Range[100 000]}], 200, "PDF"],
  Plot[K0`D[u, α] 2 Pi u, {u, 0, .999}, PlotRange → All]
 ]
]
```

Out[4511]=