Infinite 3D medium, Isotropic Point Source, Linearly-Anisotropic Scattering

Gamma-2 Random Flight

This is code to accompany the book:

A Hitchhiker’s Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



c - single-scattering albedo
Σt - extinction coefficient

r - radial position coordinate in medium (distance from point source at origin)
u = cos θ - direction cosine
b - anisotropy parameter

Namespace

In[9393]:=
```
Begin["inf3DisopointlinanisoscatterGamma2`"]
```
Out[9393]=
```
inf3DisopointlinanisoscatterGamma2`
```

Analytical results

Collision rate density

collision rate density Cc due to correlated emission:

derivation

In[9202]:=
```
f00 = Fpc[0, 0, Exp[-#] # &];
f01 = Fpc[0, 1, Exp[-#] # &];
f11 = Fpc[1, 1, Exp[-#] # &];
o = 2;
Clear[A, b, c, r, h];
A[0] := 1; A[1] := b;
hsystem = Table[
   h[k] == 2/Pi c u F[k, 0] + c Sum[A[m] × h[m] × F[k, m], {m, 0, o - 1}], {k, 0, o - 1}];
hsystemsolve = Simplify[Solve[hsystem, Table[h[i], {i, 0, o - 1}]] /. A[1] → b /.
     F[0, 0] → f00 /. F[0, 1] → f01 /. F[1, 1] → f11 /. F[1, 0] → - f01]
```
Out[9209]=
```
{{h[0] →
   (2 c u (b c u^2 + u^4 - b c (1 + u^2) ArcTan[u]^2)) / (π (u^4 (1 - c + u^2) + b c u^2 (2 - c + u^2) -
        2 b c u (1 + u^2) ArcTan[u] + b c^2 (1 + u^2) ArcTan[u]^2)),
  h[1] → (2 c u^3 (-u + (1 + u^2) ArcTan[u])) / (π (u^4 (1 - c + u^2) + b c u^2 (2 - c + u^2) -
        2 b c u (1 + u^2) ArcTan[u] + b c^2 (1 + u^2) ArcTan[u]^2))}}
```

In[9193]:=
```
Clear[r];
(2 k + 1) 1/(4 Pi r c) (h[k]) j2[k, r u] /. k → 0 /. hsystemsolve // FullSimplify
```
Out[9193]=
```
{(u (b c u^2 + u^4 - b c (1 + u^2) ArcTan[u]^2) Sin[r u]) /
  (2 π^2 r (-b (-2 + c) c u^2 + (1 + (-1 + b) c) u^4 +
       u^6 + b c (1 + u^2) ArcTan[u] (-2 u + c ArcTan[u])))}
```

result

In[9437]:=
```
Ccexact[r_, t_, c_, b_] :=
 NIntegrate[(u (b c u^2 + u^4 - b c (1 + u^2) ArcTan[u]^2) Sin[r u]) /
   (2 π^2 r (-b (-2 + c) c u^2 + (1 + (-1 + b) c) u^4 + u^6 + b c (1 + u^2) ArcTan[u]
        (-2 u + c ArcTan[u]))), {u, 0, Infinity}, Method → "LevinRule"]
```

In[9159]:=
```
TraditionalForm[
 HoldForm[Cc[r] = Integrate[(u (b c u^2 + u^4 - b c (1 + u^2) ArcTan[u]^2) Sin[r u]) /
     (2 π^2 r (-b (-2 + c) c u^2 + (1 + (-1 + b) c) u^4 + u^6 +
          b c (1 + u^2) ArcTan[u] (-2 u + c ArcTan[u]))), {u, 0, Infinity}]]]
```

Out[9159]//TraditionalForm=

$$C_c(r) = \int_0^\infty \frac{u\left(b\,c\,u^2 + u^4 - b\,c\left(1+u^2\right)\tan^{-1}(u)^2\right)\sin(r\,u)}{2\,\pi^2\,r\left(-b\,(-2+c)\,c\,u^2 + (1+(-1+b)\,c)\,u^4 + u^6 + b\,c\left(1+u^2\right)\tan^{-1}(u)\left(-2\,u + c\tan^{-1}(u)\right)\right)}\,du$$

load MC data

In[9438]:=
```
ppoints[xs_, dr_, maxx_] :=
 Table[{dr (i) - 0.5 dr, xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[9439]:=
```
ppointsu[xs_, du_, Σt_] :=
 Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[9440]:=
```
fs = FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/MCdata/
     inf3D_isotropicpoint_linanisoscatter_gamma2C*"];
```

In[9441]:=
```
index[x_] := Module[{data, c, mfp, b},
   data = Import[x, "Table"];
   mfp = data[[1, 13]];
   c = data[[2, 3]];
   b = data[[1, 16]];
   {c, mfp, b, data}];
simulations = index /@ fs;
cs = Union[#[[1]] & /@ simulations]
```

Out[9443]=
```
{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}
```

In[9444]:=
```
mfps = Union[#[[2]] & /@ simulations]
```

Out[9444]=
```
{0.3, 1}
```

In[9445]:=
```
bs = Union[#[[3]] & /@ simulations]
```

Out[9445]=
```
{-0.9, 0.7}
```

In[9446]:=
```
numcollorders = inf3Disopointlinanisoscatter`simulations[[1]][[-1]][[2, 13]];
```

Compare analytic and MC

Collision-rate density - Exact solution (1) comparison to MC

In[9447]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]},
 {ActionMenu["Set mfp", "mfp = " <> ToString[#] :> (mfp = #;) & /@ mfps],
  Dynamic[mfp]},
 {ActionMenu["Set b", "b = " <> ToString[#] :> (b = #;) & /@ bs], Dynamic[b]}}
```

Out[9447]=
```
{{Set c, 0.95}, {Set mfp, 1}, {Set b, -0.9}}
```

In[9457]:=
```
data = SelectFirst[simulations, #[[1]] == c && #[[2]] == mfp && #[[3]] == b &][[4]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
MCCollisionRate = ppoints[data[[4]], dr, maxr];
exact1CRShallow = Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], 1/mfp, c, b]}] & /@
   MCCollisionRate[[1 ;; 60]];
exact1CR = Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], 1/mfp, c, b]}] & /@
   MCCollisionRate[[61 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
     ListPlot[MCCollisionRate[[1 ;; 60]],
      PlotRange → All, PlotStyle → PointSize[.01]],
     ListPlot[exact1CRShallow, PlotRange → All, Joined → True],
     Frame → True,
     FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
     ListLogPlot[MCCollisionRate, PlotRange → All, PlotStyle → PointSize[.01]],
     ListLogPlot[exact1CR, PlotRange → All, Joined → True],
     ListLogPlot[exact1CRShallow, PlotRange → All, Joined → True],
     Frame → True,
     FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Exact solution (1)\nInfinite 3D, isotropic point source,
    linearly-anisotropic scattering, Gamma-2 random flight -
    correlated emission\nCollision-rate density Cc[r], c = " <>
   ToString[c] <> ", b = " <> ToString[b]]
```

Out[9465]=

Exact solution (1)
Infinite 3D, isotropic point source, linearly–anisotropic scattering, Gamma–2 random flight – correlated
Collision–rate density $C_c[r]$, c = 0.9, b = 0.7



Namespace

In[9392]:=
```
End[]
```
Out[9392]=
```
inf3DisopointlinanisoscatterGamma2`
```