Scattering Kernels in 3D

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Klein-Nishina

Normalized variant of Klein-Nishina - energy parameter "e" = $\frac{E_\gamma}{m_e c^2}$

```
pKleinNishina[u_, e_] := 1/(1 + e (1 - u)) 1/((2 π Log[1+2 e])/e)
```

Normalization condition

```
In[•]:= Integrate[2 Pi pKleinNishina[u, e], {u, -1, 1}, Assumptions → e > 0]
Out[•]= 1
```

Mean-cosine

```
In[•]:= Integrate[2 Pi pKleinNishina[u, e] u, {u, -1, 1}, Assumptions → e > 0]
```

Out[•]= $1 + \frac{1}{e} - \frac{2}{\text{Log}[1 + 2\,e]}$

Legendre expansion coefficients

```
In[•]:= Integrate[
         2 Pi (2 k + 1) pKleinNishina[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 0,
         {y, 0, Pi}, Assumptions → e > 0]
Out[•]= 1
```

```
In[•]:= Integrate[
         2 Pi (2 k + 1) pKleinNishina[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 1,
         {y, 0, Pi}, Assumptions → e > 0]
```

Out[•]= $3 + \frac{3}{e} - \frac{6}{\text{Log}[1 + 2\,e]}$

In[∘]:=
```
Integrate[
 2 Pi (2 k + 1) pKleinNishina[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 2,
 {y, 0, Pi}, Assumptions → e > 0]
```

Out[∘]=
$$\frac{5}{4}\left(1+\frac{3\left(2+4\,e+e^{2}-\frac{4\,e\,(1+e)}{\mathrm{Log}[1+2\,e]}\right)}{e^{2}}\right)$$

In[∘]:=
```
Integrate[
 2 Pi (2 k + 1) pKleinNishina[Cos[y], e] LegendreP[k, Cos[y]] Sin[y] /. k → 3,
 {y, 0, Pi}, Assumptions → e > 0]
```

Out[∘]=
$$\frac{7\left(15+45\,e+36\,e^{2}+6\,e^{3}-\frac{2\,e\,(15+30\,e+11\,e^{2})}{\mathrm{Log}[1+2\,e]}\right)}{6\,e^{3}}$$

sampling

In[∘]:=
```
cdf = Integrate[2 Pi pKleinNishina[u, e], {u, -1, x}, Assumptions → e > 0 && 0 < x < 1]
```

Out[∘]=
$$1-\frac{\mathrm{Log}[1+e-e\,x]}{\mathrm{Log}[1+2\,e]}$$

In[∘]:=
```
Solve[cdf == k, x]
```

Out[∘]=
$$\left\{\left\{x\to \mathrm{ConditionalExpression}\left[\frac{1+e-(1+2\,e)^{1-k}}{e},\ -\pi\le \mathrm{Im}\left[(-1+k)\,\mathrm{Log}[1+2\,e]\right]<\pi\right]\right\}\right\}$$

In[∘]:=
```
With[{e = 1.1},

 Show[
  Plot[2 Pi pKleinNishina[u, e], {u, -1, 1}],
  Histogram[
   Map[(1 + e - (1 + 2 e)^(1-#))/e &, Table[RandomReal[], {i, 1, 100 000}]], 50, "PDF"]
 ]
]
```

Out[∘]=