Infinite 3D medium, Isotropic Point Source, Isotropic Scattering

Chi-3 Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

© 2020 Eugene d'Eon

www.eugenedeon.com/hitchhikers

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[4478]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Notation



c - single-scattering albedo
r - radial position coordinate in medium (distance from point source at origin)
u = cos θ - direction cosine

Namespace

In[4479]:=
```
Begin["inf3DisopointIsotropcscatterChi3`"]
```
Out[4479]=
```
inf3DisopointIsotropcscatterChi3`
```

Analytical results

Collision rate density

collision rate density Cc due to correlated emission:

derivation

In[4480]:=
```
Clear[cpc, c];
cpc[s_] := c e^(-s^2/4) s^2 / (2 Sqrt[π])
```

In[3537]:=
```
f00 = Fpc[0, 0, cpc, u];
```

In[3538]:=
```
o = 1;
Clear[A, b, c, r, h, F];
A[n_] := 0;
A[0] := 1;
A[1] := 0;
A[2] := 0;
hsystem =
  Table[h[k] == 2/Pi u F[k, 0] + Sum[A[m] × h[m] × F[k, m], {m, 0, o - 1}], {k, 0, o - 1}];
hsystemsolve = Simplify[
  Solve[hsystem, Table[h[i], {i, 0, o - 1}]] /. F[0, 0] → f00 /. F[0, 1] → -f10 /.
      F[1, 1] → f11 /. F[1, 0] → f10 /.
    F[2, 0] → f20 /. F[0, 2] → f20 /. F[2, 2] → f22]
```
Out[3545]=
$$\left\{\left\{h[0] \to -\frac{2\,c\,u}{c\,\pi - e^{u^2}\,\pi}\right\}\right\}$$

In[3546]:=
```
Clear[r, c];
First[(2 k + 1) 1/(4 Pi c) (h[k]) u SphericalBesselJ[k, r u] /. k → 0 /. hsystemsolve //
  FullSimplify]
```
Out[3546]=
$$\frac{u^2\,\text{Sinc}[r\,u]}{2\left(-c + e^{u^2}\right)\pi^2}$$

result

In[4482]:=
```
Ccexact[r_, c_] := NIntegrate[u^2 Sinc[r u] / (2 (-c + e^(u^2)) π^2), {u, 0, Infinity}]
```

In[4513]:=
$$\text{Ccexact}[r\_, c\_, n\_] := c^{n-1}\,\frac{e^{-\frac{r^2}{4n}}}{8\,n^{3/2}\,\pi^{3/2}}$$

load MC data

In[4485]:=
```
ppoints[xs_, dr_, maxx_] :=
 Table[{dr (i) - 0.5 dr, xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[4486]:=
```
ppointsu[xs_, du_, Σt_] :=
 Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[4487]:=
```
fs = FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/MCdata/
      inf3D_isotropicpoint_isotropicscatter_chi3_*"];
```

In[4488]:=
```
index[x_] := Module[{data, c},
   data = Import[x, "Table"];
   c = data[[2, 3]];
   {c, data}];
simulations = index /@ fs;
cs = Union[#[[1]] & /@ simulations]
```

Out[4490]=
```
{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}
```

In[4491]:=
```
numcollorders = simulations[[1]][[-1]][[2, 13]];
```

Compare analytic and MC

Collision-rate density - Exact solution - comparison to MC

In[4502]:=
```
{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]}
```

Out[4502]=


I suspect the Chi3sample() is incorrect:

In[3689]:=
```
data = SelectFirst[simulations, #[[1]] == c &][[2]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
MCCollisionRate = ppoints[data[[4]], dr, maxr];
exact1CRShallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]}] & /@ MCCollisionRate[[1 ;; 60]];
exact1CR = Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]}] & /@
   MCCollisionRate[[61 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
    ListPlot[MCCollisionRate[[1 ;; 60]],
     PlotRange → All, PlotStyle → PointSize[.01]],
    ListPlot[exact1CRShallow, PlotRange → All, Joined → True, PlotStyle → Black],
    Plot[4 Pi r^2 Sum[Ccexact[r, c, n], {n, 1, 24}],
     {r, 0, 3.5}, PlotStyle → Dashed],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[MCCollisionRate, PlotRange → All, PlotStyle → PointSize[.01]],
    ListLogPlot[exact1CR, PlotRange → All, Joined → True, PlotStyle → Black],
    ListLogPlot[exact1CRShallow,
     PlotRange → All, Joined → True, PlotStyle → Black],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Infinite 3D, isotropic point source,
     Isotropic scattering, Chi-3 random flight - correlated
     emission\nCollision-rate density Cc[r], c = " <> ToString[c]]
```

Out[3697]=



nth collision

In[4503]:=
```
{{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]},
  {ActionMenu["Set collision order",
    "collisionOrder = " <> ToString[#] :> (collisionOrder = #;) & /@
     Range[0, numcollorders - 1]], Dynamic[collisionOrder]}} // TableForm
```

Out[4503]//TableForm=

Set c	0.7
Set collision order	3

In[4514]:=
```
Ci = 2 numcollorders + 14 + collisionOrder;
data = SelectFirst[simulations, #[[1]] == c &][[2]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
MCCollisionRate = ppoints[data[[Ci]], dr, maxr];
plotϕshallow = Quiet[Show[
    ListPlot[MCCollisionRate[[1 ;; 60]],
     PlotRange → All, PlotStyle → PointSize[.01]],
    Plot[4 Pi r^2 Ccexact[r, c, collisionOrder],
     {r, 0, MCCollisionRate[[60, 1]]}, PlotStyle → Black],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[MCCollisionRate, PlotRange → All, PlotStyle → PointSize[.01]],
    LogPlot[4 Pi r^2 Ccexact[r, c, collisionOrder],
     {r, 0, maxr}, PlotStyle → Black],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Infinite 3D, isotropic point source,
     Isotropic scattering, Chi-3 random flight - correlated
     emission\nCollision-rate density Cc[r], c = " <> ToString[c]]
```

Out[4521]=



Moments

In[4537]:=
```
Clear[pc]; pc[s_] := e^(-s^2/4) s^2 / (2 √π)
```

correlation emission

collision rate

In[4538]:=
```
m0Cc[c_] := 1/(1 - c)
```

In[4539]:=
```
m2Cc[c_, s_, s2_, g_] := s2/(1 - c)^2 (1 + c g 2 s^2/(s2 (1 - c g)))
```

fluence

In[4540]:=
```
m0ϕc[c_, s_] := s/(1 - c)
```

In[4541]:=
```
m2ϕc[c_, s_, s2_, s3_, g_] := (s3 (1 - c) (1 - g c) + 3 c s (2 g c s^2 + s2 (-2 g c + g + 1)))/(3 (1 - c)^2 (1 - c g))
```

In[4542]:=
```
simsC = simulations;
```

In[4543]:=
```
m0Ccs = {#[[1]], #[[-1, 8, 1]]} & /@ simsC;
m2Ccs = {#[[1]], #[[-1, 8, 3]]} & /@ simsC;
m0ϕcs = {#[[1]], #[[-1, 10, 1]]} & /@ simsC;
m2ϕcs = {#[[1]], #[[-1, 10, 3]]} & /@ simsC;
mfp = Integrate[pc[s] s, {s, 0, Infinity}];
mfp2 = Integrate[pc[s] s s, {s, 0, Infinity}];
mfp3 = Integrate[pc[s] s s s, {s, 0, Infinity}];
g = 0;
Show[
 LogPlot[{m0Cc[c], m2Cc[c, mfp, mfp2, g], m0ϕc[c, mfp],
   m2ϕc[c, mfp, mfp2, mfp3, g]}, {c, 0.01, 0.999}, PlotRange → All],
 ListLogPlot[{m0Ccs, m2Ccs, m0ϕcs, m2ϕcs}, PlotRange → All],
 PlotRange → All, Frame → True,
 FrameLabel → {{"moments",}, {c, "Chi-3 Correlated Emission 3D Isotropic
      scattering\nCollision-rate density moments"}}
]
```

Out[4551]=



Namespace

In[4552]:=
```
End[]
```

Out[4552]=
```
inf3DisopointIsotropcscatterChi3`
```