Infinite 3D medium, Isotropic Point Source, Isotropic Scattering

Chi-3 Random Flight

This is code to accompany the book:

A Hitchhiker's Guide to Multiple Scattering

www.eugenedeon.com/hitchhikers

Path Setup

Put a file at ~/.hitchhikerpath with the path to your hitchhiker repo so that these worksheets can find the MC data from the C++ simulations for verification

In[3533]:=
```
SetDirectory[Import["~/.hitchhikerpath"]]
```

Out[3533]=
```
/Users/eug/Documents/research/hitchhikersscatter
```

Notation



c - single-scattering albedo
r - radial position coordinate in medium (distance from point source at origin)
u = cos θ - direction cosine

Namespace

In[3534]:=
```
Begin["inf3DisopointIsotropcscatterChi3`"]
```
Out[3534]=
```
inf3DisopointIsotropcscatterChi3`
```

Analytical results

Collision rate density

collision rate density Cc due to correlated emission:

derivation

In[3535]:=
```
Clear[cpc, c];
cpc[s_] := c e^(-s^2/4) s^2 / (2 Sqrt[π])
```

In[3537]:=
```
f00 = Fpc[0, 0, cpc, u];
```

In[3538]:=
```
o = 1;
Clear[A, b, c, r, h, F];
A[n_] := 0;
A[0] := 1;
A[1] := 0;
A[2] := 0;
hsystem =
  Table[h[k] == 2/Pi u F[k, 0] + Sum[A[m] × h[m] × F[k, m], {m, 0, o - 1}], {k, 0, o - 1}];
hsystemsolve = Simplify[
  Solve[hsystem, Table[h[i], {i, 0, o - 1}]] /. F[0, 0] → f00 /. F[0, 1] → -f10 /.
      F[1, 1] → f11 /. F[1, 0] → f10 /.
    F[2, 0] → f20 /. F[0, 2] → f20 /. F[2, 2] → f22]
```
Out[3545]=
$$\left\{\left\{h[0] \to -\frac{2\,c\,u}{c\,\pi - e^{u^2}\,\pi}\right\}\right\}$$

In[3546]:=
```
Clear[r, c];
First[(2 k + 1) 1/(4 Pi c) (h[k]) u SphericalBesselJ[k, r u] /. k → 0 /. hsystemsolve //
  FullSimplify]
```
Out[3546]=
$$\frac{u^2\,\text{Sinc}[r\,u]}{2\left(-c + e^{u^2}\right)\pi^2}$$

result

In[3547]:=
```
Ccexact[r_, c_] := NIntegrate[u^2 Sinc[r u] / (2 (-c + e^(u^2)) π^2), {u, 0, Infinity}]
```

load MC data

In[3548]:=
```
ppoints[xs_, dr_, maxx_] :=
 Table[{dr (i) - 0.5 dr, xs[[i]]}, {i, 1, Length[xs]}][[1 ;; -2]]
```

In[3549]:=
```
ppointsu[xs_, du_, Σt_] :=
 Table[{-1.0 + du (i) - 0.5 du, xs[[i]] / (2 Σt)}, {i, 1, Length[xs]}][[1 ;; -1]]
```

In[3550]:=
```
fs = FileNames["code/3D_medium/infinite3Dmedium/Isotropicpointsource/MCdata/
     inf3D_isotropicpoint_isotropicscatter_chi3_*"];
```

In[3551]:=
```
index[x_] := Module[{data, c},
   data = Import[x, "Table"];
   c = data[[2, 3]];
   {c, data}];
simulations = index /@ fs;
cs = Union[#[[1]] & /@ simulations]
```

Out[3553]=
```
{0.01, 0.1, 0.3, 0.5, 0.7, 0.8, 0.9, 0.95, 0.99, 0.999}
```

In[3554]:=
```
numcollorders = simulations[[1]][[-1]][[2, 13]];
```

Compare analytic and MC

Collision-rate density - Exact solution - comparison to MC

In[3555]:=
```
{ActionMenu["Set c", "c = " <> ToString[#] :> (c = #;) & /@ cs], Dynamic[c]}
```

Out[3555]=
```
{Set c, 0.99}
```

I suspect the Chi3sample() is incorrect:

In[3565]:=
```
data = SelectFirst[simulations, #[[1]] == c &][[2]];
maxr = data[[2, 5]];
dr = data[[2, 7]];
MCCollisionRate = ppoints[data[[4]], dr, maxr];
exact1CRShallow =
  Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]}] & /@ MCCollisionRate[[1 ;; 60]];
exact1CR = Quiet[{#[[1]], 4 Pi #[[1]]^2 Ccexact[#[[1]], c]}] & /@
   MCCollisionRate[[61 ;; -1 ;; 10]];
plotϕshallow = Quiet[Show[
    ListPlot[MCCollisionRate[[1 ;; 60]],
     PlotRange → All, PlotStyle → PointSize[.01]],
    ListPlot[exact1CRShallow, PlotRange → All, Joined → True, PlotStyle → Black],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
logplotϕ = Quiet[Show[
    ListLogPlot[MCCollisionRate, PlotRange → All, PlotStyle → PointSize[.01]],
    ListLogPlot[exact1CR, PlotRange → All, Joined → True, PlotStyle → Black],
    ListLogPlot[exact1CRShallow,
     PlotRange → All, Joined → True, PlotStyle → Black],
    Frame → True,
    FrameLabel -> {{4 π r^2 C"c"[r],}, {r,}}
   ]];
Show[GraphicsGrid[{{plotϕshallow, logplotϕ}}, ImageSize → 800],
 PlotLabel -> "Infinite 3D, isotropic point source,
     Isotropic scattering, Chi-3 random flight - correlated
     emission\nCollision-rate density Cc[r], c = " <> ToString[c]]
```

Out[3573]=



Moments

In[3337]:= $pc[s\_] := \frac{e^{-\frac{s^2}{4}} s^2}{2\sqrt{\pi}}$

correlation emission

collision rate

In[3338]:= $m0Cc[c\_] := \frac{1}{1-c}$

In[3339]:= $m2Cc[c\_, s\_, s2\_, g\_] := \frac{s2}{(1-c)^2}\left(1 + c\,g\,\frac{2\,s^2}{s2\,(1-c\,g)}\right)$

fluence

In[ ]:= $m0\phi c[c\_, s\_] := \frac{s}{1-c}$

In[ ]:= $m2\phi c[c\_, s\_, s2\_, s3\_, g\_] := \frac{\left(s3\,(1-c)\,(1-g\,c) + 3\,c\,s\left(2\,g\,c\,s^2 + s2\left(-2\,g\,c + g + 1\right)\right)\right)}{3\,(1-c)^2\,(1-c\,g)}$

In[3340]:= `simsC = simulations;`

In[3574]:=
```
m0Ccs = {#[[1]], #[[-1, 8, 1]]} & /@ simsC;
m2Ccs = {#[[1]], #[[-1, 8, 3]]} & /@ simsC;
m0ϕcs = {#[[1]], #[[-1, 10, 1]]} & /@ simsC;
m2ϕcs = {#[[1]], #[[-1, 10, 3]]} & /@ simsC;
mfp = Integrate[pc[s] s, {s, 0, Infinity}];
mfp2 = Integrate[pc[s] s s, {s, 0, Infinity}];
mfp3 = Integrate[pc[s] s s s, {s, 0, Infinity}];
g = 0;
Show[
 LogPlot[{m0Cc[c], m2Cc[c, mfp, mfp2, g], m0ϕc[c, mfp],
   m2ϕc[c, mfp, mfp2, mfp3, g]}, {c, 0.01, 0.999}, PlotRange → All],
 ListLogPlot[{m0Ccs, m2Ccs, m0ϕcs, m2ϕcs}, PlotRange → All],
 PlotRange → All, Frame → True,
 FrameLabel → {{"moments",}, {c, "Chi-3 Correlated Emission 3D Isotropic
      scattering\nCollision-rate density moments"}}
]
```

Out[3582]=



Namespace

In[3359]:=
```
End[]
```

Out[3359]=
```
inf3DisopointIsotropcscatterChi3`
```